



The Isabelle Programming Tutorial (draft)

by Christian Urban with contributions from:

Stefan Berghofer
Jasmin Blanchette
Sascha Böhme
Jeremy Dawson
Alexander Krauss
Christian Sternagel

October 1, 2009

Contents

Contents	i
1 Introduction	1
1.1 Intended Audience and Prior Knowledge	1
1.2 Existing Documentation	1
1.3 Typographic Conventions	2
1.4 Aaaaargh! My Code Does not Work Anymore	3
1.5 Some Naming Conventions in the Isabelle Sources	4
1.6 Acknowledgements	4
2 First Steps	7
2.1 Including ML-Code	7
2.2 Debugging and Printing	9
2.3 Combinators	12
2.4 ML-Antiquotations	18
2.5 Storing Data in Isabelle (TBD)	21
3 Isabelle – The Good, the Bad and the Ugly	23
3.1 Terms and Types	23
3.2 Constructing Terms and Types Manually	26
3.3 Type-Checking	33
3.4 Theorems	35
3.5 Setups (TBD)	37
3.6 Theorem Attributes	38
3.7 Theories, Contexts and Local Theories (TBD)	44
3.8 Storing Theorems (TBD)	44
3.9 Pretty-Printing	44
3.10 Misc (TBD)	49
3.11 Name Space Issues (TBD)	50
4 Parsing	53

4.1	Building Generic Parsers	53
4.2	Parsing Theory Syntax	59
4.3	Parsers for ML-Code (TBD)	63
4.4	Context Parser (TBD)	63
4.5	Argument and Attribute Parsers (TBD)	64
4.6	Parsing Inner Syntax	64
4.7	Parsing Specifications	64
4.8	New Commands and Keyword Files	67
4.9	Methods (TBD)	71
5	Tactical Reasoning	73
5.1	Basics of Reasoning with Tactics	73
5.2	Simple Tactics	78
5.3	Tactic Combinators	84
5.4	Simplifier Tactics	89
5.5	Simprocs	96
5.6	Conversions	100
5.7	Declarations (TBD)	106
5.8	Structured Proofs (TBD)	106
6	How to Write a Definitional Package	109
6.1	Preliminaries	109
6.2	Parsing and Typing the Specification	114
6.3	The Code in a Nutshell	117
6.4	The Gory Details	121
6.5	Extensions of the Package (TBD)	137
6.6	Definitional Packages	138
A	Recipes	139
A.1	Useful Document Antiquotations	139
A.2	Restricting the Runtime of a Function	142
A.3	Measuring Time	142
A.4	Configuration Options	143
A.5	Storing Data (TBD)	145
A.6	Executing an External Application (TBD)	146
A.7	Writing an Oracle (TBD)	146
A.8	SAT Solvers	148
A.9	User Space Type-Systems (TBD)	150
B	Solutions to Most Exercises	151

<i>CONTENTS</i>	iii
C Comments for Authors	157
Bibliography	161
Index	162

Chapter 1

Introduction

“My thesis is that programming is not at the bottom of the intellectual pyramid, but at the top. It’s creative design of the highest order. It isn’t monkey or donkey work; rather, as Edsger Dijkstra famously claimed, it’s amongst the hardest intellectual tasks ever attempted.”

Richard Bornat, In *Defence of Programming*. [1]

If your next project requires you to program on the ML-level of Isabelle, then this tutorial is for you. It will guide you through the first steps of Isabelle programming, and also explain tricks of the trade. The best way to get to know the ML-level of Isabelle is by experimenting with the many code examples included in the tutorial. The code is as far as possible checked against the Isabelle distribution.¹ If something does not work, then please let us know. It is impossible for us to know every environment, operating system or editor in which Isabelle is used. If you have comments, criticism or like to add to the tutorial, please feel free—you are most welcome! The tutorial is meant to be gentle and comprehensive. To achieve this we need your feedback.

1.1 Intended Audience and Prior Knowledge

This tutorial targets readers who already know how to use Isabelle for writing theories and proofs. We also assume that readers are familiar with the functional programming language ML, the language in which most of Isabelle is implemented. If you are unfamiliar with either of these two subjects, you should first work through the Isabelle/HOL tutorial [4] or Paulson’s book on ML [6].

1.2 Existing Documentation

The following documentation about Isabelle programming already exists (and is part of the distribution of Isabelle):

¹Isabelle repository snapshot f126e68d003d (29-Sep-2009)

The Isabelle/Isar Implementation Manual describes Isabelle from a high-level perspective, documenting both the underlying concepts and some of the interfaces.

The Isabelle Reference Manual is an older document that used to be the main reference of Isabelle at a time when all proof scripts were written on the ML-level. Many parts of this manual are outdated now, but some parts, particularly the chapters on tactics, are still useful.

The Isar Reference Manual provides specification material (like grammars, examples and so on) about Isar and its implementation.

Then of course there are:

The Isabelle sources. They are the ultimate reference for how things really work. Therefore you should not hesitate to look at the way things are actually implemented. More importantly, it is often good to look at code that does similar things as you want to do and learn from it. The UNIX command `grep -R` is often your best friend while programming with Isabelle, or hypersearch if you program using jEdit under MacOSX. To understand the sources, it is often also necessary to track the change history of a file or files. The Mercurial repository² for Isabelle provides convenient interfaces to query the history of files and “change sets”.

1.3 Typographic Conventions

All ML-code in this tutorial is typeset in shaded boxes, like the following ML-expression:

```
ML {*  
  3 + 4  
*}
```

These boxes correspond to how code can be processed inside the interactive environment of Isabelle. It is therefore easy to experiment with what is displayed. However, for better readability we will drop the enclosing `ML {* ... *}` and just write:

```
3 + 4
```

Whenever appropriate we also show the response the code generates when evaluated. This response is prefixed with a `>`, like:

```
3 + 4  
> 7
```

²<http://isabelle.in.tum.de/repos/isabelle/>

The user-level commands of Isabelle (i.e., the non-ML code) are written in **bold face** (e.g., **lemma**, **apply**, **foobar** and so on). We use `$...` to indicate that a command needs to be run in a UNIX-shell, for example:

```
$ grep -R ThyOutput *
```

Pointers to further information and Isabelle files are typeset in *italic* and highlighted as follows:

Read More

Further information or pointers to files.

The pointers to Isabelle files are hyperlinked to the tip of the Mercurial repository of Isabelle at <http://isabelle.in.tum.de/repos/isabelle/>.

A few exercises are scattered around the text. Their solutions are given in Appendix B. Of course, you learn most, if you first try to solve the exercises on your own, and then look at the solutions.

1.4 Aaaaargh! My Code Does not Work Anymore

One unpleasant aspect of any code development inside a larger system is that one has to aim at a “moving target”. The Isabelle system is no exception. Every update lets potentially all hell break loose, because other developers have changed code you are relying on. Cursing is somewhat helpful in such situations, but taking the view that incompatible code changes are a fact of life might be more gratifying. Isabelle is a research project. In most circumstances it is just impossible to make research backward compatible (imagine Darwin attempting to make the Theory of Evolution backward compatible).

However, there are a few steps you can take to mitigate unwanted interferences with code changes from other developers. First, you can base your code on the latest stable release of Isabelle (it is aimed to have one such release at least once every year). This might cut you off from the latest feature implemented in Isabelle, but at least you do not have to track side-steps or dead-ends in the Isabelle development. Of course this means also you have to synchronise your code at the next stable release. Another possibility is to get your code into the Isabelle distribution. For this you have to convince other developers that your code or project is of general interest. If you managed to do this, then the problem of the moving target goes away, because when checking in new code, developers are strongly urged to test it against Isabelle’s code base. If your project is part of that code base, then maintenance is done by others. Unfortunately, this might not be a helpful advice for all types of projects. A lower threshold for inclusion has the Archive of Formalised Proofs, short AFP.³ This archive has been created mainly for formalisations that are interesting but not necessarily of general interest. If you have ML-code as part of a formalisation, then this might be the right place for you. There is no problem with updating your code after submission. At the moment developers are not as diligent with checking

³<http://afp.sourceforge.net/>

their code against the AFP than with checking against the distribution, but generally problems will be caught and the developer, who caused them, is expected to fix them. So also in this case code maintenance is done for you.

1.5 Some Naming Conventions in the Isabelle Sources

There are a few naming conventions in the Isabelle code that might aid reading and writing code. (Remember that code is written once, but read many times.) The most important conventions are:

- *t, u, trm* for (raw) terms; ML-type: *term*
- *ct, cu* for certified terms; ML-type: *cterm*
- *ty, T, U* for (raw) types; ML-type: *typ*
- *th, thm* for theorems; ML-type: *thm*
- *foo_tac* for tactics; ML-type: *tactic*
- *thy* for theories; ML-type: *theory*
- *ctxt* for proof contexts; ML-type: *Proof.context*
- *lthy* for local theories; ML-type: *local_theory*
- *context* for generic contexts; ML-type *Context.generic*
- *mx* for mixfix syntax annotations; ML-type *mixfix*
- *pvt* for pretty printing; ML-type *Pretty.T*

1.6 Acknowledgements

Financial support for this tutorial was provided by the German Research Council (DFG) under grant number URB 165/5-1. The following people contributed to the text:

- **Stefan Berghofer** wrote nearly all of the ML-code of the **simple inductive**-package and the code for the *chunk*-antiquotation. He also wrote the first version of this chapter describing the package and has been helpful *beyond measure* with answering questions about Isabelle.
- **Jasmin Blanchette** helped greatly with section 3.9.
- **Sascha Böhme** contributed the recipes in A.2, A.4, A.5, A.6 and A.7. He also wrote section 5.6 and helped with recipe A.3.
- **Jeremy Dawson** wrote the first version of the chapter about parsing.

- **Armin Heller** helped with recipe [A.8](#).
- **Alexander Krauss** wrote the first version of the “first-steps” chapter and also contributed the material on `Named_Thms`.
- **Christian Sternagel** proofread the tutorial and made many improvements to the text.

Please let me know of any omissions. Responsibility for any remaining errors lies with me.

This document is still in the process of being written! All of the text is still under construction. Sections and chapters that are under heavy construction are marked with TBD.

This document (version 2009-09-29) was compiled with:
Isabelle repository snapshot f126e68d003d (29-Sep-2009)
Poly/ML 5.2 Release RTS version: I386-5.2.1

Chapter 2

First Steps

“We will most likely never realize the full importance of painting the Tower, that it is the essential element in the conservation of metal works and the more meticulous the paint job, the longer the Tower shall endure.”

Gustave Eiffel, In his book *The 300-Meter Tower*.¹

Isabelle programming is done in ML. Just like lemmas and proofs, ML-code for Isabelle must be part of a theory. If you want to follow the code given in this chapter, we assume you are working inside the theory starting with

```
theory FirstSteps
imports Main
begin
...
```

We also generally assume you are working with the logic HOL. The examples that will be given might need to be adapted if you work in a different logic.

2.1 Including ML-Code

The easiest and quickest way to include code in a theory is by using the **ML**-command. For example:

```
ML {*
    3 + 4
*}
> 7
```

Like normal Isabelle scripts, **ML**-commands can be evaluated by using the advance and undo buttons of your Isabelle environment. The code inside the **ML**-command can also contain value and function bindings, for example

¹The Eiffel Tower has been re-painted 18 times since its initial construction, an average of once every seven years. It takes more than a year for a team of 25 painters to paint the Tower from top to bottom.

```
ML {*
  val r = ref 0
  fun f n = n + 1
*}
```

and even those can be undone when the proof script is retracted. As mentioned in the Introduction, we will drop the `ML {* ... *}` scaffolding whenever we show code. The lines prefixed with `>` are not part of the code, rather they indicate what the response is when the code is evaluated. There are also the commands `ML_val` and `ML_prf` for including ML-code. The first evaluates the given code, but any effect on the theory, in which the code is embedded, is suppressed. The second needs to be used if ML-code is defined inside a proof. For example

```
lemma test:
shows "True"
ML_prf {* writeln "Trivial!" *}
oops
```

However, both commands will only play minor roles in this tutorial (we will always arrange that the ML-code is defined outside of proofs, for example).

Once a portion of code is relatively stable, you usually want to export it to a separate ML-file. Such files can then be included somewhere inside a theory by using the command `use`. For example

```
theory FirstSteps
imports Main
uses ("file_to_be_included.ML") ...
begin
...
use "file_to_be_included.ML"
...

```

The `uses`-command in the header of the theory is needed in order to indicate the dependency of the theory on the ML-file. Alternatively, the file can be included by just writing in the header

```
theory FirstSteps
imports Main
uses "file_to_be_included.ML" ...
begin
...

```

Note that no parentheses are given this time. Note also that the included ML-file should not contain any `use` itself. Otherwise Isabelle is unable to record all file dependencies, which is a nuisance if you have to track down errors.

2.2 Debugging and Printing

During development you might find it necessary to inspect some data in your code. This can be done in a “quick-and-dirty” fashion using the function `writeln`. For example

```
writeln "any string"
> "any string"
```

will print out `"any string"` inside the response buffer of Isabelle. This function expects a string as argument. If you develop under PolyML, then there is a convenient, though again “quick-and-dirty”, method for converting values into strings, namely the function `makestring`:

```
writeln (PolyML.makestring 1)
> "1"
```

However, `makestring` only works if the type of what is converted is monomorphic and not a function.

The function `writeln` should only be used for testing purposes, because any output this function generates will be overwritten as soon as an error is raised. For printing anything more serious and elaborate, the function `tracing` is more appropriate. This function writes all output into a separate tracing buffer. For example:

```
tracing "foo"
> "foo"
```

It is also possible to redirect the “channel” where the string `foo` is printed to a separate file, e.g., to prevent ProofGeneral from choking on massive amounts of trace output. This redirection can be achieved with the code:

```
val strip_specials =
let
  fun strip ("^\^A" :: _ :: cs) = strip cs
    | strip (c :: cs) = c :: strip cs
    | strip [] = [];
in
  implode o strip o explode
end

fun redirect_tracing stream =
  Output.tracing_fn := (fn s =>
    (TextIO.output (stream, (strip_specials s));
     TextIO.output (stream, "\n");
     TextIO.flushOut stream))
```

Calling now

```
redirect_tracing (TextIO.openOut "foo.bar")
```

will cause that all tracing information is printed into the file `foo.bar`.

You can print out error messages with the function `error`; for example:

```
if 0=1 then true else (error "foo")
> Exception- ERROR "foo" raised
> At command "ML".
```

This function raises the exception `ERROR`, which will then be displayed by the infrastructure.

(FIXME Mention how to work with `debug` and `profiling`.)

Most often you want to inspect data of Isabelle's most basic data structures, namely `term`, `cterm` and `thm`. Isabelle contains elaborate pretty-printing functions for printing them (see Section 3.9), but for quick-and-dirty solutions they are a bit unwieldy. One way to transform a term into a string is to use the function `string_of_term` in structure `Syntax`, which we bind for more convenience to the toplevel.

```
val string_of_term = Syntax.string_of_term
```

It can now be used as follows

```
string_of_term @{context} @{term "1::nat"}
> "\^E\^Fterm\^E\^E\^Fconst\^Fname=HOL.one_class.one\^E1\^E\^F\^E\^E\^F\^E"
```

This produces a string corresponding to the term `1::nat` with some additional information encoded in it. The string can be properly printed by using either the function `writeln` or `tracing`:

```
writeln (string_of_term @{context} @{term "1::nat"})
> "1"
```

or

```
tracing (string_of_term @{context} @{term "1::nat"})
> "1"
```

If there are more than one `terms` to be printed, you can use the function `commas` to separate them.

```
fun string_of_terms ctxt ts =
  commas (map (string_of_term ctxt) ts)
```

A `cterm` can be transformed into a string by the following function.


```
fun string_of_cterm ctxt ct =
  string_of_term ctxt (term_of ct)
```

In this example the function `term_of` extracts the `term` from a `cterm`. More than one `cterms` can again be printed with `commas`.

```
fun string_of_cterms ctxt cts =
  commas (map (string_of_cterm ctxt) cts)
```

The easiest way to get the string of a theorem is to transform it into a `term` using the function `prop_of`.

```
fun string_of_thm ctxt thm =
  string_of_term ctxt (prop_of thm)
```

Theorems also include schematic variables, such as `?P`, `?Q` and so on. They are needed in order to be able to instantiate theorems when they are applied. For example the theorem `conjI` shown below can be used for any (typable) instantiation of `?P` and `?Q`

```
tracing (string_of_thm @{context} @{thm conjI})
> [[?P; ?Q]]  $\implies$  ?P  $\wedge$  ?Q
```

However, in order to improve the readability when printing theorems, we convert these schematic variables into free variables using the function `import`. This is similar to statements like `conjI[no_vars]` from Isabelle's user-level.

```
fun no_vars ctxt thm =
  let
    val ((_, [thm']), _) = Variable.import true [thm] ctxt
  in
    thm'
  end

fun string_of_thm_no_vars ctxt thm =
  string_of_term ctxt (prop_of (no_vars ctxt thm))
```

Theorem `conjI` is now printed as follows:

```
tracing (string_of_thm_no_vars @{context} @{thm conjI})
> [[P; Q]]  $\implies$  P  $\wedge$  Q
```

Again the function `commas` helps with printing more than one theorem.

```

fun string_of_thms ctxt thms =
  commas (map (string_of_thm ctxt) thms)

fun string_of_thms_no_vars ctxt thms =
  commas (map (string_of_thm_no_vars ctxt) thms)

```

Note, that when printing out several parcels of information that semantically belong together, like a warning message consisting for example of a term and a type, you should try to keep this information together in a single string. Therefore do not print out information as

```

tracing "First half,";
tracing "and second half."
> First half,
> and second half.

```

but as a single string with appropriate formatting. For example

```

tracing ("First half," ^ "\n" ^ "and second half.")
> First half,
> and second half.

```

To ease this kind of string manipulations, there are a number of library functions. For example, the function `cat_lines` concatenates a list of strings and inserts newlines.

```

cat_lines ["foo", "bar"]
> "foo\nbar"

```

Section 3.9 will also explain infrastructure that helps with more elaborate pretty printing.

2.3 Combinators

For beginners perhaps the most puzzling parts in the existing code of Isabelle are the combinators. At first they seem to greatly obstruct the comprehension of the code, but after getting familiar with them, they actually ease the understanding and also the programming.

The simplest combinator is I , which is just the identity function defined as

```

fun I x = x

```

Another simple combinator is K , defined as

```
fun K x = fn _ => x
```

K “wraps” a function around the argument x . However, this function ignores its argument. As a result, K defines a constant function always returning x .

The next combinator is reverse application, $/>$, defined as:

```
fun x /> f = f x
```

While just syntactic sugar for the usual function application, the purpose of this combinator is to implement functions in a “waterfall fashion”. Consider for example the function

```
1 fun inc_by_five x =
2   x /> (fn x => x + 1)
3     /> (fn x => (x, x))
4     /> fst
5     /> (fn x => x + 4)
```

which increments its argument x by 5. It proceeds by first incrementing the argument by 1 (Line 2); then storing the result in a pair (Line 3); taking the first component of the pair (Line 4) and finally incrementing the first component by 4 (Line 5). This kind of cascading manipulations of values is quite common when dealing with theories (for example by adding a definition, followed by lemmas and so on). The reverse application allows you to read what happens in a top-down manner. This kind of coding should also be familiar, if you have been exposed to Haskell’s *do*-notation. Writing the function `inc_by_five` using the reverse application is much clearer than writing

```
fun inc_by_five x = fst ((fn x => (x, x)) (x + 1)) + 4
```

or

```
fun inc_by_five x =
  ((fn x => x + 4) o fst o (fn x => (x, x)) o (fn x => x + 1)) x
```

and typographically more economical than

```
fun inc_by_five x =
let val y1 = x + 1
    val y2 = (y1, y1)
    val y3 = fst y2
    val y4 = y3 + 4
in y4 end
```

Another reason why the let-bindings in the code above are better to be avoided: it is more than easy to get the intermediate values wrong, not to mention the nightmares the maintenance of this code causes!

In Isabelle, a “real world” example for a function written in the waterfall fashion might be the following code:

```

1 fun apply_fresh_args f ctxt =
2   f |> fastype_of
3     |> binder_types
4     |> map (pair "z")
5     |> Variable.variant_frees ctxt [f]
6     |> map Free
7     |> curry list_comb f

```

This function takes a term and a context as argument. If the term is of function type, then `apply_fresh_args` returns the term with distinct variables applied to it. For example below variables are applied to the term $P::nat \Rightarrow int \Rightarrow unit \Rightarrow bool$:

```

let
  val t = @{term "P::nat ⇒ int ⇒ unit ⇒ bool"}
  val ctxt = @{context}
in
  apply_fresh_args t ctxt
  |> string_of_term ctxt
  |> tracing
end
> P z za zb

```

You can read off this behaviour from how `apply_fresh_args` is coded: in Line 2, the function `fastype_of` calculates the type of the term; `binder_types` in the next line produces the list of argument types (in the case above the list `[nat, int, unit]`); Line 4 pairs up each type with the string `z`; the function `variant_frees` generates for each `z` a unique name avoiding the given `f`; the list of name-type pairs is turned into a list of variable terms in Line 6, which in the last line is applied by the function `list_comb` to the term. In this last step we have to use the function `curry`, because `list_comb` expects the function and the variables list as a pair. This kind of functions is often needed when constructing terms with fresh variables. The infrastructure helps tremendously to avoid any name clashes. Consider for example:

```

let
  val t = @{term "za::'a ⇒ 'b ⇒ 'c"}
  val ctxt = @{context}
in
  apply_fresh_args t ctxt
  |> string_of_term ctxt
  |> tracing
end
> za z zb

```

where the *za* is correctly avoided.

The combinator `#>` is the reverse function composition. It can be used to define the following function

```
val inc_by_six =
  (fn x => x + 1)
#> (fn x => x + 2)
#> (fn x => x + 3)
```

which is the function composed of first the increment-by-one function and then increment-by-two, followed by increment-by-three. Again, the reverse function composition allows you to read the code top-down.

The remaining combinators described in this section add convenience for the “waterfall method” of writing functions. The combinator `tap` allows you to get hold of an intermediate result (to do some side-calculations for instance). The function

```
1 fun inc_by_three x =
2   x |> (fn x => x + 1)
3     |> tap (fn x => tracing (PolyML.makestring x))
4     |> (fn x => x + 2)
```

increments the argument first by 1 and then by 2. In the middle (Line 3), however, it uses `tap` for printing the “plus-one” intermediate result inside the tracing buffer. The function `tap` can only be used for side-calculations, because any value that is computed cannot be merged back into the “main waterfall”. To do this, you can use the next combinator.

The combinator `'` (a backtick) is similar to `tap`, but applies a function to the value and returns the result together with the value (as a pair). For example the function

```
fun inc_as_pair x =
  x |> '(fn x => x + 1)
    |> (fn (x, y) => (x, y + 1))
```

takes *x* as argument, and then increments *x*, but also keeps *x*. The intermediate result is therefore the pair $(x + 1, x)$. After that, the function increments the right-hand component of the pair. So finally the result will be $(x + 1, x + 1)$.

The combinators `/>>` and `//>` are defined for functions manipulating pairs. The first applies the function to the first component of the pair, defined as

```
fun (x, y) />> f = (f x, y)
```

and the second combinator to the second component, defined as

```
fun (x, y) ||> f = (x, f y)
```

These two functions can, for example, be used to avoid explicit *lets* for intermediate values in functions that return pairs. As an example, suppose you want to separate a list of integers into two lists according to a treshold. If the treshold is 5, the list $[1, 6, 2, 5, 3, 4]$ should be separated to $([1, 2, 3, 4], [6, 5])$. This function can be implemented as

```
fun separate i [] = ([], [])
  | separate i (x::xs) =
    let
      val (los, grs) = separate i xs
    in
      if i <= x then (los, x::grs) else (x::los, grs)
    end
```

where however the return value of the recursive call is bound explicitly to the pair (los, grs) . You can implement this function more concisely as

```
fun separate i [] = ([], [])
  | separate i (x::xs) =
    if i <= x
    then separate i xs ||> cons x
    else separate i xs |>> cons x
```

avoiding the explicit *let*. While in this example the gain in conciseness is only small, in more complicated situations the benefit of avoiding *lets* can be substantial.

With the combinator *|->* you can re-combine the elements from a pair. This combinator is defined as

```
fun (x, y) |-> f = f x y
```

and can be used to write the following roundabout version of the *double* function:

```
fun double x =
  x |> (fn x => (x, x))
    |-> (fn x => fn y => x + y)
```

The combinator *||>>* plays a central rôle whenever your task is to update a theory and the update also produces a side-result (for example a theorem). Functions for such tasks return a pair whose second component is the theory and the first component is the side-result. Using *||>>*, you can do conveniently the update and also accumulate the side-results. Consider the following simple function.

```

1 fun acc_incs x =
2   x /> (fn x => ("", x))
3     ||>> (fn x => (x, x + 1))
4     ||>> (fn x => (x, x + 1))
5     ||>> (fn x => (x, x + 1))

```

The purpose of Line 2 is to just pair up the argument with a dummy value (since `||>>` operates on pairs). Each of the next three lines just increment the value by one, but also nest the intermediate results to the left. For example

```

acc_incs 1
> (((("", 1), 2), 3), 4)

```

You can continue this chain with:

```

acc_incs 1 ||>> (fn x => (x, x + 2))
> (((("(", 1), 2), 3), 4), 6)

```

(FIXME: maybe give a “real world” example for this combinator)

Recall that `/>` is the reverse function application. Recall also that the related reverse function composition is `#>`. In fact all the combinators `/->`, `/>>`, `||>` and `||>>` described above have related combinators for function composition, namely `#->`, `#>>`, `##>` and `##>>`. Using `#->`, for example, the function `double` can also be written as:

```

val double =
  (fn x => (x, x))
  #-> (fn x => fn y => x + y)

```

When using combinators for writing functions in waterfall fashion, it is sometimes necessary to do some “plumbing” in order to fit functions together. We have already seen such plumbing in the function `apply_fresh_args`, where `curry` is needed for making the function `list_comb` that works over pairs to fit with the combinator `/>`. Such plumbing is also needed in situations where a functions operate over lists, but one calculates only with a single element. An example is the function `check_terms`, whose purpose is to type-check a list of terms.

```

let
  val ctxt = @{context}
in
  map (Syntax.parse_term ctxt) ["m + n", "m * n", "m - (n::nat)"]
  /> Syntax.check_terms ctxt
  /> string_of_terms ctxt
  /> tracing
end
> m + n, m * n, m - n

```

In this example we obtain three terms in which m and n are of type nat . If you have only a single term, then `check_terms` needs plumbing. This can be done with the function `singleton`.² For example

```
let
  val ctxt = @{context}
in
  Syntax.parse_term ctxt "m - (n::nat)"
  /> singleton (Syntax.check_terms ctxt)
  /> string_of_term ctxt
  /> tracing
end
> m - n
```

Read More

The most frequently used combinators are defined in the files `Pure/library.ML` and `Pure/General/basics.ML`. Also [Impl. Man., Sec. B.1] contains further information about combinators.

(FIXME: find a good exercise for combinators)

(FIXME: say something about calling conventions)

2.4 ML-Antiquotations

The main advantage of embedding all code in a theory is that the code can contain references to entities defined on the logical level of Isabelle. By this we mean definitions, theorems, terms and so on. This kind of reference is realised in Isabelle with ML-antiquotations, often just called antiquotations.³ For example, one can print out the name of the current theory by typing

```
Context.theory_name @{theory}
> "FirstSteps"
```

where `@{theory}` is an antiquotation that is substituted with the current theory (remember that we assumed we are inside the theory `FirstSteps`). The name of this theory can be extracted using the function `theory_name`.

Note, however, that antiquotations are statically linked, that is their value is determined at “compile-time”, not “run-time”. For example the function

²There is already a function `check_term` in the Isabelle sources that is defined in terms of `singleton` and `check_terms`.

³There are two kinds of antiquotations in Isabelle, which have very different purposes and infrastructures. The first kind, described in this section, are *ML-antiquotations*. They are used to refer to entities (like terms, types etc) from Isabelle’s logic layer inside ML-code. The other kind of antiquotations are *document antiquotations*. They are used only in the text parts of Isabelle and their purpose is to print logical entities inside \LaTeX -documents. Document antiquotations are part of the user level and therefore we are not interested in them in this Tutorial, except in Appendix A.1 where we show how to implement your own document antiquotations.


```
fun not_current_thyname () = Context.theory_name @{theory}
```

does *not* return the name of the current theory, if it is run in a different theory. Instead, the code above defines the constant function that always returns the string "FirstSteps", no matter where the function is called. Operationally speaking, the antiquotation `@{theory}` is *not* replaced with code that will look up the current theory in some data structure and return it. Instead, it is literally replaced with the value representing the theory name.

In a similar way you can use antiquotations to refer to proved theorems: `@{thm ...}` for a single theorem

```
@{thm allI}
> ( $\bigwedge x. ?P x$ )  $\implies$   $\forall x. ?P x$ 
```

and `@{thms ...}` for more than one

```
@{thms conj_ac}
> ( $?P \wedge ?Q$ ) = ( $?Q \wedge ?P$ )
> ( $?P \wedge ?Q \wedge ?R$ ) = ( $?Q \wedge ?P \wedge ?R$ )
> ( $(?P \wedge ?Q) \wedge ?R$ ) = ( $?P \wedge ?Q \wedge ?R$ )
```

The point of these antiquotations is that referring to theorems in this way makes your code independent from what theorems the user might have stored under this name (this becomes especially important when you deal with theorem lists; see Section 3.6).

You can also refer to the current simpset via an antiquotation. To illustrate this we implement the function that extracts the theorem names stored in a simpset.

```
fun get_thm_names_from_ss simpset =
let
  val {simps,...} = MetaSimplifier.dest_ss simpset
in
  map #1 simps
end
```

The function `dest_ss` returns a record containing all information stored in the simpset, but we are only interested in the names of the simp-rules. Now you can feed in the current simpset into this function. The current simpset can be referred to using the antiquotation `@{simpset}`.

```
get_thm_names_from_ss @{simpset}
> ["Nat.of_nat_eq_id", "Int.of_int_eq_id", "Nat.One_nat_def", ...]
```

Again, this way of referencing simpsets makes you independent from additions of lemmas to the simpset by the user that can potentially cause loops in your code.

On the ML-level of Isabelle, you often have to work with qualified names. These are strings with some additional information, such as positional information and qualifiers. Such qualified names can be generated with the antiquotation `@{binding ...}`. For example

```
@{binding "name"}
> name
```

An example where a qualified name is needed is the function `define`. This function is used below to define the constant `TrueConj` as the conjunction `True ∧ True`.

```
local_setup {*
  LocalTheory.define Thm.internalK
    ((@{binding "TrueConj"}, NoSyn),
     (Attrib.empty_binding, @{term "True ∧ True"})) #> snd *}

```

Now querying the definition you obtain:

```
thm TrueConj_def
> TrueConj ≡ True ∧ True
```

(FIXME give a better example why bindings are important; maybe give a pointer to `local_setup`; if not, then explain why `snd` is needed)

It is also possible to define your own antiquotations. But you should exercise care when introducing new ones, as they can also make your code also difficult to read. In the next chapter we will see how the (build in) antiquotation `@{term ...}` can be used to construct terms. A restriction of this antiquotation is that it does not allow you to use schematic variables. If you want to have an antiquotation that does not have this restriction, you can implement your own using the function `ML_Antiquote.inline`. The code for the antiquotation `term_pat` is as follows.

```
1 let
2   val parser = Args.context -- Scan.lift Args.name_source
3
4   fun term_pat (ctxt, str) =
5     str |> ProofContext.read_term_pattern ctxt
6       |> ML_Syntax.print_term
7       |> ML_Syntax.atomic
8 in
9   ML_Antiquote.inline "term_pat" (parser >> term_pat)
10 end
```

The parser in Line 2 provides us with a context and a string; this string is transformed into a term using the function `read_term_pattern` (Line 5); the next two lines transform the term into a string so that the ML-system can understand them. An example for the usage of this antiquotation is:

```
@{term_pat "Suc (?x::nat)}
> Const ("Suc", "nat ⇒ nat") $ Var (("x", 0), "nat")
```

which shows the internal representation of the term `Suc ?x`.

Read More

The file `Pure/ML/ml_antiquote.ML` contains the definitions for most antiquotations. Most of the basic operations on ML-syntax are implemented in `Pure/ML/ml_syntax.ML`.

2.5 Storing Data in Isabelle (TBD)

Isabelle provides mechanisms to store (and retrieve) arbitrary data. Before we explain them, let us digress: Conventional wisdom has it that ML's type-system ensures that for example an `'a list` can only hold elements of the same type, namely `'a`. Still, by some arguably accidental features of ML, one can implement a universal type. In Isabelle it is implemented as `Universal.universal`. This type allows one to inject and project elements of different type into for example into a list. We will show this by storing an integer and boolean in a list. What is important that access to the data is done in a type-safe manner.

Let us first define projection and injection functions for booleans and integers.

```
local
  val fn_int = Universal.tag () : int Universal.tag
  val fn_bool = Universal.tag () : bool Universal.tag
in
  val inject_int = Universal.tagInject fn_int;
  val inject_bool = Universal.tagInject fn_bool;
  val project_int = Universal.tagProject fn_int;
  val project_bool = Universal.tagProject fn_bool
end
```

We can now build a list injecting 13 and true as its two elements.

```
val list = [inject_int 13, inject_bool true]

project_int (nth list 0)

project_bool (nth list 1)

Context.the_thread_data ();
Context.display_names @{theory};
Context.pretty_thy @{theory}
```

```
|> Pretty.string_of  
|> tracing
```

Read More[Pure/ML-Systems/universal.ML](#)

Chapter 3

Isabelle – The Good, the Bad and the Ugly

Isabelle is build around a few central ideas. One is the LCF-approach to theorem proving where there is a small trusted core and everything else is build on top of this trusted core.

3.1 Terms and Types

One way to construct Isabelle terms, is by using the antiquotation `@{term ...}`. For example

```
@{term "(a::nat) + b = c"}
> Const ("op =", ...) $
> (Const ("HOL.plus_class.plus", ...) $ ... $ ...) $ ...
```

will show the term $a + b = c$, but printed using the internal representation corresponding to the datatype `term` defined as follows:

```
1 datatype term =
2   Const of string * typ
3 | Free of string * typ
4 | Var of indexname * typ
5 | Bound of int
6 | Abs of string * typ * term
7 | $ of term * term
```

This datatype implements lambda-terms typed in Church-style. As can be seen in Line 5, terms use the usual de Bruijn index mechanism for representing bound variables. For example in

```
@{term "\x y. x y"}
> Abs ("x", "'a  $\Rightarrow$  'b", Abs ("y", "'a", Bound 1 $ Bound 0))
```

the indices refer to the number of Abstractions (*Abs*) that we need to skip until we hit the *Abs* that binds the corresponding variable. Constructing a term with dangling de Bruijn indices is possible, but will be flagged as ill-formed when you try to typecheck or certify it (see Section 3.3). Note that the names of bound variables are kept at abstractions for printing purposes, and so should be treated only as “comments”. Application in Isabelle is realised with the term-constructor `$`.

Isabelle makes a distinction between *free* variables (term-constructor *Free* and written on the user level in blue colour) and *schematic* variables (term-constructor *Var* and written with a leading question mark). Consider the following two examples

```
let
  val v1 = Var ("x", 3), @{typ bool}
  val v2 = Var ("x1", 3), @{typ bool}
  val v3 = Free ("x", @{typ bool})
in
  string_of_terms @{context} [v1, v2, v3]
  /> tracing
end
> ?x3, ?x1.3, x
```

When constructing terms, you are usually concerned with free variables (as mentioned earlier, you cannot construct schematic variables using the antiquotation `@{term ...}`). If you deal with theorems, you have to, however, observe the distinction. The reason is that only schematic variables can be instantiated with terms when a theorem is applied. A similar distinction between free and schematic variables holds for types (see below).

Read More

Terms and types are described in detail in [Impl. Man., Sec. 2.2]. Their definition and many useful operations are implemented in [Pure/term.ML](#). For constructing terms involving HOL constants, many helper functions are defined in [HOL/Tools/hologic.ML](#).

Constructing terms via antiquotations has the advantage that only typable terms can be constructed. For example

```
@{term "x x"}
> Type unification failed: Occurs check!
```

raises a typing error, while it perfectly ok to construct the term

```
let
  val omega = Free ("x", @{typ nat}) $ Free ("x", @{typ nat})
in
  tracing (string_of_term @{context} omega)
end
> x x
```

with the raw ML-constructors.

Sometimes the internal representation of terms can be surprisingly different from what you see at the user-level, because the layers of parsing/type-checking/pretty printing can be quite elaborate.

Exercise 3.1.1. *Look at the internal term representation of the following terms, and find out why they are represented like this:*

- $\text{case } x \text{ of } 0 \Rightarrow 0 \mid \text{Suc } y \Rightarrow y$
- $\lambda(x, y). P \ y \ x$
- $\{[x] \mid x \leq -2\}$

Hint: The third term is already quite big, and the pretty printer may omit parts of it by default. If you want to see all of it, you can use the following ML-function to set the printing depth to a higher value:

```
print_depth 50
```

The antiquotation `@{prop ...}` constructs terms by inserting the usually invisible `Trueprop`-coercions whenever necessary. Consider for example the pairs

```
(@{term "P x"}, @{prop "P x"})
> (Free ("P", ...) $ Free ("x", ...),
>  Const ("Trueprop", ...) $ (Free ("P", ...) $ Free ("x", ...)))
```

where a coercion is inserted in the second component and

```
(@{term "P x ==> Q x"}, @{prop "P x ==> Q x"})
> (Const ("==>", ...) $ ... $ ...,
>  Const ("==>", ...) $ ... $ ...)
```

where it is not (since it is already constructed by a meta-implication). The purpose of the `Trueprop`-coercion is to embed formulae of an object logic, for example HOL, into the meta-logic of Isabelle. It is needed whenever a term is constructed that will be proved as a theorem.

As already seen above, types can be constructed using the antiquotation `@{typ ...}`. For example:

```
@{typ "bool => nat"}
> bool => nat
```

The corresponding datatype is

```
datatype typ =
  Type of string * typ list
| TFree of string * sort
| TVar of indexname * sort
```

Like with terms, there is the distinction between free type variables (term-constructor *TFree* and schematic type variables (term-constructor *TVar*). A type constant, like *int* or *bool*, are types with an empty list of argument types. However, it is a bit difficult to show an example, because Isabelle always pretty-prints types (unlike terms). Here is a contrived example:

```
if Type ("bool", []) = @{typ "bool"} then true else false
> true
```

Read More

Types are described in detail in [Impl. Man., Sec. 2.1]. Their definition and many useful operations are implemented in [Pure/type.ML](#).

3.2 Constructing Terms and Types Manually

While antiquotations are very convenient for constructing terms, they can only construct fixed terms (remember they are “linked” at compile-time). However, you often need to construct terms dynamically. For example, a function that returns the implication $\wedge(x::nat). P\ x \implies Q\ x$ taking *P* and *Q* as arguments can only be written as:

```
fun make_imp P Q =
let
  val x = Free ("x", @{typ nat})
in
  Logic.all x (Logic.mk_implies (P $ x, Q $ x))
end
```

The reason is that you cannot pass the arguments *P* and *Q* into an antiquotation.¹ For example the following does *not* work.

```
fun make_wrong_imp P Q = @{prop "\wedge(x::nat). P x \implies Q x"}
```

To see this, apply `@{term S}` and `@{term T}` to both functions. With *make_imp* you obtain the intended term involving the given arguments

¹At least not at the moment.


```

make_imp @{term S} @{term T}
> Const ... $
>   Abs ("x", Type ("nat", []),
>     Const ... $ (Free ("S",...) $ ...) $ (Free ("T",...) $ ...))

```

whereas with `make_wrong_imp` you obtain a term involving the P and Q from the antiquoteation.

```

make_wrong_imp @{term S} @{term T}
> Const ... $
>   Abs ("x", ...,
>     Const ... $ (Free ("P",...) $ ...) $
>     (Const ... $ (Free ("Q",...) $ ...)))

```

There are a number of handy functions that are frequently used for constructing terms. One is the function `list_comb`, which takes a term and a list of terms as arguments, and produces as output the term list applied to the term. For example

```

let
  val trm = @{term "P::nat"}
  val args = [ @{term "True"}, @{term "False"} ]
in
  list_comb (trm, args)
end
> Free ("P", "nat") $ Const ("True", "bool") $ Const ("False", "bool")

```

Another handy function is `lambda`, which abstracts a variable in a term. For example

```

let
  val x_nat = @{term "x::nat"}
  val trm = @{term "(P::nat ⇒ bool) x"}
in
  lambda x_nat trm
end
> Abs ("x", "nat", Free ("P", "bool ⇒ bool") $ Bound 0)

```

In this example, `lambda` produces a de Bruijn index (i.e. `Bound 0`), and an abstraction. It also records the type of the abstracted variable and for printing purposes also its name. Note that because of the typing annotation on P , the variable x in P is of the same type as the abstracted variable. If it is of different type, as in

```

let
  val x_int = @{term "x::int"}
  val trm = @{term "(P::nat ⇒ bool) x"}
in
  lambda x_int trm
end
> Abs ("x", "int", Free ("P", "nat ⇒ bool") $ Free ("x", "nat"))

```

then the variable *Free* ("x", "int") is *not* abstracted. This is a fundamental principle of Church-style typing, where variables with the same name still differ, if they have different type.

There is also the function *subst_free* with which terms can be replaced by other terms. For example below, we will replace in *f 0 x* the subterm *f 0* by *y*, and *x* by *True*.

```
let
  val sub1 = (@{term "(f::nat ⇒ nat ⇒ nat) 0"}, @{term "y::nat ⇒ nat"})
  val sub2 = (@{term "x::nat"}, @{term "True"})
  val trm = @{term "((f::nat ⇒ nat ⇒ nat) 0) x"}
in
  subst_free [sub1, sub2] trm
end
> Free ("y", "nat ⇒ nat") $ Const ("True", "bool")
```

As can be seen, *subst_free* does not take typability into account. However it takes alpha-equivalence into account:

```
let
  val sub = (@{term "(\λy::nat. y)"}, @{term "x::nat"})
  val trm = @{term "(\λx::nat. x)"}
in
  subst_free [sub] trm
end
> Free ("x", "nat")
```

Similarly the function *subst_bounds*, replaces lose bound variables with terms. To see how this function works, let us implement a function that strips off the outermost quantifiers in a term.

```
fun strip_all (Const ("All", _) $ Abs (n, T, t)) =
  strip_all t |>> cons (Free (n, T))
| strip_all t = ([], t)
```

The function returns a pair consisting of the stripped off variables and the body of the universal quantifications. For example

```
strip_all @{term "∀x y. x = (y::bool)"}
> ([Free ("x", "bool"), Free ("y", "bool")],
>  Const ("op =", ...) $ Bound 1 $ Bound 0)
```

After calling *strip_all*, you obtain a term with lose bound variables. With the function *subst_bounds*, you can replace these lose *Bounds* with the stripped off variables.

```

1 let
2   val (vrs, trm) = strip_all @term "∀ x y. x = (y::bool)"
3 in
4   subst_bounds (rev vrs, trm)
5   |> string_of_term @context
6   |> tracing
7 end
8 > x = y

```

Note that in Line 4 we had to reverse the list of variables that `strip_all` returned. The reason is that the head of the list the function `subst_bounds` takes is the replacement for *Bound 0*, the next element for *Bound 1* and so on.

There are many convenient functions that construct specific HOL-terms. For example `mk_eq` constructs an equality out of two terms. The types needed in this equality are calculated from the type of the arguments. For example

```

let
  val eq = Hologic.mk_eq (@term "True"}, @term "False"})
in
  string_of_term @context eq
  |> tracing
end
> True = False

```

Read More

There are many functions in [Pure/term.ML](#), [Pure/logic.ML](#) and [HOL/Tools/hologic.ML](#) that make such manual constructions of terms and types easier.

When constructing terms manually, there are a few subtle issues with constants. They usually crop up when pattern matching terms or types, or when constructing them. While it is perfectly ok to write the function `is_true` as follows

```

fun is_true @term True} = true
  | is_true _ = false

```

this does not work for picking out \forall -quantified terms. Because the function

```

fun is_all (@term All} $ _) = true
  | is_all _ = false

```

will not correctly match the formula `"∀ x::nat. P x"`:

```

is_all @term "∀ x::nat. P x"}
> false

```

The problem is that the `@term`-antiquotation in the pattern fixes the type of the constant `All` to be $(\text{'a} \Rightarrow \text{bool}) \Rightarrow \text{bool}$ for an arbitrary, but fixed type `'a`. A properly working alternative for this function is

```
fun is_all (Const ("All", _) $ _) = true
  | is_all _ = false
```

because now

```
is_all @{term "\x::nat. P x"}
> true
```

matches correctly (the first wildcard in the pattern matches any type and the second any term).

However there is still a problem: consider the similar function that attempts to pick out `Nil`-terms:

```
fun is_nil (Const ("Nil", _)) = true
  | is_nil _ = false
```

Unfortunately, also this function does *not* work as expected, since

```
is_nil @{term "Nil"}
> false
```

The problem is that on the ML-level the name of a constant is more subtle than you might expect. The function `is_all` worked correctly, because `All` is such a fundamental constant, which can be referenced by `Const ("All", some_type)`. However, if you look at

```
@{term "Nil"}
> Const ("List.list.Nil", ...)
```

the name of the constant `Nil` depends on the theory in which the term constructor is defined (`List`) and also in which datatype (`list`). Even worse, some constants have a name involving type-classes. Consider for example the constants for `zero` and `(op *)`:

```
(@{term "0::nat"}, @{term "(op *)"})
> (Const ("HOL.zero_class.zero", ...),
> Const ("HOL.times_class.times", ...))
```

While you could use the complete name, for example `Const ("List.list.Nil", some_type)`, for referring to or matching against `Nil`, this would make the code

rather brittle. The reason is that the theory and the name of the datatype can easily change. To make the code more robust, it is better to use the antiquotation `@{const_name ...}`. With this antiquotation you can harness the variable parts of the constant's name. Therefore a function for matching against constants that have a polymorphic type should be written as follows.

```
fun is_nil_or_all (Const (@{const_name "Nil"}, _) = true
  | is_nil_or_all (Const (@{const_name "All"}, _) $ _) = true
  | is_nil_or_all _ = false
```

The antiquotation for properly referencing type constants is `@{type_name ...}`. For example

```
@{type_name "list"}
> "List.list"
```

(FIXME: Explain the following better.)

Occasionally you have to calculate what the “base” name of a given constant is. For this you can use the function `Sign.extern_const` or `Long_Name.base_name`. For example:

```
Sign.extern_const @{theory} "List.list.Nil"
> "Nil"
```

The difference between both functions is that `extern_const` returns the smallest name that is still unique, whereas `base_name` always strips off all qualifiers.

Read More

Functions about naming are implemented in [Pure/General/name_space.ML](#); functions about signatures in [Pure/sign.ML](#).

Although types of terms can often be inferred, there are many situations where you need to construct types manually, especially when defining constants. For example the function returning a function type is as follows:

```
fun make_fun_type ty1 ty2 = Type ("fun", [ty1, ty2])
```

This can be equally written with the combinator `-->` as:

```
fun make_fun_type ty1 ty2 = ty1 --> ty2
```

If you want to construct a function type with more than one argument type, then you can use `---->`.

```
fun make_fun_types tys ty = tys ---> ty
```

A handy function for manipulating terms is `map_types`: it takes a function and applies it to every type in a term. You can, for example, change every `nat` in a term into an `int` using the function:

```
fun nat_to_int ty =
  (case ty of
   @{typ nat} => @{typ int}
  | Type (s, tys) => Type (s, map nat_to_int tys)
  | _ => ty)
```

Here is an example:

```
map_types nat_to_int @{term "a = (1::nat)"}
> Const ("op =", "int => int => bool")
> $ Free ("a", "int") $ Const ("HOL.one_class.one", "int")
```

If you want to obtain the list of free type-variables of a term, you can use the function `add_tfrees` (similarly `add_tvvars` for the schematic type-variables). One would expect that such functions take a term as input and return a list of types. But their type is actually

```
Term.term -> (string * Term.sort) list -> (string * Term.sort) list
```

that is they take, besides a term, also a list of type-variables as input. So in order to obtain the list of type-variables of a term you have to call them as follows

```
Term.add_tfrees @{term "(a, b)"} []
> [( "'b", ["HOL.type"]), ("'a", ["HOL.type"])]
```

The reason for this definition is that `add_tfrees` can be easily folded over a list of terms. Similarly for all functions named `add_*` in *Pure/term.ML*.

Exercise 3.2.1. Write a function `rev_sum` : `term -> term` that takes a term of the form $t_1 + t_2 + \dots + t_n$ (whereby n might be one) and returns the reversed sum $t_n + \dots + t_2 + t_1$. Assume the t_i can be arbitrary expressions and also note that `+` associates to the left. Try your function on some examples.

Exercise 3.2.2. Write a function which takes two terms representing natural numbers in unary notation (like `Suc (Suc (Suc 0))`), and produces the number representing their sum.

Exercise 3.2.3.² Implement the function, which we below name `deBruijn`, that depends on a natural number $n > 0$ and constructs terms of the form:

²Personal communication of de Bruijn to Dyckhoff.

$$\begin{aligned}
\text{rhs } n &\stackrel{\text{def}}{=} \bigwedge_{i=1..n} P_i \\
\text{lhs } n &\stackrel{\text{def}}{=} \bigwedge_{i=1..n} P_i = P (i + 1 \bmod n) \longrightarrow \text{rhs } n \\
\text{deBruijn } n &\stackrel{\text{def}}{=} \text{lhs } n \longrightarrow \text{rhs } n
\end{aligned}$$

For $n=3$ this function returns the term

$$\begin{aligned}
&(P\ 1 = P\ 2 \longrightarrow P\ 1 \wedge P\ 2 \wedge P\ 3) \wedge \\
&(P\ 2 = P\ 3 \longrightarrow P\ 1 \wedge P\ 2 \wedge P\ 3) \wedge \\
&(P\ 3 = P\ 1 \longrightarrow P\ 1 \wedge P\ 2 \wedge P\ 3) \longrightarrow P\ 1 \wedge P\ 2 \wedge P\ 3
\end{aligned}$$

Make sure you use the functions defined in `HOL/Tools/hologic.ML` for constructing the terms for the logical connectives.

3.3 Type-Checking

You can freely construct and manipulate *terms* and *types*, since they are just arbitrary unchecked trees. However, you eventually want to see if a term is well-formed, or type-checks, relative to a theory. Type-checking is done via the function `cterm_of`, which converts a *term* into a *cterm*, a *certified* term. Unlike *terms*, which are just trees, *cterms* are abstract objects that are guaranteed to be type-correct, and they can only be constructed via “official interfaces”.

Type-checking is always relative to a theory context. For now we use the `@{theory}` antiquotation to get hold of the current theory. For example you can write:

```
cterm_of @{theory} @{term "(a::nat) + b = c"}
> a + b = c
```

This can also be written with an antiquotation:

```
@{cterm "(a::nat) + b = c"}
> a + b = c
```

Attempting to obtain the certified term for

```
@{cterm "1 + True"}
> Type unification failed ...
```

yields an error (since the term is not typable). A slightly more elaborate example that type-checks is:

```

let
  val natT = @{typ "nat"}
  val zero = @{term "0::nat"}
in
  cterm_of @{theory}
    (Const (@{const_name plus}, natT --> natT --> natT) $ zero $ zero)
end
> 0 + 0

```

In Isabelle not just terms need to be certified, but also types. For example, you obtain the certified type for the Isabelle type $\text{nat} \Rightarrow \text{bool}$ on the ML-level as follows:

```

ctyp_of @{theory} (@{typ nat} --> @{typ bool})
> nat  $\Rightarrow$  bool

```

or with the antiquotation:

```

@{ctyp "nat  $\Rightarrow$  bool"}
> nat  $\Rightarrow$  bool

```

Read More

For functions related to *cterm*s and *ctyp*s see the file [Pure/thm.ML](#).

Remember Isabelle follows the Church-style typing for terms, i.e., a term contains enough typing information (constants, free variables and abstractions all have typing information) so that it is always clear what the type of a term is. Given a well-typed term, the function *type_of* returns the type of a term. Consider for example:

```

type_of (@{term "f::nat  $\Rightarrow$  bool"} $ @{term "x::nat"})
> bool

```

To calculate the type, this function traverses the whole term and will detect any typing inconsistency. For example changing the type of the variable *x* from *nat* to *int* will result in the error message:

```

type_of (@{term "f::nat  $\Rightarrow$  bool"} $ @{term "x::int"})
> *** Exception- TYPE ("type_of: type mismatch in application" ...

```

Since the complete traversal might sometimes be too costly and not necessary, there is the function *fastype_of*, which also returns the type of a term.

```

fastype_of (@{term "f::nat  $\Rightarrow$  bool"} $ @{term "x::nat"})
> bool

```

However, efficiency is gained on the expense of skipping some tests. You can see this in the following example


```
fastype_of (@{term "f::nat ⇒ bool"} $ @{term "x::int"})
> bool
```

where no error is detected.

Sometimes it is a bit inconvenient to construct a term with complete typing annotations, especially in cases where the typing information is redundant. A short-cut is to use the “place-holder” type *dummyT* and then let type-inference figure out the complete type. An example is as follows:

```
let
  val c = Const (@{const_name "plus"}, dummyT)
  val o = @{term "1::nat"}
  val v = Free ("x", dummyT)
in
  Syntax.check_term @{context} (c $ o $ v)
end
> Const ("HOL.plus_class.plus", "nat ⇒ nat ⇒ nat") $
> Const ("HOL.one_class.one", "nat") $ Free ("x", "nat")
```

Instead of giving explicitly the type for the constant *plus* and the free variable *x*, type-inference fills in the missing information.

Read More

See [Pure/Syntax/syntax.ML](#) where more functions about reading, checking and pretty-printing of terms are defined. Functions related to type-inference are implemented in [Pure/type.ML](#) and [Pure/type_infer.ML](#).

(FIXME: say something about sorts)

Exercise 3.3.1. Check that the function defined in Exercise 3.2.1 returns a result that type-checks. See what happens to the solutions of this exercise given in [B](#) when they receive an ill-typed term as input.

3.4 Theorems

Just like *cterm*s, theorems are abstract objects of type *thm* that can only be built by going through interfaces. As a consequence, every proof in Isabelle is correct by construction. This follows the tradition of the LCF approach [3].

To see theorems in “action”, let us give a proof on the ML-level for the following statement:

```
lemma
  assumes  $assm_1: "\bigwedge(x::nat). P x \implies Q x"$ 
  and  $assm_2: "P t"$ 
  shows  $"Q t"$ 
```

The corresponding ML-code is as follows:

```

let
  val assm1 = @{cprop "\(\(x:\text{nat}). P x \implies Q x\)"}
  val assm2 = @{cprop "(P:\text{nat}\implies\text{bool}) t"}

  val Pt_implies_Qt =
    assume assm1
    /> forall_elim @{cterm "t:\text{nat}"};

  val Qt = implies_elim Pt_implies_Qt (assume assm2);
in
  Qt
  /> implies_intr assm2
  /> implies_intr assm1
end
> [\(\(x. P x \implies Q x; P t)\) \implies Q t

```

This code-snippet constructs the following proof:

$$\frac{\frac{\frac{}{\Lambda x. P x \implies Q x \vdash \Lambda x. P x \implies Q x} (\text{assume})}{\Lambda x. P x \implies Q x \vdash P t \implies Q t} (\wedge\text{-elim})}{\Lambda x. P x \implies Q x, P t \vdash Q t} (\implies\text{-intro})}{\vdash [\Lambda x. P x \implies Q x; P t] \implies Q t} (\implies\text{-intro})$$

However, while we obtained a theorem as result, this theorem is not yet stored in Isabelle's theorem database. So it cannot be referenced later on. How to store theorems will be explained in Section 3.8.

Read More

For the functions `assume`, `forall_elim` etc see [Impl. Man., Sec. 2.3]. The basic functions for theorems are defined in `Pure/thm.ML`.

(FIXME: handy functions working on theorems, like `rulify` and so on)

(FIXME: `prove`)

(FIXME: how to add case-names to goal states - maybe in the next section)

(FIXME: example for how to add theorem styles)

```

fun strip_assums_all (params, Const("all",_) $ Abs(a, T, t)) =
  strip_assums_all ((a, T)::params, t)
  | strip_assums_all (params, B) = (params, B)

fun style_parm_premise i ctxt t =
  let val prems = Logic.strip_imp_premis t in
    if i <= length prems
    then let val (params,t) = strip_assums_all([], nth prems (i - 1))
          in subst_bounds(map Free params, t) end
    else error ("Not enough premises for prem" ^ string_of_int i ^

```

```

    " in proposition: " ^ string_of_term ctxt t)
  end;

strip_assums_all ([], @{term "\x y. A x y"})

```

```

setup {*
  TermStyle.add_style "no_all_prem1" (style_parm_premise 1) #>
  TermStyle.add_style "no_all_prem2" (style_parm_premise 2)
*}

```

```

lemma
  shows "A  $\implies$  B"
  and   "C  $\implies$  D"
oops

```

3.5 Setups (TBD)

In the previous section we used **setup** in order to make a theorem attribute known to Isabelle. What happens behind the scenes is that **setup** expects a function of type *theory* \rightarrow *theory*: the input theory is the current theory and the output the theory where the theory attribute has been stored.

This is a fundamental principle in Isabelle. A similar situation occurs for example with declaring constants. The function that declares a constant on the ML-level is `add_consts_i`. If you write³

```

Sign.add_consts_i [(@{binding "BAR"}, @{typ "nat"}, NoSyn)] @{theory}

```

for declaring the constant `BAR` with type `nat` and run the code, then you indeed obtain a theory as result. But if you query the constant on the Isabelle level using the command **term**

```

term "BAR"
> "BAR" :: "'a"

```

you do not obtain a constant of type `nat`, but a free variable (printed in blue) of polymorphic type. The problem is that the ML-expression above did not register the declaration with the current theory. This is what the command **setup** is for. The constant is properly declared with

³Recall that ML-code needs to be enclosed in `ML {* ... *}`.

```
setup {* Sign.add_consts_i [(@{binding "BAR"}, @{typ "nat"}, NoSyn)] *}
```

Now

```
term "BAR"
> "BAR" :: "nat"
```

returns a (black) constant with the type *nat*.

A similar command is **local_setup**, which expects a function of type *local_theory* \rightarrow *local_theory*. Later on we will also use the commands **method_setup** for installing methods in the current theory and **simproc_setup** for adding new simprocs to the current simpset.

3.6 Theorem Attributes

Theorem attributes are *[symmetric]*, *[THEN ...]*, *[simp]* and so on. Such attributes are *neither* tags *nor* flags annotated to theorems, but functions that do further processing once a theorem is proved. In particular, it is not possible to find out what are all theorems that have a given attribute in common, unless of course the function behind the attribute stores the theorems in a retrievable data structure.

If you want to print out all currently known attributes a theorem can have, you can use the Isabelle command

print_attributes

```
> COMP: direct composition with rules (no lifting)
> HOL.dest: declaration of Classical destruction rule
> HOL.elim: declaration of Classical elimination rule
> ...
```

The theorem attributes fall roughly into two categories: the first category manipulates the proved theorem (for example *[symmetric]* and *[THEN ...]*), and the second stores the proved theorem somewhere as data (for example *[simp]*, which adds the theorem to the current simpset).

To explain how to write your own attribute, let us start with an extremely simple version of the attribute *[symmetric]*. The purpose of this attribute is to produce the “symmetric” version of an equation. The main function behind this attribute is

```
val my_symmetric = Thm.rule_attribute (fn _ => fn thm => thm RS @{thm sym})
```

where the function *rule_attribute* expects a function taking a context (which we ignore in the code above) and a theorem (*thm*), and returns another theorem (namely *thm* resolved with the theorem *sym*: $s = t \implies t = s$; the function *RS* is explained in Section 5.2). The function *rule_attribute* then returns an attribute.

Before we can use the attribute, we need to set it up. This can be done using the Isabelle command **attribute_setup** as follows:

```
attribute_setup my_sym = {* Scan.succeed my_symmetric *}
  "applying the sym rule"
```

Inside the `{* ... *}`, we have to specify a parser for the theorem attribute. Since the attribute does not expect any further arguments (unlike `[THEN ...]`, for example), we use the parser `Scan.succeed`. Later on we will also consider attributes taking further arguments. An example for the attribute `[my_sym]` is the proof

```
lemma test[my_sym]: "2 = Suc (Suc 0)" by simp
```

which stores the theorem `Suc (Suc 0) = 2` under the name `test`. You can see this, if you query the lemma:

```
thm test
> Suc (Suc 0) = 2
```

We can also use the attribute when referring to this theorem:

```
thm test[my_sym]
> 2 = Suc (Suc 0)
```

An alternative for setting up an attribute is the function `setup`. So instead of using `attribute_setup`, you can also set up the attribute as follows:

```
Attrib.setup @{binding "my_sym"} (Scan.succeed my_symmetric)
  "applying the sym rule"
```

This gives a function from `Context.theory -> Context.theory`, which can be used for example with `setup`.

As an example of a slightly more complicated theorem attribute, we implement our own version of `[THEN ...]`. This attribute will take a list of theorems as argument and resolve the proved theorem with this list (one theorem after another). The code for this attribute is

```
fun MY_THEN thms =
  Thm.rule_attribute (fn _ => fn thm => foldl ((op RS) o swap) thm thms)
```

where `swap` swaps the components of a pair. The setup of this theorem attribute uses the parser `thms`, which parses a list of theorems.

```
attribute_setup MY_THEN = {* Attrib.thms >> MY_THEN *}
  "resolving the list of theorems with the proved theorem"
```

You can, for example, use this theorem attribute to turn an equation into a meta-equation:

```
thm test[MY_THEN eq_reflection]
> Suc (Suc 0)  $\equiv$  2
```

If you need the symmetric version as a meta-equation, you can write

```
thm test[MY_THEN sym eq_reflection]
> 2 ≡ Suc (Suc 0)
```

It is also possible to combine different theorem attributes, as in:

```
thm test[my_sym, MY_THEN eq_reflection]
> 2 ≡ Suc (Suc 0)
```

However, here also a weakness of the concept of theorem attributes shows through: since theorem attributes can be arbitrary functions, they do not in general commute. If you try

```
thm test[MY_THEN eq_reflection, my_sym]
> exception THM 1 raised: RSN: no unifiers
```

you get an exception indicating that the theorem *sym* does not resolve with meta-equations.

The purpose of *rule_attribute* is to directly manipulate theorems. Another usage of theorem attributes is to add and delete theorems from stored data. For example the theorem attribute *[simp]* adds or deletes a theorem from the current simpset. For these applications, you can use *declaration_attribute*. To illustrate this function, let us introduce a reference containing a list of theorems.

```
val my_thms = ref ([] : thm list)
```

The purpose of this reference is to store a list of theorems. We are going to modify it by adding and deleting theorems. However, a word of warning: such references must not be used in any code that is meant to be more than just for testing purposes! Here it is only used to illustrate matters. We will show later how to store data properly without using references.

We need to provide two functions that add and delete theorems from this list. For this we use the two functions:

```
fun my_thm_add thm ctxt =
  (my_thms := Thm.add_thm thm (!my_thms); ctxt)

fun my_thm_del thm ctxt =
  (my_thms := Thm.del_thm thm (!my_thms); ctxt)
```

These functions take a theorem and a context and, for what we are explaining here it is sufficient that they just return the context unchanged. They change however the reference *my_thms*, whereby the function *add_thm* adds a theorem if it is not already included in the list, and *del_thm* deletes one (both functions use the predicate *eq_thm_prop*, which compares theorems according to their proved propositions modulo alpha-equivalence).

You can turn functions *my_thm_add* and *my_thm_del* into attributes with the code

```
val my_add = Thm.declaration_attribute my_thm_add
val my_del = Thm.declaration_attribute my_thm_del
```

and set up the attributes as follows

```
attribute_setup my_thms = {* Attrib.add_del my_add my_del *}
  "maintaining a list of my_thms - rough test only!"
```

The parser `add_del` is a predefined parser for adding and deleting lemmas. Now if you prove the next lemma and attach to it the attribute `[my_thms]`

```
lemma trueI_2[my_thms]: "True" by simp
```

then you can see it is added to the initially empty list.

```
!my_thms
> ["True"]
```

You can also add theorems using the command **declare**.

```
declare test[my_thms] trueI_2[my_thms add]
```

With this attribute, the `add` operation is the default and does not need to be explicitly given. These three declarations will cause the theorem list to be updated as:

```
!my_thms
> ["True", "Suc (Suc 0) = 2"]
```

The theorem `trueI_2` only appears once, since the function `add_thm` tests for duplicates, before extending the list. Deletion from the list works as follows:

```
declare test[my_thms del]
```

After this, the theorem list is again:

```
!my_thms
> ["True"]
```

We used in this example two functions declared as `declaration_attribute`, but there can be any number of them. We just have to change the parser for reading the arguments accordingly.

However, as said at the beginning of this example, using references for storing theorems is *not* the received way of doing such things. The received way is to start a “data slot”, below called `MyThmsData`, generated by the functor `GenericDataFun`:

```
structure MyThmsData = GenericDataFun
  (type T = thm list
   val empty = []
   val extend = I
   fun merge _ = Thm.merge_thms)
```

The type T of this data slot is *thm list*.⁴ To use this data slot, you only have to change *my_thm_add* and *my_thm_del* to:

```
val my_thm_add = MyThmsData.map o Thm.add_thm
val my_thm_del = MyThmsData.map o Thm.del_thm
```

where *MyThmsData.map* updates the data appropriately. The corresponding theorem attributes are

```
val my_add = Thm.declaration_attribute my_thm_add
val my_del = Thm.declaration_attribute my_thm_del
```

and the setup is as follows

```
attribute_setup my_thms2 = {* Attrib.add_del my_add my_del *}
  "properly maintaining a list of my_thms"
```

Initially, the data slot is empty

```
MyThmsData.get (Context.Proof @{context})
> []
```

but if you prove

```
lemma three[my_thms2]: "3 = Suc (Suc (Suc 0))" by simp
```

then the lemma is recorded.

```
MyThmsData.get (Context.Proof @{context})
> ["3 = Suc (Suc (Suc 0))"]
```

With theorem attribute *my_thms2* you can also nicely see why it is important to store data in a “data slot” and *not* in a reference. Backtrack to the point just before the lemma *three* was proved and check the the content of *MyThmsData*: it should be empty. The addition has been properly retracted. Now consider the proof:

```
lemma four[my_thms]: "4 = Suc (Suc (Suc (Suc 0)))" by simp
```

Checking the content of *my_thms* gives

```
!my_thms
> ["4 = Suc (Suc (Suc (Suc 0)))", "True"]
```

⁴FIXME: give a pointer to where data slots are explained properly.

as expected, but if you backtrack before the lemma *four*, the content of *my_thms* is unchanged. The backtracking mechanism of Isabelle is completely oblivious about what to do with references, but properly treats “data slots”!

Since storing theorems in a list is such a common task, there is the special functor `Named_Thms`, which does most of the work for you. To obtain a named theorem list, you just declare

```
structure FooRules = Named_Thms
  (val name = "foo"
   val description = "Rules for foo")
```

and set up the *FooRules* with the command

```
setup {* FooRules.setup *}
```

This code declares a data slot where the theorems are stored, an attribute *foo* (with the *add* and *del* options for adding and deleting theorems) and an internal ML interface to retrieve and modify the theorems.

Furthermore, the facts are made available on the user-level under the dynamic fact name *foo*. For example you can declare three lemmas to be of the kind *foo* by:

```
lemma rule1[foo]: "A" sorry
lemma rule2[foo]: "B" sorry
lemma rule3[foo]: "C" sorry
```

and undeclare the first one by:

```
declare rule1[foo del]
```

and query the remaining ones with:

```
thm foo
> ?C
> ?B
```

On the ML-level the rules marked with *foo* can be retrieved using the function `FooRules.get`:

```
FooRules.get @{context}
> ["?C", "?B"]
```

Read More

For more information see [Pure/Tools/named_thms.ML](#).

(FIXME What are: *theory_attributes*, *proof_attributes*?)

Read More

FIXME: Pure/more_thm.ML; parsers for attributes is in Pure/Isar/attrib.ML...also explained in the chapter about parsing.

3.7 Theories, Contexts and Local Theories (TBD)

There are theories, proof contexts and local theories (in this order, if you want to order them).

In contrast to an ordinary theory, which simply consists of a type signature, as well as tables for constants, axioms and theorems, a local theory contains additional context information, such as locally fixed variables and local assumptions that may be used by the package. The type *local_theory* is identical to the type of *proof contexts* *Proof.context*, although not every proof context constitutes a valid local theory.

3.8 Storing Theorems (TBD)

add_thms_dynamic

```
local_setup {*
  LocalTheory.note Thm.theoremK
    ((@{binding "allI_alt"}, []), [@{thm allI}]) #> snd *}

```

3.9 Pretty-Printing

So far we printed out only plain strings without any formatting except for occasional explicit line breaks using `"\n"`. This is sufficient for “quick-and-dirty” printouts. For something more sophisticated, Isabelle includes an infrastructure for properly formatting text. This infrastructure is loosely based on a paper by Oppen [5]. Most of its functions do not operate on *strings*, but on instances of the type:

Pretty.T

The function *str* transforms a (plain) string into such a pretty type. For example

```
Pretty.str "test"
> String ("test", 4)

```

where the result indicates that we transformed a string with length 4. Once you have a pretty type, you can, for example, control where linebreaks may occur in case

the text wraps over a line, or with how much indentation a text should be printed. However, if you want to actually output the formatted text, you have to transform the pretty type back into a *string*. This can be done with the function *string_of*. In what follows we will use the following wrapper function for printing a pretty type:

```
fun pprint prt = tracing (Pretty.string_of prt)
```

The point of the pretty-printing infrastructure is to give hints about how to layout text and let Isabelle do the actual layout. Let us first explain how you can insert places where a line break can occur. For this assume the following function that replicates a string *n* times:

```
fun rep n str = implode (replicate n str)
```

and suppose we want to print out the string:

```
val test_str = rep 8 "foooooooooooooooooobaaaaaaaaaaaar "
```

We deliberately chose a large string so that it spans over more than one line. If we print out the string using the usual “quick-and-dirty” method, then we obtain the ugly output:

```
tracing test_str
> foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar fooooooooooooo
> oooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaa
> aaaaaaar foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar fo
> oooooooooooooooooobaaaaaaaaaaaar
```

We obtain the same if we just use

```
pprint (Pretty.str test_str)
> foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar fooooooooooooo
> oooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaa
> aaaaaaar foooooooooooooooooobaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaar fo
> oooooooooooooooooobaaaaaaaaaaaar
```

However by using pretty types you have the ability to indicate a possible line break for example at each space. You can achieve this with the function *breaks*, which expects a list of pretty types and inserts a possible line break in between every two elements in this list. To print this list of pretty types as a single string, we concatenate them with the function *blk* as follows:

```
let
  val ptrs = map Pretty.str (space_explode " " test_str)
in
```

```

pprint (Pretty.blk (0, Pretty.breaks ptrs))
end
> foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
> foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
> foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
> foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar

```

Here the layout of `test_str` is much more pleasing to the eye. The `0` in `blk` stands for no indentation of the printed string. You can increase the indentation and obtain

```

let
  val ptrs = map Pretty.str (space_explode " " test_str)
in
  pprint (Pretty.blk (3, Pretty.breaks ptrs))
end
> foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>   foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>   foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>   foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar

```

where starting from the second line the indent is 3. If you want that every line starts with the same indent, you can use the function `indent` as follows:

```

let
  val ptrs = map Pretty.str (space_explode " " test_str)
in
  pprint (Pretty.indent 10 (Pretty.blk (0, Pretty.breaks ptrs)))
end
>           foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>           foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>           foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar
>           foooooooooooooooooobaaaaaaaaaaaaar foooooooooooooooooobaaaaaaaaaaaaar

```

If you want to print out a list of items separated by commas and have the linebreaks handled properly, you can use the function `commas`. For example

```

let
  val ptrs = map (Pretty.str o string_of_int) (99998 upto 100020)
in
  pprint (Pretty.blk (0, Pretty.commas ptrs))
end
> 99998, 99999, 100000, 100001, 100002, 100003, 100004, 100005, 100006,
> 100007, 100008, 100009, 100010, 100011, 100012, 100013, 100014, 100015,
> 100016, 100017, 100018, 100019, 100020

```

where `upto` generates a list of integers. You can print out this list as a “set”, that means enclosed inside `"{"` and `"}"`, and separated by commas using the function `enum`. For example

```

let
  val ptrs = map (Pretty.str o string_of_int) (99998 upto 100020)
in
  pprint (Pretty.enum ", " "{" "}" ptrs)
end
> {99998, 99999, 100000, 100001, 100002, 100003, 100004, 100005, 100006,
>   100007, 100008, 100009, 100010, 100011, 100012, 100013, 100014, 100015,
>   100016, 100017, 100018, 100019, 100020}

```

As can be seen, this function prints out the “set” so that starting from the second, each new line as an indentation of 2.

If you print out something that goes beyond the capabilities of the standard functions, you can do relatively easily the formatting yourself. Assume you want to print out a list of items where like in “English” the last two items are separated by “and”. For this you can write the function

```

1 fun and_list [] = []
2   | and_list [x] = [x]
3   | and_list xs =
4     let
5       val (front, last) = split_last xs
6     in
7       (Pretty.commas front) @
8       [Pretty.brk 1, Pretty.str "and", Pretty.brk 1, last]
9     end

```

where Line 7 prints the beginning of the list and Line 8 the last item. We have to use `Pretty.brk 1` in order to insert a space (of length 1) before the “and”. This space is also a place where a line break can occur. We do the same after the “and”. This gives you for example

```

let
  val ptrs = map (Pretty.str o string_of_int) (1 upto 22)
in
  pprint (Pretty.blk (0, and_list ptrs))
end
> 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21
> and 22

```

Next we like to pretty-print a term and its type. For this we use the function `tell_type`.

```

1 fun tell_type ctxt t =
2   let
3     fun pstr s = Pretty.breaks (map Pretty.str (space_explode " " s))
4     val ptrm = Pretty.quote (Syntax.pretty_term ctxt t)
5     val pty  = Pretty.quote (Syntax.pretty_typ  ctxt (fastype_of t))
6   in

```

```

7   pprint (Pretty.blk (0,
8     (pstr "The term " @ [ptrm] @ pstr " has type "
9     @ [pty, Pretty.str "."]))))
10  end

```

In Line 3 we define a function that inserts possible linebreaks in places where a space is. In Lines 4 and 5 we pretty-print the term and its type using the functions `pretty_term` and `pretty_ttyp`. We also use the function `quote` in order to enclose the term and type inside quotation marks. In Line 9 we add a period right after the type without the possibility of a line break, because we do not want that a line break occurs there.

Now you can write

```

tell_type @{context} @{term "min (Suc 0)"}
> The term "min (Suc 0)" has type "nat ⇒ nat".

```

To see the proper line breaking, you can try out the function on a bigger term and type. For example:

```

tell_type @{context} @{term "op = (op = (op = (op = op =)))"}
> The term "op = (op = (op = (op = op =)))" has type
> "((((('a ⇒ 'a ⇒ bool) ⇒ bool) ⇒ bool) ⇒ bool) ⇒ bool) ⇒ bool".

```

FIXME: TBD below

```

pprint (Pretty.big_list "header" (map (Pretty.str o string_of_int) (4
upto 10)))

```

`chunks` inserts forced breaks (unlike `blk` where you have to do this yourself)

```

(Pretty.chunks [Pretty.str "a", Pretty.str "b"],
  Pretty.blk (0, [Pretty.str "a", Pretty.str "b"]))

fun setmp_show_all_types f =
  setmp show_all_types
    (! show_types orelse ! show_sorts orelse ! show_all_types) f;

val ctxt = @{context};
val t1 = @{term "undefined::nat"};
val t2 = @{term "Suc y"};
val pty = Pretty.block (Pretty.breaks
  [(setmp show_question_marks false o setmp_show_all_types)
    (Syntax.pretty_term ctxt) t1,
    Pretty.str "=", Syntax.pretty_term ctxt t2]);
pty |> Pretty.string_of |> priority

```

the infrastructure of Syntax-pretty-term makes sure it is printed nicely

```
Pretty.mark Markup.hilite (Pretty.str "foo") |> Pretty.string_of |>
tracing
(Pretty.str "bar") |> Pretty.string_of |> tracing

Pretty.mark Markup.subgoal (Pretty.str "foo") |> Pretty.string_of |>
tracing
(Pretty.str "bar") |> Pretty.string_of |> tracing
```

Still to be done:

What happens with big formulae?

Read More

The general infrastructure for pretty-printing is implemented in the file [Pure/General/pretty.ML](#). The file [Pure/Syntax/syntax.ML](#) contains pretty-printing functions for terms, types, theorems and so on.

[Pure/General/markup.ML](#)

printing into the goal buffer as part of the proof state

```
Pretty.mark Markup.hilite (Pretty.str "foo") |> Pretty.string_of |>
tracing
(Pretty.str "bar") |> Pretty.string_of |> tracing
```

writing into the goal buffer

```
fun my_hook interactive state =
  (interactive ? Proof.goal_message (fn () => Pretty.str
"foo")) state
```

```
setup {* Context.theory_map (Specification.add_theorem_hook my_hook) *}
```

```
lemma "False"
oops
```

3.10 Misc (TBD)

```
Datatype.get_info @{theory} "List.list"
```

FIXME: association lists: [Pure/General/alist.ML](#)

FIXME: calling the ML-compiler

3.11 Name Space Issues (TBD)

```
let
  val do_nothing = Scan.succeed (LocalTheory.theory I)
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar" "description of foobar" kind do_nothing
end

let
  fun trace_prop str =
    LocalTheory.theory (fn lthy => (tracing str; lthy))
  val trace_prop_parser = OuterParse.prop >> trace_prop
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar_trace" "traces a proposition"
    kind trace_prop_parser
end

let
  fun prove_prop str lthy =
    let
      val prop = Syntax.read_prop lthy str
    in
      Proof.theorem_i NONE (K I) [(prop, [])] lthy
    end;
  val prove_prop_parser = OuterParse.prop >> prove_prop
  val kind = OuterKeyword.thy_goal
in
  OuterSyntax.local_theory_to_proof "foobar_goal" "proving a proposition"
    kind prove_prop_parser
end

val r = ref (NONE:(unit -> term) option)
```



```
let
  fun after_qed thm_name thms lthy =
    LocalTheory.note Thm.theoremK (thm_name, (flat thms)) lthy |> snd

  fun setup_proof (thm_name, (txt, pos)) lthy =
    let
      val trm = ML_Context.evaluate lthy true ("r", r) txt
    in
      Proof.theorem_i NONE (after_qed thm_name) [[(trm, [])]] lthy
    end

  val parser = SpecParse.opt_thm_name ":" -- OuterParse.ML_source

in
  OuterSyntax.local_theory_to_proof "foobar_prove" "proving a proposition"
  OuterKeyword.thy_goal (parser >> setup_proof)
end
```


Chapter 4

Parsing

Isabelle distinguishes between *outer* and *inner* syntax. Commands, such as **definition**, **inductive** and so on, belong to the outer syntax, whereas terms, types and so on belong to the inner syntax. For parsing inner syntax, Isabelle uses a rather general and sophisticated algorithm, which is driven by priority grammars. Parsers for outer syntax are built up by functional parsing combinators. These combinators are a well-established technique for parsing, which has, for example, been described in Paulson's classic ML-book [6]. Isabelle developers are usually concerned with writing these outer syntax parsers, either for new definitional packages or for calling methods with specific arguments.

Read More

The library for writing parser combinators is split up, roughly, into two parts: The first part consists of a collection of generic parser combinators defined in the structure `Scan` in the file `Pure/General/scan.ML`. The second part of the library consists of combinators for dealing with specific token types, which are defined in the structure `OuterParse` in the file `Pure/Isar/outer_parse.ML`. In addition specific parsers for packages are defined in `Pure/Isar/spec_parse.ML`. Parsers for method arguments are defined in `Pure/Isar/args.ML`.

4.1 Building Generic Parsers

Let us first have a look at parsing strings using generic parsing combinators. The function `$$` takes a string as argument and will “consume” this string from a given input list of strings. “Consume” in this context means that it will return a pair consisting of this string and the rest of the input list. For example:

```
($$ "h") (Symbol.explode "hello")  
> ("h", ["e", "l", "l", "o"])
```

```
($$ "w") (Symbol.explode "world")  
> ("w", ["o", "r", "l", "d"])
```

The function `$$` will either succeed (as in the two examples above) or raise the exception `FAIL` if no string can be consumed. For example trying to parse

```
($$ "x") (Symbol.explode "world")
> Exception FAIL raised
```

will raise the exception *FAIL*. There are three exceptions used in the parsing combinators:

- *FAIL* is used to indicate that alternative routes of parsing might be explored.
- *MORE* indicates that there is not enough input for the parser. For example in `($$ "h") []`.
- *ABORT* is the exception that is raised when a dead end is reached. It is used for example in the function `!!` (see below).

However, note that these exceptions are private to the parser and cannot be accessed by the programmer (for example to handle them).

In the examples above we use the function `Symbol.explode`, instead of the more standard library function `explode`, for obtaining an input list for the parser. The reason is that `Symbol.explode` is aware of character sequences, for example `\<foo>`, that have a special meaning in Isabelle. To see the difference consider

```
let
  val input = "\<foo> bar"
in
  (explode input, Symbol.explode input)
end
> ([ "\"", "<", "f", "o", "o", ">", " ", "b", "a", "r"],
> ["\<foo>", " ", "b", "a", "r"])
```

Slightly more general than the parser `$$` is the function `one`, in that it takes a predicate as argument and then parses exactly one item from the input list satisfying this predicate. For example the following parser either consumes an `"h"` or a `"w"`:

```
let
  val hw = Scan.one (fn x => x = "h" orelse x = "w")
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
  (hw input1, hw input2)
end
> (("h", ["e", "l", "l", "o"]), ("w", ["o", "r", "l", "d"]))
```

Two parsers can be connected in sequence by using the function `--`. For example parsing `h`, `e` and `l` (in this order) you can achieve by:

```
($$ "h" -- $$ "e" -- $$ "l") (Symbol.explode "hello")
> (("h", "e"), "l"), ["l", "o"])
```

Note how the result of consumed strings builds up on the left as nested pairs.

If, as in the previous example, you want to parse a particular string, then you can use the function *this_string*.

```
Scan.this_string "hell" (Symbol.explode "hello")
> ("hell", ["o"])
```

Parsers that explore alternatives can be constructed using the function *//*. The parser (*p // q*) returns the result of *p*, in case it succeeds, otherwise it returns the result of *q*. For example:

```
let
  val hw = $$ "h" || $$ "w"
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
  (hw input1, hw input2)
end
> (("h", ["e", "l", "l", "o"]), ("w", ["o", "r", "l", "d"]))
```

The functions */--* and *--|* work like the sequencing function for parsers, except that they discard the item being parsed by the first (respectively second) parser. For example:

```
let
  val just_e = $$ "h" /-- $$ "e"
  val just_h = $$ "h" --| $$ "e"
  val input = Symbol.explode "hello"
in
  (just_e input, just_h input)
end
> (("e", ["l", "l", "o"]), ("h", ["l", "l", "o"]))
```

The parser *Scan.optional p x* returns the result of the parser *p*, if it succeeds; otherwise it returns the default value *x*. For example:

```
let
  val p = Scan.optional ($$ "h") "x"
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
  (p input1, p input2)
end
> (("h", ["e", "l", "l", "o"]), ("x", ["w", "o", "r", "l", "d"]))
```

The function `option` works similarly, except no default value can be given. Instead, the result is wrapped as an `option`-type. For example:

```
let
  val p = Scan.option ($$ "h")
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
  (p input1, p input2)
end
> ((SOME "h", ["e", "l", "l", "o"]), (NONE, ["w", "o", "r", "l", "d"]))
```

The function `ahead` parses some input, but leaves the original input unchanged. For example:

```
Scan.ahead (Scan.this_string "foo") (Symbol.explode "foo")
> ("foo", ["f", "o", "o"])
```

The function `!!` helps with producing appropriate error messages during parsing. For example if you want to parse `p` immediately followed by `q`, or start a completely different parser `r`, you might write:

```
(p -- q) || r
```

However, this parser is problematic for producing a useful error message, if the parsing of `(p -- q)` fails. Because with the parser above you lose the information that `p` should be followed by `q`. To see this assume that `p` is present in the input, but it is not followed by `q`. That means `(p -- q)` will fail and hence the alternative parser `r` will be tried. However, in many circumstances this will be the wrong parser for the input “`p`-followed-by-something” and therefore will also fail. The error message is then caused by the failure of `r`, not by the absence of `q` in the input. This kind of situation can be avoided when using the function `!!`. This function aborts the whole process of parsing in case of a failure and prints an error message. For example if you invoke the parser

```
!! (fn _ => "foo") ($$ "h")
```

on `"hello"`, the parsing succeeds

```
(!! (fn _ => "foo") ($$ "h")) (Symbol.explode "hello")
> ("h", ["e", "l", "l", "o"])
```

but if you invoke it on `"world"`

```
(!! (fn _ => "foo") ($$ "h")) (Symbol.explode "world")
> Exception ABORT raised
```

then the parsing aborts and the error message *foo* is printed. In order to see the error message properly, you need to prefix the parser with the function *error*. For example:

```
Scan.error (!! (fn _ => "foo") ($$ "h"))
> Exception Error "foo" raised
```

This “prefixing” is usually done by wrappers such as *local_theory* (see Section 4.8 which explains this function in more detail).

Let us now return to our example of parsing $(p \text{ -- } q) \text{ || } r$. If you want to generate the correct error message for failure of parsing *p*-followed-by-*q*, then you have to write:

```
fun p_followed_by_q p q r =
let
  val err_msg = fn _ => p ^ " is not followed by " ^ q
in
  ($$ p -- (!! err_msg ($$ q))) || ($$ r -- $$ r)
end
```

Running this parser with the arguments "h", "e" and "w", and the input "holle"

```
Scan.error (p_followed_by_q "h" "e" "w") (Symbol.explode "holle")
> Exception ERROR "h is not followed by e" raised
```

produces the correct error message. Running it with

```
Scan.error (p_followed_by_q "h" "e" "w") (Symbol.explode "wworld")
> (("w", "w"), ["o", "r", "l", "d"])
```

yields the expected parsing.

The function *Scan.repeat p* will apply a parser *p* as often as it succeeds. For example:

```
Scan.repeat ($$ "h") (Symbol.explode "hhhhello")
> (["h", "h", "h", "h"], ["e", "l", "l", "o"])
```

Note that *repeat* stores the parsed items in a list. The function *repeat1* is similar, but requires that the parser *p* succeeds at least once.

Also note that the parser would have aborted with the exception *MORE*, if you had it run with the string "hhhh". This can be avoided by using the wrapper *finite* and the “stopper-token” *stopper*. With them you can write:

```
Scan.finite Symbol.stopper (Scan.repeat ($$ "h")) (Symbol.explode "hhhh")
> (["h", "h", "h", "h"], [])
```

The function *stopper* is the “end-of-input” indicator for parsing strings; other stoppers need to be used when parsing, for example, tokens. However, this kind of manually wrapping is often already done by the surrounding infrastructure.

The function *repeat* can be used with *one* to read any string as in

```
let
  val p = Scan.repeat (Scan.one Symbol.not_eof)
  val input = Symbol.explode "foo bar foo"
in
  Scan.finite Symbol.stopper p input
end
> (["f", "o", "o", " ", "b", "a", "r", " ", "f", "o", "o"], [])
```

where the function *not_eof* ensures that we do not read beyond the end of the input string (i.e. stopper symbol).

The function *unless* takes two parsers: if the first one can parse the input, then the whole parser fails; if not, then the second is tried. Therefore

```
Scan.unless ($$ "h") ($$ "w") (Symbol.explode "hello")
> Exception FAIL raised
```

fails, while

```
Scan.unless ($$ "h") ($$ "w") (Symbol.explode "world")
> ("w", ["o", "r", "l", "d"])
```

succeeds.

The functions *repeat* and *unless* can be combined to read any input until a certain marker symbol is reached. In the example below the marker symbol is a ***.

```
let
  val p = Scan.repeat (Scan.unless ($$ "*") (Scan.one Symbol.not_eof))
  val input1 = Symbol.explode "fooooo"
  val input2 = Symbol.explode "foo*ooo"
in
  (Scan.finite Symbol.stopper p input1,
   Scan.finite Symbol.stopper p input2)
end
> ((["f", "o", "o", "o", "o", "o"], []),
>  (["f", "o", "o"], ["*", "o", "o", "o"]))
```

After parsing is done, you almost always want to apply a function to the parsed items. One way to do this is the function *>>* where $(p \gg f)$ runs first the parser *p* and upon successful completion applies the function *f* to the result. For example


```
let
  fun double (x, y) = (x ^ x, y ^ y)
  val parser = $$ "h" -- $$ "e"
in
  (parser >> double) (Symbol.explode "hello")
end
> (("hh", "ee"), ["l", "l", "o"])
```

doubles the two parsed input strings; or

```
let
  val p = Scan.repeat (Scan.one Symbol.not_eof)
  val input = Symbol.explode "foo bar foo"
in
  Scan.finite Symbol.stopper (p >> implode) input
end
> ("foo bar foo", [])
```

where the single-character strings in the parsed output are transformed back into one string.

The function *lift* takes a parser and a pair as arguments. This function applies the given parser to the second component of the pair and leaves the first component untouched. For example

```
Scan.lift ($$ "h" -- $$ "e") (1, Symbol.explode "hello")
> (("h", "e"), (1, ["l", "l", "o"]))
```

(FIXME: In which situations is this useful? Give examples.)

Exercise 4.1.1. Write a parser that parses an input string so that any comment enclosed within *(...*)* is replaced by the same comment but enclosed within *(**...**)* in the output string. To enclose a string, you can use the function *enclose s1 s2 s* which produces the string $s1 \wedge s \wedge s2$. Hint: To simplify the task ignore the proper nesting of comments.

4.2 Parsing Theory Syntax

Most of the time, however, Isabelle developers have to deal with parsing tokens, not strings. These token parsers have the type:

```
type 'a parser = OuterLex.token list -> 'a * OuterLex.token list
```

The reason for using token parsers is that theory syntax, as well as the parsers for the arguments of proof methods, use the type *OuterLex.token*.

Read More

The parser functions for the theory syntax are contained in the structure `OuterParse` defined in the file `Pure/Isar/outer_parse.ML`. The definition for tokens is in the file `Pure/Isar/outer_lex.ML`.

The structure `OuterLex` defines several kinds of tokens (for example `Ident` for identifiers, `Keyword` for keywords and `Command` for commands). Some token parsers take into account the kind of tokens. The first example shows how to generate a token list out of a string using the function `scan`. It is given the argument `Position.none` since, at the moment, we are not interested in generating precise error messages. The following code¹

```
OuterSyntax.scan Position.none "hello world"
> [Token (...,(Ident, "hello"),...),
>  Token (...,(Space, " "),...),
>  Token (...,(Ident, "world"),...)]
```

produces three tokens where the first and the last are identifiers, since `"hello"` and `"world"` do not match any other syntactic category. The second indicates a space.

We can easily change what is recognised as a keyword with the function `keyword`. For example calling it with

```
val _ = OuterKeyword.keyword "hello"
```

then lexing `"hello world"` will produce

```
OuterSyntax.scan Position.none "hello world"
> [Token (...,(Keyword, "hello"),...),
>  Token (...,(Space, " "),...),
>  Token (...,(Ident, "world"),...)]
```

Many parsing functions later on will require white space, comments and the like to have already been filtered out. So from now on we are going to use the functions `filter` and `is_proper` to do this. For example:

```
let
  val input = OuterSyntax.scan Position.none "hello world"
in
  filter OuterLex.is_proper input
end
> [Token (...,(Ident, "hello"), ...), Token (...,(Ident, "world"), ...)]
```

For convenience we define the function:

¹Note that because of a possible bug in the PolyML runtime system, the result is printed as `"?"`, instead of the tokens.

```
fun filtered_input str =
  filter OuterLex.is_proper (OuterSyntax.scan Position.none str)
```

If you now parse

```
filtered_input "inductive | for"
> [Token (...,(Command, "inductive"),...),
>  Token (...,(Keyword, "|"),...),
>  Token (...,(Keyword, "for"),...)]
```

you obtain a list consisting of only one command and two keyword tokens. If you want to see which keywords and commands are currently known to Isabelle, type:

```
let
  val (keywords, commands) = OuterKeyword.get_lexicons ()
in
  (Scan.dest_lexicon commands, Scan.dest_lexicon keywords)
end
> (["}", "{", ...], [" $\Rightarrow$ ", " $\Leftarrow$ ", ...])
```

You might have to adjust the `print_depth` in order to see the complete list. The parser `$$$` parses a single keyword. For example:

```
let
  val input1 = filtered_input "where for"
  val input2 = filtered_input "| in"
in
  (OuterParse.$$$ "where" input1, OuterParse.$$$ "|" input2)
end
> (("where",...), ("|",...))
```

Any non-keyword string can be parsed with the function `reserved`. For example:

```
let
  val p = OuterParse.reserved "bar"
  val input = filtered_input "bar"
in
  p input
end
> ("bar", [])
```

Like before, you can sequentially connect parsers with `--`. For example:

```
let
  val input = filtered_input "| in"
in
  (OuterParse.$$$ "|" -- OuterParse.$$$ "in") input
end
> ("|", "in"), [])
```

The parser `OuterParse.enum s p` parses a possibly empty list of items recognised by the parser `p`, where the items being parsed are separated by the string `s`. For example:

```
let
  val input = filtered_input "in | in | in foo"
in
  (OuterParse.enum "|" (OuterParse.$$$ "in")) input
end
> (["in", "in", "in"], [...])
```

The function `enum1` works similarly, except that the parsed list must be non-empty. Note that we had to add a string `"foo"` at the end of the parsed string, otherwise the parser would have consumed all tokens and then failed with the exception `MORE`. Like in the previous section, we can avoid this exception using the wrapper `Scan.finite`. This time, however, we have to use the “stopper-token” `OuterLex.stopper`. We can write:

```
let
  val input = filtered_input "in | in | in"
  val p = OuterParse.enum "|" (OuterParse.$$$ "in")
in
  Scan.finite OuterLex.stopper p input
end
> (["in", "in", "in"], [])
```

The following function will help to run examples.

```
fun parse p input = Scan.finite OuterLex.stopper (Scan.error p) input
```

The function `!!!` can be used to force termination of the parser in case of a dead end, just like `Scan.!!` (see previous section). A difference, however, is that the error message of `OuterParse.!!!` is fixed to be `"Outer syntax error"` together with a relatively precise description of the failure. For example:

```
let
  val input = filtered_input "in |"
  val parse_bar_then_in = OuterParse.$$$ "|" -- OuterParse.$$$ "in"
in
  parse (OuterParse.!!! parse_bar_then_in) input
end
> Exception ERROR "Outer syntax error: keyword "|" expected,
> but keyword in was found" raised
```

Exercise 4.2.1. (FIXME) A type-identifier, for example `'a`, is a token of kind `Keyword`. It can be parsed using the function `type_ident`.

(FIXME: or give parser for numbers)

Whenever there is a possibility that the processing of user input can fail, it is a good idea to give all available information about where the error occurred. For this Isabelle can attach positional information to tokens and then thread this information up the “processing chain”. To see this, modify the function `filtered_input`, described earlier, as follows

```
fun filtered_input' str =
  filter OuterLex.is_proper (OuterSyntax.scan (Position.line 7) str)
```

where we pretend the parsed string starts on line 7. An example is

```
filtered_input' "foo \n bar"
> [Token ("foo", ({line=7, end_line=7}, {line=7})), (Ident, "foo"), ...],
> Token ("bar", ({line=8, end_line=8}, {line=8})), (Ident, "bar"), ...]
```

in which the “\n” causes the second token to be in line 8.

By using the parser `position` you can access the token position and return it as part of the parser result. For example

```
let
  val input = filtered_input' "where"
in
  parse (OuterParse.position (OuterParse.$$$ "where")) input
end
> ("where", {line=7, end_line=7}), []
```

Read More

The functions related to positions are implemented in the file [Pure/General/position.ML](#).

4.3 Parsers for ML-Code (TBD)

`ML_source`

4.4 Context Parser (TBD)

`Args.context`

`Args.context`

Used for example in `attribute_setup` and `method_setup`.

4.5 Argument and Attribute Parsers (TBD)

4.6 Parsing Inner Syntax

There is usually no need to write your own parser for parsing inner syntax, that is for terms and types: you can just call the predefined parsers. Terms can be parsed using the function `term`. For example:

```
let
  val input = OuterSyntax.scan Position.none "foo"
in
  OuterParse.term input
end
> ("^E^Ftoken^Efoo^E^F^E", [])
```

The function `prop` is similar, except that it gives a different error message, when parsing fails. As you can see, the parser not just returns the parsed string, but also some encoded information. You can decode the information with the function `parse` in `YXML`. For example

```
YXML.parse "^E^Ftoken^Efoo^E^F^E"
> XML.Elem ("token", [], [XML.Text "foo"])
```

The result of the decoding is an XML-tree. You can see better what is going on if you replace `Position.none` by `Position.line 42`, say:

```
let
  val input = OuterSyntax.scan (Position.line 42) "foo"
in
  YXML.parse (fst (OuterParse.term input))
end
> XML.Elem ("token", [("line", "42"), ("end_line", "42")], [XML.Text "foo"])
```

The positional information is stored as part of an XML-tree so that code called later on will be able to give more precise error messages.

Read More

The functions to do with input and output of XML and YXML are defined in [Pure/General/xml.ML](#) and [Pure/General/yxml.ML](#).

4.7 Parsing Specifications

There are a number of special purpose parsers that help with parsing specifications of function definitions, inductive predicates and so on. In Chapter 6, for example, we will need to parse specifications for inductive predicates of the form:

simple_inductive

even and odd

where

even0: "even 0"

| evenS: "odd n \implies even (Suc n)"

| oddS: "even n \implies odd (Suc n)"

For this we are going to use the parser:

```

1 val spec_parser =
2   OuterParse.fixes --
3   Scan.optional
4     (OuterParse.$$$ "where" |--
5      OuterParse.!!!
6      (OuterParse.enum1 "|"
7       (SpecParse.opt_thm_name ":" -- OuterParse.prop))) []

```

Note that the parser must not parse the keyword **simple_inductive**, even if it is meant to process definitions as shown above. The parser of the keyword will be given by the infrastructure that will eventually call *spec_parser*.

To see what the parser returns, let us parse the string corresponding to the definition of *even* and *odd*:

```

let
  val input = filtered_input
    ("even and odd " ^
     "where " ^
     "  even0[intro]: \"even 0\" " ^
     "| evenS[intro]: \"odd n  $\implies$  even (Suc n)\\" ^
     "| oddS[intro]: \"even n  $\implies$  odd (Suc n)\\"")
in
  parse spec_parser input
end
> ((([even, NONE, NoSyn], [odd, NONE, NoSyn]),
>   [(even0, ...), "\^E\^Ftoken\^Even 0\^E\^F\^E"),
>   [(evenS, ...), "\^E\^Ftoken\^Eodd n  $\implies$  even (Suc n)\^E\^F\^E"),
>   [(oddS, ...), "\^E\^Ftoken\^Eeven n  $\implies$  odd (Suc n)\^E\^F\^E"]]), [])

```

As you see, the result is a pair consisting of a list of variables with optional type-annotation and syntax-annotation, and a list of rules where every rule has optionally a name and an attribute.

The function *fixes* in Line 2 of the parser reads an **and**-separated list of variables that can include optional type annotations and syntax translations. For example:²

```

let
  val input = filtered_input

```

²Note that in the code we need to write `"int \implies bool"` in order to properly escape the double quotes in the compound type.

```

"foo::\"int ⇒ bool\" and bar::nat (\"BAR\" 100) and blonk"
in
  parse OuterParse.fixes input
end
> [(foo, SOME "\^E\^Ftoken\^Eint ⇒ bool\^E\^F\^E", NoSyn),
>  (bar, SOME "\^E\^Ftoken\^Enat\^E\^F\^E", Mixfix ("BAR", [], 100)),
>  (blonk, NONE, NoSyn)], []

```

Whenever types are given, they are stored in the *SOME*s. The types are not yet used to type the variables: this must be done by type-inference later on. Since types are part of the inner syntax they are strings with some encoded information (see previous section). If a mixfix-syntax is present for a variable, then it is stored in the *Mixfix* data structure; no syntax translation is indicated by *NoSyn*.

Read More

The data structure for mixfix annotations is defined in [Pure/Syntax/mixfix.ML](#).

Lines 3 to 7 in the function *spec_parser* implement the parser for a list of introduction rules, that is propositions with theorem annotations such as rule names and attributes. The introduction rules are propositions parsed by *prop*. However, they can include an optional theorem name plus some attributes. For example

```

let
  val input = filtered_input "foo_lemma[intro,dest!]:"
  val ((name, attrib), _) = parse (SpecParse.thm_name ":") input
in
  (name, map Args.dest_src attrib)
end
> (foo_lemma, [(("intro", []), ...), (("dest", [...]), ...)])

```

The function *opt_thm_name* is the “optional” variant of *thm_name*. Theorem names can contain attributes. The name has to end with “:”—see the argument of the function *SpecParse.opt_thm_name* in Line 7.

Read More

Attributes and arguments are implemented in the files [Pure/Isar/attrib.ML](#) and [Pure/Isar/args.ML](#).

Exercise 4.7.1. Have a look at how the parser *SpecParse.where_alt_specs* is implemented in file [Pure/Isar/spec_parse.ML](#). This parser corresponds to the “where-part” of the introduction rules given above. Below we paraphrase the code of *where_alt_specs* adapted to our purposes.

```

1 val spec_parser' =
2   OuterParse.fixes --
3   Scan.optional
4   (OuterParse.$$$ "where" |--
5   OuterParse.!!!

```



```

6      (OuterParse.enum1 "|"
7        ((SpecParse.opt_thm_name ":" -- OuterParse.prop) --|
8          Scan.option (Scan.ahead (OuterParse.name ||
9            OuterParse.$$$ "[" --
10           OuterParse.!!! (OuterParse.$$$ "|"))))) []

```

Both parsers accept the same input, but if you look closely, you can notice an additional “tail” (Lines 8 to 10) in `spec_parser`. What is the purpose of this additional “tail”?

(FIXME: `OuterParse.type_args`, `OuterParse.typ`, `OuterParse.opt_mixfix`)

4.8 New Commands and Keyword Files

Often new commands, for example for providing new definitional principles, need to be implemented. While this is not difficult on the ML-level, new commands, in order to be useful, need to be recognised by ProofGeneral. This results in some subtle configuration issues, which we will explain in this section.

To keep things simple, let us start with a “silly” command that does nothing at all. We shall name this command **foobar**. On the ML-level it can be defined as:

```

let
  val do_nothing = Scan.succeed (LocalTheory.theory I)
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar" "description of foobar" kind do_nothing
end

```

The crucial function `local_theory` expects a name for the command, a short description, a kind indicator (which we will explain later more thoroughly) and a parser producing a local theory transition (its purpose will also be explained later).

While this is everything you have to do on the ML-level, you need a keyword file that can be loaded by ProofGeneral. This is to enable ProofGeneral to recognise **foobar** as a command. Such a keyword file can be generated with the command-line:

```
$ isabelle keywords -k foobar some_log_files
```

The option `-k foobar` indicates which postfix the name of the keyword file will be assigned. In the case above the file will be named `isar-keywords-foobar.el`. This command requires log files to be present (in order to extract the keywords from them). To generate these log files, you first need to package the code above into a separate theory file named `Command.thy`, say—see Figure 4.1 for the complete code.

For our purposes it is sufficient to use the log files of the theories *Pure*, *HOL* and *Pure-ProofGeneral*, as well as the log file for the theory `Command.thy`, which contains the new **foobar**-command. If you target other logics besides HOL, such as Nominal or ZF, then you need to adapt the log files appropriately.

Pure and *HOL* are usually compiled during the installation of Isabelle. So log files for them should be already available. If not, then they can be conveniently compiled with the help of the build-script from the Isabelle distribution.

```

theory Command
imports Main
begin
ML {*
  let
    val do_nothing = Scan.succeed (LocalTheory.theory I)
    val kind = OuterKeyword.thy_decl
  in
    OuterSyntax.local_theory "foobar" "description of foobar" kind do_nothing
  end
*}
end

```

Figure 4.1: This file can be used to generate a log file. This log file in turn can be used to generate a keyword file containing the command **foobar**.

```

$ ./build -m "Pure"
$ ./build -m "HOL"

```

The *Pure-ProofGeneral* theory needs to be compiled with:

```
$ ./build -m "Pure-ProofGeneral" "Pure"
```

For the theory *Command.thy*, you first need to create a “managed” subdirectory with:

```
$ isabelle mkdir FoobarCommand
```

This generates a directory containing the files:

```

./IsaMakefile
./FoobarCommand/ROOT.ML
./FoobarCommand/document
./FoobarCommand/document/root.tex

```

You need to copy the file *Command.thy* into the directory *FoobarCommand* and add the line

```
no_document use_thy "Command";
```

to the file *./FoobarCommand/ROOT.ML*. You can now compile the theory by just typing:

```
$ isabelle make
```

If the compilation succeeds, you have finally created all the necessary log files. They are stored in the directory

```
~/isabelle/heaps/Isabelle2009/polym1-5.2.1_x86-linux/log
```

or something similar depending on your Isabelle distribution and architecture. One quick way to assign a shell variable to this directory is by typing

```
$ ISABELLE_LOGS="$(isabelle getenv -b ISABELLE_OUTPUT)/log
```

on the Unix prompt. If you now type `ls $ISABELLE_LOGS`, then the directory should include the files:

```
Pure.gz
HOL.gz
Pure-ProofGeneral.gz
HOL-FoobarCommand.gz
```

From them you can create the keyword files. Assuming the name of the directory is in `$ISABELLE_LOGS`, then the Unix command for creating the keyword file is:

```
$ isabelle keywords -k foobar
  $ISABELLE_LOGS/{Pure.gz,HOL.gz,Pure-ProofGeneral.gz,HOL-FoobarCommand.gz}
```

The result is the file `isar-keywords-foobar.el`. It should contain the string `foobar` twice.³ This keyword file needs to be copied into the directory `~/isabelle/etc`. To make ProofGeneral aware of it, you have to start Isabelle with the option `-k foobar`, that is:

```
$ isabelle emacs -k foobar a_theory_file
```

If you now build a theory on top of `Command.thy`, then you can use the command **foobar**. You can just write

```
foobar
```

but you will not see any action as we chose to implement this command to do nothing. The point of this command is to only show the procedure of how to interact with ProofGeneral. A similar procedure has to be done with any other new command, and also any new keyword that is introduced with `OuterKeyword.keyword`.

```
val _ = OuterKeyword.keyword "blink"
```

At the moment the command **foobar** is not very useful. Let us refine it a bit next by letting it take a proposition as argument and printing this proposition inside the tracing buffer.

The crucial part of a command is the function that determines the behaviour of the command. In the code above we used a “do-nothing”-function, which because of `succeed` does not parse any argument, but immediately returns the simple function `LocalTheory.theory I`. We can replace this code by a function that first parses a

³To see whether things are fine, check that `grep foobar` on this file returns something non-empty.

proposition (using the parser `OuterParse.prop`), then prints out the tracing information (using a new function `trace_prop`) and finally does nothing. For this you can write:

```
let
  fun trace_prop str =
    LocalTheory.theory (fn lthy => (tracing str; lthy))

  val trace_prop_parser = OuterParse.prop >> trace_prop
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar_trace" "traces a proposition"
    kind trace_prop_parser
end
```

The command is now **foobar_trace** and can be used to see the proposition in the tracing buffer.

```
foobar_trace "True  $\wedge$  False"
```

Note that so far we used `thy_decl` as the kind indicator for the command. This means that the command finishes as soon as the arguments are processed. Examples of this kind of commands are **definition** and **declare**. In other cases, commands are expected to parse some arguments, for example a proposition, and then “open up” a proof in order to prove the proposition (for example **lemma**) or prove some other properties (for example **function**). To achieve this kind of behaviour, you have to use the kind indicator `thy_goal` and the function `local_theory_to_proof` to set up the command. Note, however, once you change the “kind” of a command from `thy_decl` to `thy_goal` then the keyword file needs to be re-created!

Below we change **foobar_trace** so that it takes a proposition as argument and then starts a proof in order to prove it. Therefore in Line 13, we set the kind indicator to `thy_goal`.

```
1 let
2   fun prove_prop str lthy =
3     let
4       val prop = Syntax.read_prop lthy str
5     in
6       Proof.theorem_i NONE (K I) [[(prop, [])]] lthy
7     end;
8
9   val prove_prop_parser = OuterParse.prop >> prove_prop
10  val kind = OuterKeyword.thy_goal
11 in
12   OuterSyntax.local_theory_to_proof "foobar_goal" "proving a proposition"
13     kind prove_prop_parser
14 end
```

The function `prove_prop` in Lines 2 to 7 takes a string (the proposition to be proved) and a context as argument. The context is necessary in order to be able to use

`read_prop`, which converts a string into a proper proposition. In Line 6 the function `theorem_i` starts the proof for the proposition. Its argument `NONE` stands for a locale (which we chose to omit); the argument `(K I)` stands for a function that determines what should be done with the theorem once it is proved (we chose to just forget about it). Line 9 contains the parser for the proposition.

If you now type `foobar_goal "True \wedge True"`, you obtain the following proof state

```
foobar_goal "True  $\wedge$  True"
```

```
goal (1 subgoal):
  1. True  $\wedge$  True
```

and can prove the proposition as follows.

```
apply(rule conjI)
apply(rule TrueI)+
done
```

(FIXME: read a name and show how to store theorems; see *note*)

```
val r = ref (NONE:(unit -> term) option)
let
  fun after_qed thm_name thms lthy =
    LocalTheory.note Thm.theoremK (thm_name, (flat thms)) lthy |> snd

  fun setup_proof (thm_name, (txt, pos)) lthy =
    let
      val trm = ML_Context.evaluate lthy true ("r", r) txt
    in
      Proof.theorem_i NONE (after_qed thm_name) [[(trm, [])]] lthy
    end

  val parser = SpecParse.opt_thm_name ":" -- OuterParse.ML_source
in
  OuterSyntax.local_theory_to_proof "foobar_prove" "proving a proposition"
  OuterKeyword.thy_goal (parser >> setup_proof)
end
```

```
foobar_prove test: {* @{prop "True"} *}
apply(rule TrueI)
done
```

4.9 Methods (TBD)

(FIXME: maybe move to after the tactic section)

Methods are central to Isabelle. They are the ones you use for example in `apply`. To print out all currently known methods you can use the Isabelle command:

```
print_methods
> methods:
```

```
> -: do nothing (insert current facts only)
> HOL.default: apply some intro/elim rule (potentially classical)
> ...
```

An example of a very simple method is:

```
method_setup foo =
  {* Scan.succeed
    (K (SIMPLE_METHOD ((etac @{thm conjE} THEN' rtac @{thm conjI}) 1))) *}
  "foo method for conjE and conjI"
```

It defines the method *foo*, which takes no arguments (therefore the parser *Scan.succeed*) and only applies a single tactic, namely the tactic which applies *conjE* and then *conjI*. The function *SIMPLE_METHOD* turns such a tactic into a method. The method *foo* can be used as follows

```
lemma shows "A ∧ B ⇒ C ∧ D"
apply(foo)
```

where it results in the goal state

goal (2 subgoals):

1. $\llbracket A; B \rrbracket \Rightarrow C$
2. $\llbracket A; B \rrbracket \Rightarrow D$

(FIXME: explain a version of rule-tac)

Chapter 5

Tactical Reasoning

One of the main reason for descending to the ML-level of Isabelle is to be able to implement automatic proof procedures. Such proof procedures usually lessen considerably the burden of manual reasoning, for example, when introducing new definitions. These proof procedures are centred around refining a goal state using tactics. This is similar to the **apply**-style reasoning at the user-level, where goals are modified in a sequence of proof steps until all of them are solved. However, there are also more structured operations available on the ML-level that help with the handling of variables and assumptions.

5.1 Basics of Reasoning with Tactics

To see how tactics work, let us first transcribe a simple **apply**-style proof into ML. Suppose the following proof.

```
lemma disj_swap: "P  $\vee$  Q  $\implies$  Q  $\vee$  P"
  apply(erule disjE)
  apply(rule disjI2)
  apply(assumption)
  apply(rule disjI1)
  apply(assumption)
done
```

This proof translates to the following ML-code.

```
let
  val ctxt = @{context}
  val goal = @{prop "P  $\vee$  Q  $\implies$  Q  $\vee$  P"}
in
  Goal.prove ctxt ["P", "Q"] [] goal
  (fn _ =>
    etac @{thm disjE} 1
    THEN rtac @{thm disjI2} 1
    THEN atac 1
    THEN rtac @{thm disjI1} 1
    THEN atac 1)
end
```

```
> ?P ∨ ?Q ⇒ ?Q ∨ ?P
```

To start the proof, the function `Goal.prove ctxt xs As C tac` sets up a goal state for proving the goal C (that is $P \vee Q \Rightarrow Q \vee P$ in the proof at hand) under the assumptions As (happens to be empty) with the variables xs that will be generalised once the goal is proved (in our case P and Q).¹ The `tac` is the tactic that proves the goal; it can make use of the local assumptions (there are none in this example). The tactics `etac`, `rtac` and `atac` in the code above correspond roughly to `erule`, `rule` and `assumption`, respectively. The operator `THEN` strings the tactics together.

Read More

To learn more about the function `prove` see [Impl.Man., Sec. 4.3] and the file `Pure/goal.ML`. See `Pure/tactic.ML` and `Pure/tactical.ML` for the code of basic tactics and tactic combinators; see also Chapters 3 and 4 in the old Isabelle Reference Manual, and Chapter 3 in the Isabelle/Isar Implementation Manual.

Note that in the code above we use antiquotations for referencing the theorems. Many theorems also have ML-bindings with the same name. Therefore, we could also just have written `etac disjE 1`, or in case where there is no ML-binding obtain the theorem dynamically using the function `thm`; for example `etac (thm "disjE") 1`. Both ways however are considered bad style! The reason is that the binding for `disjE` can be re-assigned by the user and thus one does not have complete control over which theorem is actually applied. This problem is nicely prevented by using antiquotations, because then the theorems are fixed statically at compile-time.

During the development of automatic proof procedures, you will often find it necessary to test a tactic on examples. This can be conveniently done with the command `apply(tactic {* ... *})`. Consider the following sequence of tactics

```
val foo_tac =
  (etac @{thm disjE} 1
   THEN rtac @{thm disjI2} 1
   THEN atac 1
   THEN rtac @{thm disjI1} 1
   THEN atac 1)
```

and the Isabelle proof:

```
lemma "P ∨ Q ⇒ Q ∨ P"
  apply(tactic {* foo_tac *})
  done
```

By using `tactic {* ... *}` you can call from the user-level of Isabelle the tactic `foo_tac` or any other function that returns a tactic.

The tactic `foo_tac` is just a sequence of simple tactics stringed together by `THEN`. As can be seen, each simple tactic in `foo_tac` has a hard-coded number that stands for the subgoal analysed by the tactic (`1` stands for the first, or top-most, subgoal). This hard-coding of goals is sometimes wanted, but usually it is not. To avoid the explicit numbering, you can write

¹FIXME: explain prove earlier


```

val foo_tac' =
  (etac @{thm disjE}
   THEN' rtac @{thm disjI2}
   THEN' atac
   THEN' rtac @{thm disjI1}
   THEN' atac)

```

where *THEN'* is used instead of *THEN*. With *foo_tac'* you can give the number for the subgoal explicitly when the tactic is called. So in the next proof you can first discharge the second subgoal, and subsequently the first.

```

lemma "P1 ∨ Q1 ⇒ Q1 ∨ P1"
  and "P2 ∨ Q2 ⇒ Q2 ∨ P2"
apply(tactic {* foo_tac' 2 *})
apply(tactic {* foo_tac' 1 *})
done

```

This kind of addressing is more difficult to achieve when the goal is hard-coded inside the tactic. For most operators that combine tactics (*THEN* is only one such operator) a “primed” version exists.

The tactics *foo_tac* and *foo_tac'* are very specific for analysing goals being only of the form $P \vee Q \implies Q \vee P$. If the goal is not of this form, then these tactics return the error message:

```

*** empty result sequence -- proof command failed
*** At command "apply".

```

This means they failed.² The reason for this error message is that tactics are functions mapping a goal state to a (lazy) sequence of successor states. Hence the type of a tactic is:

```

type tactic = thm -> thm Seq.seq

```

By convention, if a tactic fails, then it should return the empty sequence. Therefore, if you write your own tactics, they should not raise exceptions willy-nilly; only in very grave failure situations should a tactic raise the exception *THM*.

The simplest tactics are *no_tac* and *all_tac*. The first returns the empty sequence and is defined as

```

fun no_tac thm = Seq.empty

```

which means *no_tac* always fails. The second returns the given theorem wrapped in a single member sequence; it is defined as

²To be precise, tactics do not produce this error message, it originates from the **apply** wrapper.

```
fun all_tac thm = Seq.single thm
```

which means `all_tac` always succeeds, but also does not make any progress with the proof.

The lazy list of possible successor goal states shows through at the user-level of Isabelle when using the command `back`. For instance in the following proof there are two possibilities for how to apply `foo_tac'`: either using the first assumption or the second.

```
lemma "[P ∨ Q; P ∨ Q] ⇒ Q ∨ P"
```

```
apply(tactic {* foo_tac' 1 *})
```

```
back
```

```
done
```

By using `back`, we construct the proof that uses the second assumption. While in the proof above, it does not really matter which assumption is used, in more interesting cases provability might depend on exploring different possibilities.

Read More

See [Pure/General/seq.ML](#) for the implementation of lazy sequences. In day-to-day Isabelle programming, however, one rarely constructs sequences explicitly, but uses the pre-defined tactics and tactic combinators instead.

It might be surprising that tactics, which transform one goal state to the next, are functions from theorems to theorem (sequences). The surprise resolves by knowing that every goal state is indeed a theorem. To shed more light on this, let us modify the code of `all_tac` to obtain the following tactic

```
fun my_print_tac ctxt thm =
let
  val _ = tracing (string_of_thm_no_vars ctxt thm)
in
  Seq.single thm
end
```

which prints out the given theorem (using the string-function defined in Section 2.2) and then behaves like `all_tac`. With this tactic we are in the position to inspect every goal state in a proof. Consider now the proof in Figure 5.1: as can be seen, internally every goal state is an implication of the form

$$A_1 \implies \dots \implies A_n \implies \#C$$

where C is the goal to be proved and the A_i are the subgoals. So after setting up the lemma, the goal state is always of the form $C \implies \#C$; when the proof is finished we are left with $\#C$. Since the goal C can potentially be an implication, there is a “protector” wrapped around it (the wrapper is the outermost constant `Const` (“prop”, `bool ⇒ bool`); in the figure we make it visible as an `#`). This wrapper prevents that premises of C are misinterpreted as open subgoals. While tactics can operate on the subgoals (the A_i above), they are expected to leave the conclusion

```

notation (output) "prop" ("#_" [1000] 1000)

lemma shows "[[A; B]] ==> A & B"
apply(tactic {* my_print_tac @{context} *})
goal (1 subgoal):
  1. [[A; B]] ==> A & B

internal goal state:
([A; B] ==> A & B) ==> #([A; B] ==> A & B)

apply(rule conjI)
apply(tactic {* my_print_tac @{context} *})
goal (2 subgoals):
  1. [[A; B]] ==> A
  2. [[A; B]] ==> B

internal goal state:
([[A; B] ==> A; [A; B] ==> B] ==> #([A; B] ==> A & B))

apply(assumption)
apply(tactic {* my_print_tac @{context} *})
goal (1 subgoal):
  1. [[A; B]] ==> B

internal goal state:
([A; B] ==> B) ==> #([A; B] ==> A & B)

apply(assumption)
apply(tactic {* my_print_tac @{context} *})
No subgoals!

internal goal state:
#([A; B] ==> A & B)

```

Figure 5.1: The figure shows a proof where each intermediate goal state is printed by the Isabelle system and by `my_print_tac`. The latter shows the goal state as represented internally (highlighted boxes). This tactic shows that every goal state in Isabelle is represented by a theorem: when you start the proof of $\llbracket A; B \rrbracket \implies A \wedge B$ the theorem is $(\llbracket A; B \rrbracket \implies A \wedge B) \implies \#(\llbracket A; B \rrbracket \implies A \wedge B)$; when you finish the proof the theorem is $\#(\llbracket A; B \rrbracket \implies A \wedge B)$.

C intact, with the exception of possibly instantiating schematic variables. If you use the predefined tactics, which we describe in the next section, this will always be the case.

Read More

For more information about the internals of goals see [Impl. Man., Sec. 3.1].

5.2 Simple Tactics

Let us start with explaining the simple tactic `print_tac`, which is quite useful for low-level debugging of tactics. It just prints out a message and the current goal state. Unlike `my_print_tac` shown earlier, it prints the goal state as the user would see it. For example, processing the proof

```
lemma shows "False  $\implies$  True"
apply(tactic {* print_tac "foo message" *})
```

gives:

```
foo message
```

```
False  $\implies$  True
1. False  $\implies$  True
```

A simple tactic for easy discharge of any proof obligations is `cheat_tac`. This tactic corresponds to the Isabelle command **sorry** and is sometimes useful during the development of tactics.

```
lemma shows "False" and "Goldbach_conjecture"
apply(tactic {* SkipProof.cheat_tac @theory *})
```

```
No subgoals!
```

This tactic works however only if the quick-and-dirty mode of Isabelle is switched on.

Another simple tactic is the function `atac`, which, as shown in the previous section, corresponds to the assumption command.

```
lemma shows "P  $\implies$  P"
apply(tactic {* atac 1 *})
```

```
No subgoals!
```

Similarly, `rtac`, `dtac`, `etac` and `ftac` correspond (roughly) to `rule`, `drule`, `erule` and `frule`, respectively. Each of them takes a theorem as argument and attempts to apply it to a goal. Below are three self-explanatory examples.

```
lemma shows "P  $\wedge$  Q"
apply(tactic {* rtac @thm conjI 1 *})
```

```
goal (2 subgoals):
 1. P
 2. Q
```

```
lemma shows "P ∧ Q ⇒ False"
apply(tactic {* etac @[thm conjE] 1 *})
```

```
goal (1 subgoal):
1. [[P; Q]] ⇒ False
```

```
lemma shows "False ∧ True ⇒ False"
apply(tactic {* dtac @[thm conjunct2] 1 *})
```

```
goal (1 subgoal):
1. True ⇒ False
```

The function `resolve_tac` is similar to `rtac`, except that it expects a list of theorems as arguments. From this list it will apply the first applicable theorem (later theorems that are also applicable can be explored via the lazy sequences mechanism). Given the code

```
val resolve_xmp_tac = resolve_tac [@[thm impI], @[thm conjI]]
```

an example for `resolve_tac` is the following proof where first an outermost implication is analysed and then an outermost conjunction.

```
lemma shows "C ⇒ (A ∧ B)" and "(A ⇒ B) ∧ C"
apply(tactic {* resolve_xmp_tac 1 *})
apply(tactic {* resolve_xmp_tac 2 *})
```

```
goal (3 subgoals):
1. C ⇒ A ∧ B
2. A ⇒ B
3. C
```

Similar versions taking a list of theorems exist for the tactics `dtac` (`dresolve_tac`), `etac` (`eresolve_tac`) and so on.

Another simple tactic is `cut_facts_tac`. It inserts a list of theorems into the assumptions of the current goal state. For example

```
lemma shows "True ≠ False"
apply(tactic {* cut_facts_tac [@[thm True_def], @[thm False_def]] 1 *})
```

produces the goal state

```
goal (1 subgoal):
1. [[True ≡ (λx. x) = (λx. x); False ≡ ∀P. P]] ⇒ True ≠ False
```

Since rules are applied using higher-order unification, an automatic proof procedure might become too fragile, if it just applies inference rules as shown above. The reason is that a number of rules introduce meta-variables into the goal state. Consider for example the proof

```
lemma shows "∀x ∈ A. P x ⇒ Q x"
apply(tactic {* dtac @[thm bspec] 1 *})
```

```
goal (2 subgoals):
1. ?x ∈ A
2. P ?x ⇒ Q x
```

where the application of rule *bspec* generates two subgoals involving the meta-variable *?x*. Now, if you are not careful, tactics applied to the first subgoal might instantiate this meta-variable in such a way that the second subgoal becomes unprovable. If it is clear what the *?x* should be, then this situation can be avoided by introducing a more constrained version of the *bspec*-rule. Such constraints can be given by pre-instantiating theorems with other theorems. One function to do this is *RS*

```
@{thm disjI1} RS @{thm conjI}
> [[?P1; ?Q]] ==> (?P1 ∨ ?Q1) ∧ ?Q
```

which in the example instantiates the first premise of the *conjI*-rule with the rule *disjI1*. If the instantiation is impossible, as in the case of

```
@{thm conjI} RS @{thm mp}
> *** Exception- THM ("RSN: no unifiers", 1,
> ["[[?P; ?Q]] ==> ?P ∧ ?Q", "[[?P → ?Q; ?P]] ==> ?Q"]) raised
```

then the function raises an exception. The function *RSN* is similar to *RS*, but takes an additional number as argument that makes explicit which premise should be instantiated.

To improve readability of the theorems we shall produce below, we will use the function *no_vars* from Section 2.2, which transforms schematic variables into free ones. Using this function for the first *RS*-expression above produces the more readable result:

```
no_vars @{context} (@{thm disjI1} RS @{thm conjI})
> [[P; Q]] ==> (P ∨ Qa) ∧ Q
```

If you want to instantiate more than one premise of a theorem, you can use the function *MRS*:

```
no_vars @{context} ([@{thm disjI1}, @{thm disjI2}] MRS @{thm conjI})
> [[P; Q]] ==> (P ∨ Qa) ∧ (Pa ∨ Q)
```

If you need to instantiate lists of theorems, you can use the functions *RL* and *MRL*. For example in the code below, every theorem in the second list is instantiated with every theorem in the first.

```
map (no_vars @{context})
  ([@{thm impI}, @{thm disjI2}] RL [@{thm conjI}, @{thm disjI1}])
> [[P ==> Q; Qa]] ==> (P → Q) ∧ Qa,
> [[Q; Qa]] ==> (P ∨ Q) ∧ Qa,
> (P ==> Q) ==> (P → Q) ∨ Qa,
> Q ==> (P ∨ Q) ∨ Qa]
```

Read More

The combinators for instantiating theorems are defined in [Pure/drule.ML](#).

Often proofs on the ML-level involve elaborate operations on assumptions and \wedge -quantified variables. To do such operations using the basic tactics shown so far is very unwieldy and brittle. Some convenience and safety is provided by the functions *FOCUS* and *SUBPROOF*. These tactics fix the parameters and bind the various components of a goal state to a record. To see what happens, use the function defined in Figure 5.2, which takes a record and just prints out the contents of this record. Consider now the proof:

lemma shows " $\wedge x y. A x y \implies B y x \longrightarrow C (?z y) x$ "
apply(tactic {* Subgoal.FOCUS foc_tac @{context} 1 *})

The tactic produces the following printout:

```
params:      x, y
schematics:  ?z
assumptions: A x y
conclusion:   B y x  $\longrightarrow$  C (z y) x
premises:    A x y
```

(FIXME: Find out how nowadays the schematics are handled)

Notice in the actual output the brown colour of the variables x , and y . Although they are parameters in the original goal, they are fixed inside the tactic. By convention these fixed variables are printed in brown colour. Similarly the schematic variable $?z$. The assumption, or premise, $A x y$ is bound as *cterm* to the record-variable *asms*, but also as *thm* to *prems*.

If we continue the proof script by applying the *impI*-rule

apply(rule *impI*)
apply(tactic {* Subgoal.FOCUS foc_tac @{context} 1 *})

then the tactic prints out:

```
params:      x, y
schematics:  ?z
assumptions: A x y, B y x
conclusion:   C (z y) x
premises:    A x y, B y x
```

Now also $B y x$ is an assumption bound to *asms* and *prems*.

The difference between the tactics *SUBPROOF* and *FOCUS* is that the former expects that the goal is solved completely, which the latter does not. *SUBPROOF* can also not instantiate an schematic variables.

One convenience of both *FOCUS* and *SUBPROOF* is that we can apply the assumptions using the usual tactics, because the parameter *prems* contains them as theorems. With this you can easily implement a tactic that behaves almost like *atac*:

```

fun foc_tac {prems, params, asms, concl, context, schematics} =
let
  val string_of_params = string_of_cterms context (map snd params)
  val string_of_asms = string_of_cterms context asms
  val string_of_concl = string_of_cterm context concl
  val string_of_prems = string_of_thms_no_vars context prems
  val string_of_schms = string_of_cterms context (map fst (snd schematics))

  val strs = ["params: " ^ string_of_params,
              "schematics: " ^ string_of_schms,
              "assumptions: " ^ string_of_asms,
              "conclusion: " ^ string_of_concl,
              "premises: " ^ string_of_prems]
in
  tracing (cat_lines strs); all_tac
end

```

Figure 5.2: A function that prints out the various parameters provided by *FOCUS* and *SUBPROOF*. It uses the functions defined in Section 2.2 for extracting strings from *cterms* and *thms*.

```

val atac' = Subgoal.FOCUS (fn {prems, ...} => resolve_tac prems 1)

```

If you apply *atac'* to the next lemma

```

lemma shows "[B x y; A x y; C x y]  $\implies$  A x y"
apply(tactic {* atac' @context 1 *})

```

it will produce

No subgoals!

Notice that *atac'* inside *FOCUS* calls *resolve_tac* with the subgoal number 1 and also the outer call to *FOCUS* in the **apply**-step uses 1. This is another advantage of *FOCUS* and *SUBPROOF*: the addressing inside it is completely local to the tactic inside the subproof. It is therefore possible to also apply *atac'* to the second goal by just writing:

```

lemma shows "True" and "[B x y; A x y; C x y]  $\implies$  A x y"
apply(tactic {* atac' @context 2 *})
apply(rule TrueI)
done

```

Read More

The functions *FOCUS* and *SUBPROOF* are defined in *Pure/subgoal.ML* and also described in [Impl. Man., Sec. 4.3].

Similar but less powerful functions than *FOCUS*, respectively *SUBPROOF*, are *SUBGOAL* and *CSUBGOAL*. They allow you to inspect a given subgoal (the former presents the

subgoal as a *term*, while the latter as a *cterm*). With this you can implement a tactic that applies a rule according to the topmost logic connective in the subgoal (to illustrate this we only analyse a few connectives). The code of the tactic is as follows.

```

1 fun select_tac (t, i) =
2   case t of
3     @{term "Trueprop"} $ t' => select_tac (t', i)
4   | @{term "op ==>"} $ _ $ t' => select_tac (t', i)
5   | @{term "op ^"} $ _ $ _ => rtac @{thm conjI} i
6   | @{term "op ->"} $ _ $ _ => rtac @{thm impI} i
7   | @{term "Not"} $ _ => rtac @{thm notI} i
8   | Const (@{const_name "All"}, _) $ _ => rtac @{thm allI} i
9   | _ => all_tac

```

The input of the function is a term representing the subgoal and a number specifying the subgoal of interest. In Line 3 you need to descend under the outermost *Trueprop* in order to get to the connective you like to analyse. Otherwise goals like $A \wedge B$ are not properly analysed. Similarly with meta-implications in the next line. While for the first five patterns we can use the *@term*-antiquotation to construct the patterns, the pattern in Line 8 cannot be constructed in this way. The reason is that an antiquotation would fix the type of the quantified variable. So you really have to construct the pattern using the basic term-constructors. This is not necessary in other cases, because their type is always fixed to function types involving only the type *bool*. (See Section 3.2 about constructing terms manually.) For the catch-all pattern, we chose to just return *all_tac*. Consequently, *select_tac* never fails.

Let us now see how to apply this tactic. Consider the four goals:

lemma shows "A \wedge B" and "A \longrightarrow B \longrightarrow C" and " $\forall x. D x$ " and "E \implies F"

```

apply(tactic {* SUBGOAL select_tac 4 *})
apply(tactic {* SUBGOAL select_tac 3 *})
apply(tactic {* SUBGOAL select_tac 2 *})
apply(tactic {* SUBGOAL select_tac 1 *})

```

goal (5 subgoals):

1. A
2. B
3. $A \implies B \longrightarrow C$
4. $\bigwedge x. D x$
5. $E \implies F$

where in all but the last the tactic applied an introduction rule. Note that we applied the tactic to the goals in “reverse” order. This is a trick in order to be independent from the subgoals that are produced by the rule. If we had applied it in the other order

lemma shows "A \wedge B" and "A \longrightarrow B \longrightarrow C" and " $\forall x. D x$ " and "E \implies F"

```

apply(tactic {* SUBGOAL select_tac 1 *})
apply(tactic {* SUBGOAL select_tac 3 *})
apply(tactic {* SUBGOAL select_tac 4 *})
apply(tactic {* SUBGOAL select_tac 5 *})

```


and call `foo_tac1`.

With the combinators `THEN'`, `EVERY'` and `EVERY1` it must be guaranteed that all component tactics successfully apply; otherwise the whole tactic will fail. If you rather want to try out a number of tactics, then you can use the combinator `ORELSE'` for two tactics, and `FIRST'` (or `FIRST1`) for a list of tactics. For example, the tactic

```
val orelse_xmp_tac = rtac @{thm disjI1} ORELSE' rtac @{thm conjI}
```

will first try out whether rule `disjI` applies and in case of failure will try `conjI`. To see this consider the proof

```
lemma shows "True  $\wedge$  False" and "Foo  $\vee$  Bar"
apply(tactic {* orelse_xmp_tac 2 *})
apply(tactic {* orelse_xmp_tac 1 *})
```

which results in the goal state

```
goal (3 subgoals):
  1. True
  2. False
  3. Foo
```

Using `FIRST'` we can simplify our `select_tac` from Page 83 as follows:

```
val select_tac' = FIRST' [rtac @{thm conjI}, rtac @{thm impI},
                          rtac @{thm notI}, rtac @{thm allI}, K all_tac]
```

Since we like to mimic the behaviour of `select_tac` as closely as possible, we must include `all_tac` at the end of the list, otherwise the tactic will fail if no rule applies (we also have to wrap `all_tac` using the `K`-combinator, because it does not take a subgoal number as argument). You can test the tactic on the same goals:

```
lemma shows "A  $\wedge$  B" and "A  $\longrightarrow$  B  $\longrightarrow$  C" and " $\forall x. D x$ " and "E  $\implies$  F"
apply(tactic {* select_tac' 4 *})
apply(tactic {* select_tac' 3 *})
apply(tactic {* select_tac' 2 *})
apply(tactic {* select_tac' 1 *})
```

```
goal (5 subgoals):
  1. A
  2. B
  3. A  $\implies$  B  $\longrightarrow$  C
  4.  $\bigwedge x. D x$ 
  5. E  $\implies$  F
```

Since such repeated applications of a tactic to the reverse order of *all* subgoals is quite common, there is the tactic combinator `ALLGOALS` that simplifies this. Using this combinator you can simply write:

```
lemma shows "A  $\wedge$  B" and "A  $\longrightarrow$  B  $\longrightarrow$  C" and " $\forall x. D x$ " and "E  $\implies$  F"
apply(tactic {* ALLGOALS select_tac' *})
```

```
goal (5 subgoals):
  1. A
  2. B
  3.  $A \implies B \longrightarrow C$ 
  4.  $\bigwedge x. D x$ 
  5.  $E \implies F$ 
```

Remember that we chose to implement `select_tac'` so that it always succeeds by adding `all_tac` at the end of the tactic list. The same can be achieved with the tactic combinator `TRY`. For example:

```
val select_tac'' = TRY o FIRST' [rtac @{thm conjI}, rtac @{thm impI},
                                rtac @{thm notI}, rtac @{thm allI}]
```

This tactic behaves in the same way as `select_tac'`: it tries out one of the given tactics and if none applies leaves the goal state unchanged. This, however, can be potentially very confusing when visible to the user, for example, in cases where the goal is the form

```
lemma shows "E  $\implies$  F"
apply(tactic {* select_tac' 1 *})
```

```
goal (1 subgoal):
  1.  $E \implies F$ 
```

In this case no rule applies, but because of `TRY` or the inclusion of `all_tac` the tactics do not fail. The problem with this is that for the user there is little chance to see whether or not progress in the proof has been made. By convention therefore, tactics visible to the user should either change something or fail.

To comply with this convention, we could simply delete the `K all_tac` from the end of the theorem list. As a result `select_tac'` would only succeed on goals where it can make progress. But for the sake of argument, let us suppose that this deletion is *not* an option. In such cases, you can use the combinator `CHANGED` to make sure the subgoal has been changed by the tactic. Because now

```
lemma shows "E  $\implies$  F"
apply(tactic {* CHANGED (select_tac' 1) *})
```

gives the error message:

```
*** empty result sequence -- proof command failed
*** At command "apply".
```

We can further extend `select_tac'` so that it not just applies to the topmost connective, but also to the ones immediately “underneath”, i.e. analyse the goal completely. For this you can use the tactic combinator `REPEAT`. As an example suppose the following tactic

```
val repeat_xmp_tac = REPEAT (CHANGED (select_tac' 1))
```

which applied to the proof

lemma shows " $((\neg A) \wedge (\forall x. B\ x)) \wedge (C \longrightarrow D)$ "
apply(tactic {** repeat_xmp_tac **})

produces

goal (3 subgoals):

1. $A \implies \text{False}$
2. $\forall x. B\ x$
3. $C \longrightarrow D$

Here it is crucial that *select_tac'* is prefixed with *CHANGED*, because otherwise *REPEAT* runs into an infinite loop (it applies the tactic as long as it succeeds). The function *REPEAT1* is similar, but runs the tactic at least once (failing if this is not possible).

If you are after the “primed” version of *repeat_xmp_tac*, then you can implement it as

```
val repeat_xmp_tac' = REPEAT o CHANGED o select_tac'
```

since there are no “primed” versions of *REPEAT* and *CHANGED*.

If you look closely at the goal state above, then you see the tactics *repeat_xmp_tac* and *repeat_xmp_tac'* are not yet quite what we are after: the problem is that goals 2 and 3 are not analysed. This is because the tactic is applied repeatedly only to the first subgoal. To analyse also all resulting subgoals, you can use the tactic combinator *REPEAT_ALL_NEW*. Suppose the tactic

```
val repeat_all_new_xmp_tac = REPEAT_ALL_NEW (CHANGED o select_tac')
```

you see that the following goal

lemma shows " $((\neg A) \wedge (\forall x. B\ x)) \wedge (C \longrightarrow D)$ "
apply(tactic {** repeat_all_new_xmp_tac 1 **})

goal (3 subgoals):

1. $A \implies \text{False}$
2. $\bigwedge x. B\ x$
3. $C \implies D$

is completely analysed according to the theorems we chose to include in *select_tac'*.

Recall that tactics produce a lazy sequence of successor goal states. These states can be explored using the command **back**. For example

lemma " $\llbracket P1 \vee Q1; P2 \vee Q2 \rrbracket \implies R$ "
apply(tactic {** etac @{\thm disjE} 1 **})

applies the rule to the first assumption yielding the goal state:

goal (2 subgoals):

1. $\llbracket P2 \vee Q2; P1 \rrbracket \implies R$
2. $\llbracket P2 \vee Q2; Q1 \rrbracket \implies R$

After typing

back

the rule now applies to the second assumption.

```
goal (2 subgoals):
  1.  $\llbracket P1 \vee Q1; P2 \rrbracket \implies R$ 
  2.  $\llbracket P1 \vee Q1; Q2 \rrbracket \implies R$ 
```

Sometimes this leads to confusing behaviour of tactics and also has the potential to explode the search space for tactics. These problems can be avoided by prefixing the tactic with the tactic combinator *DETERM*.

lemma " $\llbracket P1 \vee Q1; P2 \vee Q2 \rrbracket \implies R$ "
apply(tactic {* *DETERM* (etac @*{thm disjE}* 1) *)})

```
goal (2 subgoals):
  1.  $\llbracket P2 \vee Q2; P1 \rrbracket \implies R$ 
  2.  $\llbracket P2 \vee Q2; Q1 \rrbracket \implies R$ 
```

This combinator will prune the search space to just the first successful application. Attempting to apply **back** in this goal states gives the error message:

```
*** back: no alternatives
*** At command "back".
```

Recall that we implemented *select_tac'* on Page 85 specifically so that it always succeeds. We achieved this by adding at the end the tactic *all_tac*. We can achieve this also by using the combinator *TRY*. Suppose, for example the tactic

```
val select_tac'' = FIRST' [rtac @{thm conjI}, rtac @{thm impI},
                          rtac @{thm notI}, rtac @{thm allI}]
```

which will fail if none of the rules applies. However, if you prefix it as follows

```
val select_tac''' = TRY o select_tac''
```

then the tactic *select_tac''* will be tried out and any failure is harnessed. We again have to use the construction with *TRY o ...* since there is no primed version of *TRY*. The tactic combinator *TRYALL* will try out a tactic on all subgoals. For example the tactic

```
val triv_tac = TRYALL (rtac @{thm TrueI}) ORELSE' etac @{thm FalseE})
```

will solve all trivial subgoals involving *True* or *False*.

(FIXME: say something about *COND* and *COND'*)

(FIXME: PARALLEL-CHOICE PARALLEL-GOALS)

Read More

Most tactic combinators described in this section are defined in [Pure/tactical.ML](#). Some combinators for the purpose of proof search are implemented in [Pure/search.ML](#).

Exercise 5.3.1. Dyckhoff presents in [2] inference rules for a sequent calculus, called *G4ip*, in which no contraction rule is needed in order to be complete. As a result the rules applied in any order give a simple decision procedure for propositional intuitionistic logic. His rules are

$$\begin{array}{c}
\frac{}{A, \Gamma \Rightarrow A} \text{Ax} \qquad \frac{}{\text{False}, \Gamma \Rightarrow G} \text{False} \\
\frac{A, B, \Gamma \Rightarrow G}{A \wedge B, \Gamma \Rightarrow G} \wedge_L \qquad \frac{\Gamma \Rightarrow A \quad \Gamma \Rightarrow B}{\Gamma \Rightarrow A \wedge B} \wedge_R \\
\frac{A, \Gamma \Rightarrow G \quad B, \Gamma \Rightarrow G}{A \vee B, \Gamma \Rightarrow G} \vee_L \qquad \frac{\Gamma \Rightarrow A}{\Gamma \Rightarrow A \vee B} \vee_{R1} \quad \frac{\Gamma \Rightarrow B}{\Gamma \Rightarrow A \vee B} \vee_{R2} \\
\frac{B, A, \Gamma \Rightarrow G}{A \longrightarrow B, A, \Gamma \Rightarrow G} \longrightarrow_{L1} \qquad \frac{A, \Gamma \Rightarrow B}{\Gamma \Rightarrow A \longrightarrow B} \longrightarrow_R \\
\frac{C \longrightarrow (D \longrightarrow B), \Gamma \Rightarrow G}{(C \wedge D) \longrightarrow B, \Gamma \Rightarrow G} \longrightarrow_{L2} \qquad \frac{C \longrightarrow B, D \longrightarrow B, \Gamma \Rightarrow G}{(C \vee D) \longrightarrow B, \Gamma \Rightarrow G} \longrightarrow_{L3} \\
\frac{D \longrightarrow B, \Gamma \Rightarrow C \longrightarrow D \quad B, \Gamma \Rightarrow G}{(C \longrightarrow D) \longrightarrow B, \Gamma \Rightarrow G} \longrightarrow_{L4}
\end{array}$$

Implement a tactic that explores all possibilities of applying these rules to a propositional formula until a goal state is reached in which all subgoals are discharged. Note that in Isabelle the left-rules need to be implemented as elimination rules. You need to prove separate lemmas corresponding to $\longrightarrow_{L1..4}$. In order to explore all possibilities of applying the rules, use the tactic combinator `DEPTH_SOLVE`, which searches for a state in which all subgoals are solved. Add also rules for equality and run your tactic on the de Bruijn formulae discussed in Exercise 3.2.3.

5.4 Simplifier Tactics

A lot of convenience in the reasoning with Isabelle derives from its powerful simplifier. The power of the simplifier is a strength and a weakness at the same time, because you can easily make the simplifier run into a loop and in general its behaviour can be difficult to predict. There is also a multitude of options that you can configure to control the behaviour of the simplifier. We describe some of them in this and the next section.

There are the following five main tactics behind the simplifier (in parentheses is their user-level counterpart):

<code>simp_tac</code>	<code>(simp (no_asm))</code>
<code>asm_simp_tac</code>	<code>(simp (no_asm_simp))</code>
<code>full_simp_tac</code>	<code>(simp (no_asm_use))</code>
<code>asm_lr_simp_tac</code>	<code>(simp (asm_lr))</code>
<code>asm_full_simp_tac</code>	<code>(simp)</code>

All of the tactics take a simpset and an integer as argument (the latter as usual to specify the goal to be analysed). So the proof

```
lemma "Suc (1 + 2) < 3 + 2"
apply(simp)
done
```

corresponds on the ML-level to the tactic

```
lemma "Suc (1 + 2) < 3 + 2"
apply(tactic {* asm_full_simp_tac @{simpset} 1 *})
done
```

If the simplifier cannot make any progress, then it leaves the goal unchanged, i.e., does not raise any error message. That means if you use it to unfold a definition for a constant and this constant is not present in the goal state, you can still safely apply the simplifier.

(FIXME: show rewriting of cterms)

When using the simplifier, the crucial information you have to provide is the simpset. If this information is not handled with care, then the simplifier can easily run into a loop. Therefore a good rule of thumb is to use simpsets that are as minimal as possible. It might be surprising that a simpset is more complex than just a simple collection of theorems used as simplification rules. One reason for the complexity is that the simplifier must be able to rewrite inside terms and should also be able to rewrite according to rules that have preconditions.

The rewriting inside terms requires congruence rules, which are meta-equalities typical of the form

$$\frac{t_1 \equiv s_1 \dots t_n \equiv s_n}{\text{constr } t_1 \dots t_n \equiv \text{constr } s_1 \dots s_n}$$

with *constr* being a constant, like *If* or *Let*. Every simpset contains only one congruence rule for each term-constructor, which however can be overwritten. The user can declare lemmas to be congruence rules using the attribute *[cong]*. In HOL, the user usually states these lemmas as equations, which are then internally transformed into meta-equations.

The rewriting with rules involving preconditions requires what is in Isabelle called a subgoal, a solver and a looper. These can be arbitrary tactics that can be installed in a simpset and which are called at various stages during simplification. However, simpsets also include simprocs, which can produce rewrite rules on demand (see next section). Another component are split-rules, which can simplify for example the “then” and “else” branches of if-statements under the corresponding preconditions.

Read More

For more information about the simplifier see [Pure/meta_simplifier.ML](#) and [Pure/simplifier.ML](#). The simplifier for HOL is set up in [HOL/Tools/simpdata.ML](#). The generic splitter is implemented in [Provers/splitter.ML](#).

Read More

FIXME: Find the right place: Discrimination nets are implemented in [Pure/net.ML](#).

The most common combinators to modify simpsets are:


```

addsimps      delsimps
addcongs      delcongs
addsimprocs   delsimprocs

```

(FIXME: What about splitters? *addsplits*, *delsplits*)

To see how they work, consider the function in Figure 5.3, which prints out some parts of a simpset. If you use it to print out the components of the empty simpset, i.e., *empty_ss*

```

print_ss @{context} empty_ss
> Simplification rules:
> Congruences rules:
> Simproc patterns:

```

you can see it contains nothing. This simpset is usually not useful, except as a building block to build bigger simpsets. For example you can add to *empty_ss* the simplification rule *Diff_Int* as follows:

```

val ss1 = empty_ss addsimps [@{thm Diff_Int} RS @{thm eq_reflection}]

```

Printing then out the components of the simpset gives:

```

print_ss @{context} ss1
> Simplification rules:
>   ??.unknown: A - B ∩ C ≡ A - B ∪ (A - C)
> Congruences rules:
> Simproc patterns:

```

(FIXME: Why does it print out ??.unknown)

Adding also the congruence rule *UN_cong*

```

val ss2 = ss1 addcongs [@{thm UN_cong} RS @{thm eq_reflection}]

```

gives

```

print_ss @{context} ss2
> Simplification rules:
>   ??.unknown: A - B ∩ C ≡ A - B ∪ (A - C)
> Congruences rules:
>   UNION: [A = B; ∧x. x ∈ B ⇒ C x = D x] ⇒ ∪x∈A. C x ≡ ∪x∈B. D x
> Simproc patterns:

```

Notice that we had to add these lemmas as meta-equations. The *empty_ss* expects this form of the simplification and congruence rules. However, even when adding these lemmas to *empty_ss* we do not end up with anything useful yet.

In the context of HOL, the first really useful simpset is *HOL_basic_ss*. While printing out the components of this simpset

```

fun print_ss ctxt ss =
let
  val {simps, congs, procs, ...} = Simplifier.dest_ss ss

  fun name_thm (nm, thm) =
    " " ^ nm ^ ": " ^ (string_of_thm_no_vars ctxt thm)
  fun name_ctrm (nm, ctrm) =
    " " ^ nm ^ ": " ^ (string_of_cterms ctxt ctrm)

  val s = ["Simplification rules:" @ map name_thm simps @
    ["Congruences rules:" @ map name_thm congs @
    ["Simproc patterns:" @ map name_ctrm procs
in
  s |> cat_lines
    |> tracing
end

```

Figure 5.3: The function `dest_ss` returns a record containing all printable information stored in a simpset. We are here only interested in the simplification rules, congruence rules and simprocs.

```

print_ss @{context} HOL_basic_ss
> Simplification rules:
> Congruences rules:
> Simproc patterns:

```

also produces “nothing”, the printout is misleading. In fact the `HOL_basic_ss` is setup so that it can solve goals of the form

```
True, t = t, t ≡ t and False ⇒ P;
```

and also resolve with assumptions. For example:

```

lemma
  "True" and "t = t" and "t ≡ t" and "False ⇒ Foo" and "[A; B; C] ⇒ A"
apply(tactic {* ALLGOALS (simp_tac HOL_basic_ss) *})
done

```

This behaviour is not because of simplification rules, but how the subgoal, solver and looper are set up in `HOL_basic_ss`.

The simpset `HOL_ss` is an extension of `HOL_basic_ss` containing already many useful simplification and congruence rules for the logical connectives in HOL.

```

print_ss @{context} HOL_ss
> Simplification rules:
> Pure.triv_forall_equality: (∧x. PROP V) ≡ PROP V
> HOL.the_eq_trivial: THE x. x = y ≡ y
> HOL.the_sym_eq_trivial: THE ya. y = ya ≡ y

```

```

> ...
> Congruences rules:
>   HOL.simp_implies: ...
>      $\implies (PROP P =simp=> PROP Q) \equiv (PROP P' =simp=> PROP Q')$ 
>   op -->:  $\llbracket P \equiv P'; P' \implies Q \equiv Q' \rrbracket \implies P \longrightarrow Q \equiv P' \longrightarrow Q'$ 
> Simproc patterns:
> ...

```

The simplifier is often used to unfold definitions in a proof. For this the simplifier implements the tactic `rewrite_goals_tac`.³ Suppose for example the definition

definition `"MyTrue \equiv True"`

then in the following proof we can unfold this constant

lemma shows `"MyTrue \implies True \wedge True"`

`apply(rule conjI)`

`apply(tactic {* rewrite_goals_tac @{thms MyTrue_def} *})`

producing the goal state

`goal (2 subgoals):`

1. `True \implies True`
2. `True \implies True`

As you can see, the tactic unfolds the definitions in all subgoals.

The simplifier is often used in order to bring terms into a normal form. Unfortunately, often the situation arises that the corresponding simplification rules will cause the simplifier to run into an infinite loop. Consider for example the simple theory about permutations over natural numbers shown in Figure 5.4. The purpose of the lemmas is to push permutations as far inside as possible, where they might disappear by Lemma `perm_rev`. However, to fully normalise all instances, it would be desirable to add also the lemma `perm_compose` to the simplifier for pushing permutations over other permutations. Unfortunately, the right-hand side of this lemma is again an instance of the left-hand side and so causes an infinite loop. There seems to be no easy way to reformulate this rule and so one ends up with clunky proofs like:

lemma

`fixes c d::"nat" and pi1 pi2::"prm"`

`shows "pi1.(c, pi2.(rev pi1).d) = (pi1.c, (pi1.pi2).d)"`

`apply(simp)`

`apply(rule trans)`

`apply(rule perm_compose)`

`apply(simp)`

`done`

It is however possible to create a single simplifier tactic that solves such proofs. The trick is to introduce an auxiliary constant for permutations and split the simplification into two phases (below actually three). Let assume the auxiliary constant is

definition

`perm_aux :: "prm \Rightarrow 'a \Rightarrow 'a" ("_ .aux _" [80,80] 80)`

³FIXME: see LocalDefs infrastructure.

```

types prm = "(nat × nat) list"
consts perm :: "prm ⇒ 'a ⇒ 'a" ("_ · _" [80,80] 80)

overloading
  perm_nat ≡ "perm :: prm ⇒ nat ⇒ nat"
  perm_prod ≡ "perm :: prm ⇒ ('a×'b) ⇒ ('a×'b)"
  perm_list ≡ "perm :: prm ⇒ 'a list ⇒ 'a list"
begin

fun swap::"nat ⇒ nat ⇒ nat ⇒ nat"
where
  "swap a b c = (if c=a then b else (if c=b then a else c))"

primrec perm_nat
where
  "perm_nat [] c = c"
| "perm_nat (ab#pi) c = swap (fst ab) (snd ab) (perm_nat pi c)"

fun perm_prod
where
  "perm_prod pi (x, y) = (pi·x, pi·y)"

primrec perm_list
where
  "perm_list pi [] = []"
| "perm_list pi (x#xs) = (pi·x)#(perm_list pi xs)"

end

lemma perm_append[simp]:
fixes c::"nat" and pi1 pi2::"prm"
shows "((pi1@pi2)·c) = (pi1·(pi2·c))"
by (induct pi1) (auto)

lemma perm_bij[simp]:
fixes c d::"nat" and pi::"prm"
shows "(pi·c = pi·d) = (c = d)"
by (induct pi) (auto)

lemma perm_rev[simp]:
fixes c::"nat" and pi::"prm"
shows "pi·((rev pi)·c) = c"
by (induct pi arbitrary: c) (auto)

lemma perm_compose:
fixes c::"nat" and pi1 pi2::"prm"
shows "pi1·(pi2·c) = (pi1·pi2)·(pi1·c)"
by (induct pi2) (auto)

```

Figure 5.4: A simple theory about permutations over *nats*. The point is that the lemma *perm_compose* cannot be directly added to the simplifier, as it would cause the simplifier to loop. It can still be used as a simplification rule if the permutation in the right-hand side is sufficiently protected.

where

```
"pi · aux c ≡ pi · c"
```

Now the two lemmas

```
lemma perm_aux_expand:
fixes c::"nat" and pi1 pi2::"prm"
shows "pi1·(pi2·c) = pi1 ·aux (pi2·c)"
unfolding perm_aux_def by (rule refl)
```

```
lemma perm_compose_aux:
fixes c::"nat" and pi1 pi2::"prm"
shows "pi1·(pi2·aux c) = (pi1·pi2) ·aux (pi1·c)"
unfolding perm_aux_def by (rule perm_compose)
```

are simple consequence of the definition and *perm_compose*. More importantly, the lemma *perm_compose_aux* can be safely added to the simplifier, because now the right-hand side is not anymore an instance of the left-hand side. In a sense it freezes all redexes of permutation compositions after one step. In this way, we can split simplification of permutations into three phases without the user noticing anything about the auxiliary constant. We first freeze any instance of permutation compositions in the term using lemma "*perm_aux_expand*" (Line 9); then simplify all other permutations including pushing permutations over other permutations by rule *perm_compose_aux* (Line 10); and finally “unfreeze” all instances of permutation compositions by unfolding the definition of the auxiliary constant.

```
1 val perm_simp_tac =
2 let
3   val thms1 = [@{thm perm_aux_expand}]
4   val thms2 = [@{thm perm_append}, @{thm perm_bij}, @{thm perm_rev},
5               @{thm perm_compose_aux}] @ @{thms perm_prod.simps} @
6               @{thms perm_list.simps} @ @{thms perm_nat.simps}
7   val thms3 = [@{thm perm_aux_def}]
8 in
9   simp_tac (HOL_basic_ss addsimps thms1)
10  THEN' simp_tac (HOL_basic_ss addsimps thms2)
11  THEN' simp_tac (HOL_basic_ss addsimps thms3)
12 end
```

For all three phases we have to build simpsets adding specific lemmas. As is sufficient for our purposes here, we can add these lemmas to *HOL_basic_ss* in order to obtain the desired results. Now we can solve the following lemma

```
lemma
fixes c d::"nat" and pi1 pi2::"prm"
shows "pi1·(c, pi2·((rev pi1)·d)) = (pi1·c, (pi1·pi2)·d)"
apply(tactic {* perm_simp_tac 1 *})
done
```

in one step. This tactic can deal with most instances of normalising permutations. In order to solve all cases we have to deal with corner-cases such as the lemma being an exact instance of the permutation composition lemma. This can often be done easier

by implementing a *simproc* or a conversion. Both will be explained in the next two chapters.

(FIXME: Is it interesting to say something about $op =simp=>?$)

(FIXME: What are the second components of the congruence rules—something to do with weak congruence constants?)

(FIXME: Anything interesting to say about *Simplifier.clear_ss*?)

(FIXME: *ObjectLogic.full_atomize_tac*, *ObjectLogic.rulify_tac*)

(FIXME: what are *mksimps_pairs*; used in *Nominal.thy*)

(FIXME: explain *simplify* and *Simplifier.rewrite_rule* etc.)

5.5 Simprocs

In Isabelle you can also implement custom simplification procedures, called *simprocs*. Simprocs can be triggered by the simplifier on a specified term-pattern and rewrite a term according to a theorem. They are useful in cases where a rewriting rule must be produced on “demand” or when rewriting by simplification is too unpredictable and potentially loops.

To see how simprocs work, let us first write a simproc that just prints out the pattern which triggers it and otherwise does nothing. For this you can use the function:

```

1 fun fail_simproc simpset redex =
2   let
3     val ctxt = Simplifier.the_context simpset
4     val _ = tracing ("The redex: " ^ (string_of_cterm ctxt redex))
5   in
6     NONE
7   end

```

This function takes a *simpset* and a *redex* (a *cterm*) as arguments. In Lines 3 and 4, we first extract the context from the given *simpset* and then print out a message containing the *redex*. The function returns *NONE* (standing for an optional *thm*) since at the moment we are *not* interested in actually rewriting anything. We want that the simproc is triggered by the pattern *Suc n*. This can be done by adding the simproc to the current *simpset* as follows

```

simproc_setup fail ("Suc n") = {* K fail_simproc *}

```

where the second argument specifies the pattern and the right-hand side contains the code of the simproc (we have to use *K* since we are ignoring an argument about morphisms). After this, the simplifier is aware of the simproc and you can test whether it fires on the lemma:

```

lemma shows "Suc 0 = 1"
apply (simp)

```

```
> The redex: Suc 0
> The redex: Suc 0
```

This will print out the message twice: once for the left-hand side and once for the right-hand side. The reason is that during simplification the simplifier will at some point reduce the term 1 to $Suc\ 0$, and then the simproc “fires” also on that term.

We can add or delete the simproc from the current simpset by the usual **declare**-statement. For example the simproc will be deleted with the declaration

```
declare [[simproc del: fail]]
```

If you want to see what happens with just *this* simproc, without any interference from other rewrite rules, you can call *fail* as follows:

```
lemma shows "Suc 0 = 1"
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [@[simproc fail]]) 1*})
```

Now the message shows up only once since the term 1 is left unchanged.

Setting up a simproc using the command **simproc.setup** will always add automatically the simproc to the current simpset. If you do not want this, then you have to use a slightly different method for setting up the simproc. First the function *fail_simproc* needs to be modified to

```
fun fail_simproc' simpset redex =
let
  val ctxt = Simplifier.the_context simpset
  val _ = tracing ("The redex: " ^ (Syntax.string_of_term ctxt redex))
in
  NONE
end
```

Here the redex is given as a *term*, instead of a *cterm* (therefore we printing it out using the function *string_of_term*). We can turn this function into a proper simproc using the function *Simplifier.simproc_i*:

```
val fail' =
let
  val thy = @{theory}
  val pat = [@[term "Suc n"]]
in
  Simplifier.simproc_i thy "fail_simproc'" pat (K fail_simproc')
end
```

Here the pattern is given as *term* (instead of *cterm*). The function also takes a list of patterns that can trigger the simproc. Now the simproc is set up and can be explicitly added using *addsimprocs* to a simpset whenever needed.

Simprocs are applied from inside to outside and from left to right. You can see this in the proof

```
lemma shows "Suc (Suc 0) = (Suc 1)"
```

```
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [fail']) 1*})
```

The simproc *fail'* prints out the sequence

```
> Suc 0
> Suc (Suc 0)
> Suc 1
```

To see how a simproc applies a theorem, let us implement a simproc that rewrites terms according to the equation:

```
lemma plus_one:
  shows "Suc n  $\equiv$  n + 1" by simp
```

Simprocs expect that the given equation is a meta-equation, however the equation can contain preconditions (the simproc then will only fire if the preconditions can be solved). To see that one has relatively precise control over the rewriting with simprocs, let us further assume we want that the simproc only rewrites terms “greater” than *Suc 0*. For this we can write

```
fun plus_one_simproc ss redex =
  case redex of
    @{term "Suc 0"} => NONE
  | _ => SOME @{thm plus_one}
```

and set up the simproc as follows.

```
val plus_one =
let
  val thy = @{theory}
  val pat = [@{term "Suc n"}]
in
  Simplifier.simproc_i thy "sproc +1" pat (K plus_one_simproc)
end
```

Now the simproc is set up so that it is triggered by terms of the form *Suc n*, but inside the simproc we only produce a theorem if the term is not *Suc 0*. The result you can see in the following proof

```
lemma shows "P (Suc (Suc (Suc 0))) (Suc 0)"
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [plus_one]) 1*})
```

where the simproc produces the goal state

```
goal (1 subgoal):
  1. P (Suc 0 + 1 + 1) (Suc 0)
```

As usual with rewriting you have to worry about looping: you already have a loop with *plus_one*, if you apply it with the default simpset (because the default simpset contains a rule which just does the opposite of *plus_one*, namely rewriting “+ 1” to a successor). So you have to be careful in choosing the right simpset to which you add a simproc.

Next let us implement a simproc that replaces terms of the form $Suc\ n$ with the number n increased by one. First we implement a function that takes a term and produces the corresponding integer value.

```
fun dest_suc_trm ((Const (@{const_name "Suc"}, _)) $ t) = 1 + dest_suc_trm t
  | dest_suc_trm t = snd (HOLLogic.dest_number t)
```

It uses the library function `dest_number` that transforms (Isabelle) terms, like 0 , 1 , 2 and so on, into integer values. This function raises the exception `TERM`, if the term is not a number. The next function expects a pair consisting of a term t (containing `Sucs`) and the corresponding integer value n .

```
1 fun get_thm ctxt (t, n) =
2   let
3     val num = HOLLogic.mk_number @{typ "nat"} n
4     val goal = Logic.mk_equals (t, num)
5   in
6     Goal.prove ctxt [] [] goal (K (Arith_Data.arith_tac ctxt 1))
7   end
```

From the integer value it generates the corresponding number term, called `num` (Line 3), and then generates the meta-equation $t \equiv num$ (Line 4), which it proves by the arithmetic tactic in Line 6.

For our purpose at the moment, proving the meta-equation using `arith_tac` is fine, but there is also an alternative employing the simplifier with a special simpset. For the kind of lemmas we want to prove here, the simpset `num_ss` should suffice.

```
fun get_thm_alt ctxt (t, n) =
let
  val num = HOLLogic.mk_number @{typ "nat"} n
  val goal = Logic.mk_equals (t, num)
  val num_ss = HOL_ss addsimps [
    @{thm One_nat_def}, @{thm Let_def}] @
    @{thms nat_number} @ @{thms neg_simps} @ @{thms plus_nat_simps}
in
  Goal.prove ctxt [] [] goal (K (simp_tac num_ss 1))
end
```

The advantage of `get_thm_alt` is that it leaves very little room for something to go wrong; in contrast it is much more difficult to predict what happens with `arith_tac`, especially in more complicated circumstances. The disadvantage of `get_thm_alt` is to find a simpset that is sufficiently powerful to solve every instance of the lemmas we like to prove. This requires careful tuning, but is often necessary in “production code”.⁴

Anyway, either version can be used in the function that produces the actual theorem for the simproc.

⁴It would be of great help if there is another way than tracing the simplifier to obtain the lemmas that are successfully applied during simplification. Alas, there is none.

```

fun nat_number_simproc ss t =
let
  val ctxt = Simplifier.the_context ss
in
  SOME (get_thm ctxt (t, dest_suc_trm t))
  handle TERM _ => NONE
end

```

This function uses the fact that `dest_suc_trm` might raise an exception `TERM`. In this case there is nothing that can be rewritten and therefore no theorem is produced (i.e. the function returns `NONE`). To try out the simproc on an example, you can set it up as follows:

```

val nat_number =
let
  val thy = @{theory}
  val pat = [ @{term "Suc n"} ]
in
  Simplifier.simproc_i thy "nat_number" pat (K nat_number_simproc)
end

```

Now in the lemma

```

lemma "P (Suc (Suc 2)) (Suc 99) (0::nat) (Suc 4 + Suc 0) (Suc (0 + 0))"
apply(tactic {* simp_tac (HOL_ss addsimprocs [nat_number]) 1*})

```

you obtain the more legible goal state

```

goal (1 subgoal):
  1. P 4 100 0 (5 + 1) (Suc (0 + 0))

```

where the simproc rewrites all `Sucs` except in the last argument. There it cannot rewrite anything, because it does not know how to transform the term `Suc (0 + 0)` into a number. To solve this problem have a look at the next exercise.

Exercise 5.5.1. Write a simproc that replaces terms of the form $t_1 + t_2$ by their result. You can assume the terms are “proper” numbers, that is of the form 0, 1, 2 and so on.

(FIXME: We did not do anything with morphisms. Anything interesting one can say about them?)

5.6 Conversions

Conversions are a thin layer on top of Isabelle’s inference kernel, and can be viewed as a controllable, bare-bone version of Isabelle’s simplifier. One difference between conversions and the simplifier is that the former act on *cterms* while the latter acts on *thms*. However, we will also show in this section how conversions can be applied to theorems via tactics. The type for conversions is

```
type conv = cterm -> thm
```

whereby the produced theorem is always a meta-equality. A simple conversion is the function `all_conv`, which maps a `cterm` to an instance of the (meta)reflexivity theorem. For example:

```
Conv.all_conv @{cterm "Foo ∨ Bar"}
> Foo ∨ Bar ≡ Foo ∨ Bar
```

Another simple conversion is `no_conv` which always raises the exception `CTERM`.

```
Conv.no_conv @{cterm True}
> *** Exception- CTERM ("no conversion", []) raised
```

A more interesting conversion is the function `beta_conversion`: it produces a meta-equation between a term and its beta-normal form. For example

```
let
  val add = @{cterm "λx y. x + (y::nat)"}
  val two = @{cterm "2::nat"}
  val ten = @{cterm "10::nat"}
  val ctrm = Thm.capply (Thm.capply add two) ten
in
  Thm.beta_conversion true ctrm
end
> ((λx y. x + y) 2) 10 ≡ 2 + 10
```

If you run this example, you will notice that the actual response is the seemingly nonsensical `2 + 10 ≡ 2 + 10`. The reason is that the pretty-printer for `cterm`s eta-normalises terms and therefore produces this output. If we get hold of the “raw” representation of the produced theorem, we obtain the expected result.

```
let
  val add = @{cterm "λx y. x + (y::nat)"}
  val two = @{cterm "2::nat"}
  val ten = @{cterm "10::nat"}
  val ctrm = Thm.capply (Thm.capply add two) ten
in
  Thm.prop_of (Thm.beta_conversion true ctrm)
end
> Const ("==", ...) $
>   (Abs ("x", ..., Abs ("y", ..., ...)) $...$...) $
>     (Const ("HOL.plus_class.plus", ...) $ ... $ ...)
```

The argument `true` in `beta_conversion` indicates that the right-hand side should be fully beta-normalised. If instead `false` is given, then only a single beta-reduction is performed on the outer-most level.

The main point of conversions is that they can be used for rewriting *cterm*s. One example is the function *rewr_conv*, which expects a meta-equation as an argument. Suppose the following meta-equation.

```
lemma true_conj1: "True  $\wedge$  P  $\equiv$  P" by simp
```

It can be used for example to rewrite $\text{True} \wedge (\text{Foo} \longrightarrow \text{Bar})$ to $\text{Foo} \longrightarrow \text{Bar}$. The code is as follows.

```
let
  val ctrm = @{cterm "True  $\wedge$  (Foo  $\longrightarrow$  Bar)"}
in
  Conv.rewr_conv @{thm true_conj1} ctrm
end
> True  $\wedge$  (Foo  $\longrightarrow$  Bar)  $\equiv$  Foo  $\longrightarrow$  Bar
```

Note, however, that the function *rewr_conv* only rewrites the outer-most level of the *cterm*. If the given *cterm* does not match exactly the left-hand side of the theorem, then *rewr_conv* fails, raising the exception *CTERM*.

This very primitive way of rewriting can be made more powerful by combining several conversions into one. For this you can use conversion combinators. The simplest conversion combinator is *then_conv*, which applies one conversion after another. For example

```
let
  val conv1 = Thm.beta_conversion false
  val conv2 = Conv.rewr_conv @{thm true_conj1}
  val ctrm = Thm.capply @{cterm " $\lambda x. x \wedge \text{False}$ "} @{cterm "True"}
in
  (conv1 then_conv conv2) ctrm
end
> ( $\lambda x. x \wedge \text{False}$ ) True  $\equiv$  False
```

where we first beta-reduce the term and then rewrite according to *true_conj1*. (When running this example recall the problem with the pretty-printer normalising all terms.)

The conversion combinator *else_conv* tries out the first one, and if it does not apply, tries the second. For example

```
let
  val conv = Conv.rewr_conv @{thm true_conj1} else_conv Conv.all_conv
  val ctrm1 = @{cterm "True  $\wedge$  Q"}
  val ctrm2 = @{cterm "P  $\vee$  (True  $\wedge$  Q)"}
in
  (conv ctrm1, conv ctrm2)
end
> (True  $\wedge$  Q  $\equiv$  Q, P  $\vee$  True  $\wedge$  Q  $\equiv$  P  $\vee$  True  $\wedge$  Q)
```

Here the conversion of *true_conj1* only applies in the first case, but fails in the second. The whole conversion does not fail, however, because the combinator *else_conv* will then try out *all_conv*, which always succeeds.

The conversion combinator *try_conv* constructs a conversion which is tried out on a term, but in case of failure just does nothing. For example

```
let
  val conv = Conv.try_conv (Conv.rewr_conv @{thm true_conj1})
  val ctrm = @{cterm "True ∨ P"}
in
  conv ctrm
end
> True ∨ P ≡ True ∨ P
```

Apart from the function *beta_conversion*, which is able to fully beta-normalise a term, the conversions so far are restricted in that they only apply to the outermost level of a *cterm*. In what follows we will lift this restriction. The combinators *fun_conv* and *arg_conv* will apply a conversion to the first, respectively second, argument of an application. For example

```
let
  val conv = Conv.arg_conv (Conv.rewr_conv @{thm true_conj1})
  val ctrm = @{cterm "P ∨ (True ∧ Q)"}
in
  conv ctrm
end
> P ∨ (True ∧ Q) ≡ P ∨ Q
```

The reason for this behaviour is that (*op* ∨) expects two arguments. Therefore the term must be of the form (*Const* ... \$ *t1*) \$ *t2*. The conversion is then applied to *t2*, which in the example above stands for *True* ∧ *Q*. The function *fun_conv* would apply the conversion to the term (*Const* ... \$ *t1*).

The function *abs_conv* applies a conversion under an abstraction. For example:

```
let
  val conv = Conv.rewr_conv @{thm true_conj1}
  val ctrm = @{cterm "λP. True ∧ (P ∧ Foo)"}
in
  Conv.abs_conv (K conv) @{context} ctrm
end
> λP. True ∧ (P ∧ Foo) ≡ λP. P ∧ Foo
```

Note that this conversion needs a context as an argument. We also give the conversion as (*K conv*), which is a function that ignores its argument (the argument being a sufficiently freshened version of the variable that is abstracted and a context). The conversion that goes under an application is *combination_conv*. It expects two conversions as arguments, each of which is applied to the corresponding “branch” of the

application. An abbreviation for this conversion is the function `comb_conv`, which applies the same conversion to both branches.

We can now apply all these functions in a conversion that recursively descends a term and applies a “`true_conj1`”-conversion in all possible positions.

```

1 fun all_true1_conv ctxt ctrm =
2   case (Thm.term_of ctrm) of
3     @{term "op ^"} $ @{term True} $ _ =>
4       (Conv.arg_conv (all_true1_conv ctxt) then_conv
5         Conv.rewr_conv @{thm true_conj1}) ctrm
6   / _ $ _ => Conv.comb_conv (all_true1_conv ctxt) ctrm
7   / Abs _ => Conv.abs_conv (fn (_, ctxt) => all_true1_conv ctxt) ctxt ctrm
8   / _ => Conv.all_conv ctrm

```

This function “fires” if the terms is of the form $(True \wedge \dots)$. It descends under applications (Line 6 and 7) and abstractions (Line 8); otherwise it leaves the term unchanged (Line 9). In Line 2 we need to transform the `ctrm` into a `term` in order to be able to pattern-match the term. To see this conversion in action, consider the following example:

```

let
  val conv = all_true1_conv @{context}
  val ctrm = @{cterm "distinct [1, x] → True ∧ 1 ≠ x"}
in
  conv ctrm
end
> distinct [1, x] → True ∧ 1 ≠ x ≡ distinct [1, x] → 1 ≠ x

```

To see how much control you have about rewriting by using conversions, let us make the task a bit more complicated by rewriting according to the rule `true_conj1`, but only in the first arguments of `If`s. Then the conversion should be as follows.

```

fun if_true1_conv ctxt ctrm =
  case Thm.term_of ctrm of
    Const (@{const_name If}, _) $ _ =>
      Conv.arg_conv (all_true1_conv ctxt) ctrm
  / _ $ _ => Conv.comb_conv (if_true1_conv ctxt) ctrm
  / Abs _ => Conv.abs_conv (fn (_, ctxt) => if_true1_conv ctxt) ctxt ctrm
  / _ => Conv.all_conv ctrm

```

Here is an example for this conversion:

```

let
  val conv = if_true1_conv @{context}
  val ctrm =
    @{cterm "if P (True ∧ 1 ≠ 2) then True ∧ True else True ∧ False"}
in
  conv ctrm

```

```
end
> if P (True ∧ 1 ≠ 2) then True ∧ True else True ∧ False
> ≡ if P (1 ≠ 2) then True ∧ True else True ∧ False
```

So far we only applied conversions to *cterm*s. Conversions can, however, also work on theorems using the function *fconv_rule*. As an example, consider the conversion *all_true1_conv* and the lemma:

lemma *foo_test*: " $P \vee (True \wedge \neg P)$ " **by** *simp*

Using the conversion *all_true1_conv* you can transform this theorem into a new theorem as follows

```
let
  val conv = Conv.fconv_rule (all_true1_conv @{context})
  val thm = @{thm foo_test}
in
  conv thm
end
> ?P ∨ ¬ ?P
```

Finally, conversions can also be turned into tactics and then applied to goal states. This can be done with the help of the function *CONVERSION*, and also some predefined conversion combinators that traverse a goal state. The combinators for the goal state are:

- *params_conv* for converting under parameters (i.e. where goals are of the form $\bigwedge x. P\ x \implies Q\ x$)
- *prems_conv* for applying a conversion to all premises of a goal, and
- *concl_conv* for applying a conversion to the conclusion of a goal.

Assume we want to apply *all_true1_conv* only in the conclusion of the goal, and *if_true1_conv* should only apply to the premises. Here is a tactic doing exactly that:

```
fun true1_tac ctxt =
  CONVERSION
  (Conv.params_conv ~1 (fn ctxt =>
    (Conv.prems_conv ~1 (if_true1_conv ctxt) then_conv
      Conv.concl_conv ~1 (all_true1_conv ctxt))) ctxt)
```

We call the conversions with the argument *~1*. This is to analyse all parameters, premises and conclusions. If we call them with a non-negative number, say *n*, then these conversions will only be called on *n* premises (similar for parameters and conclusions). To test the tactic, consider the proof

lemma

```
"if True ∧ P then P else True ∧ False ⇒
 (if True ∧ Q then True ∧ Q else P) → True ∧ (True ∧ Q)"
apply(tactic {* true1_tac @{context} 1 *})
```

where the tactic yields the goal state

goal (1 subgoal):

```
1. if P then P else True ∧ False ⇒ (if Q then Q else P) → Q
```

As you can see, the premises are rewritten according to `if_true1_conv`, while the conclusion according to `all_true1_conv`.

To sum up this section, conversions are more general than the simplifier or `simpprocs`, but you have to do more work yourself. Also conversions are often much less efficient than the simplifier. The advantage of conversions, however, that they provide much less room for non-termination.

Exercise 5.6.1. Write a tactic that does the same as the `simpproc` in exercise 5.5.1, but is based on conversions. You can make the same assumptions as in 5.5.1.

Exercise 5.6.2. Compare your solutions of Exercises 5.5.1 and 5.6.1, and try to determine which way of rewriting such terms is faster. For this you might have to construct quite large terms. Also see Recipe A.3 for information about timing.

Read More

See `Pure/conv.ML` for more information about conversion combinators. Some basic conversions are defined in `Pure/thm.ML`, `Pure/drule.ML` and `Pure/meta_simplifier.ML`.

(FIXME: check whether `Pattern.match_rew` and `Pattern.rewrite_term` are of any use/efficient)

5.7 Declarations (TBD)

5.8 Structured Proofs (TBD)

TBD

lemma True

proof

```
{
  fix A B C
  assume r: "A & B ⇒ C"
  assume A B
  then have "A & B" ..
  then have C by (rule r)
}
```

```
{
  fix A B C
  assume r: "A & B ⇒ C"
  assume A B
  note conjI [OF this]
```



```
    note r [OF this]
  }
oops
```

```
val ctxt0 = @{context};
val ctxt = ctxt0;
val (_, ctxt) = Variable.add_fixes ["A", "B", "C"] ctxt;
val ([r], ctxt) = Assumption.add_assumes [@"A & B ==> C"] ctxt
val (this, ctxt) = Assumption.add_assumes [@"A"}, @@"B"]
ctxt;
val this = [@"thm conjI"} OF this];
val this = r OF this;
val this = Assumption.export false ctxt ctxt0 this
val this = Variable.export ctxt ctxt0 [this]
```


Chapter 6

How to Write a Definitional Package

HOL is based on just a few primitive constants, like equality and implication, whose properties are described by axioms. All other concepts, such as inductive predicates, datatypes or recursive functions, are defined in terms of those primitives, and the desired properties, for example induction theorems or recursion equations, are derived from the definitions by a formal proof. Since it would be very tedious for a user to define inductive predicates or datatypes “by hand” just using the primitive operators of higher order logic, *definitional packages* have been implemented to automate such work. Thanks to those packages, the user can give a high-level specification, for example a list of introduction rules or constructors, and the package then does all the low-level definitions and proofs behind the scenes. In this chapter we explain how such a package can be implemented.

As the running example we have chosen a rather simple package for defining inductive predicates. To keep things really simple, we will not use the general Knaster-Tarski fixpoint theorem on complete lattices, which forms the basis of Isabelle/HOL’s standard inductive definition package. Instead, we will describe a simpler *impredicative* (i.e. involving quantification on predicate variables) encoding of inductive predicates. Due to its simplicity, this package will necessarily have a reduced functionality. It does neither support introduction rules involving arbitrary monotonic operators, nor does it prove case analysis rules (also called inversion rules). Moreover, it only proves a weaker form of the induction principle for inductive predicates. But it illustrates the implementation of a typical package in Isabelle.

6.1 Preliminaries

The user will just give a specification of inductive predicate(s) and expects from the package to produce a convenient reasoning infrastructure. This infrastructure needs to be derived from the definition that correspond to the specified predicate(s). Before we start with explaining all parts of the package, let us first give some examples showing how to define inductive predicates and then also how to generate a reason-

ing infrastructure for them. From the examples we will figure out a general method for defining inductive predicates. This is usually the first step in writing a package for Isabelle. The aim in this section is *not* to write proofs that are as beautiful as possible, but as close as possible to the ML-implementation we will develop in later sections.

We first consider the transitive closure of a relation R . The “pencil-and-paper” specification for the transitive closure is:

$$\frac{}{trcl\ R\ x\ x} \qquad \frac{R\ x\ y \quad trcl\ R\ y\ z}{trcl\ R\ x\ z}$$

As mentioned before, the package has to make an appropriate definition for $trcl$. Since an inductively defined predicate is the least predicate closed under a collection of introduction rules, the predicate $trcl\ R\ x\ y$ can be defined so that it holds if and only if $P\ x\ y$ holds for every predicate P closed under the rules above. This gives rise to the definition

definition `"trcl ≡`
 $\lambda R\ x\ y. \forall P. (\forall x. P\ x\ x)$
 $\longrightarrow (\forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z) \longrightarrow P\ x\ y"$

Note we have to use the object implication \longrightarrow and object quantification \forall for stating this definition (there is no other way for definitions in HOL). However, the introduction rules and induction principles associated with the transitive closure should use the meta-connectives, since they simplify the reasoning for the user.

With this definition, the proof of the induction principle for $trcl$ is almost immediate. It suffices to convert all the meta-level connectives in the lemma to object-level connectives using the proof method `atomize` (Line 4 below), expand the definition of $trcl$ (Line 5 and 6), eliminate the universal quantifier contained in it (Line 7), and then solve the goal by `assumption` (Line 8).

```
1 lemma trcl_induct:
2 assumes "trcl R x y"
3 shows "( $\bigwedge x. P\ x\ x$ )  $\implies$  ( $\bigwedge x\ y\ z. R\ x\ y \implies P\ y\ z \implies P\ x\ z$ )  $\implies P\ x\ y$ "
4 apply(atomize (full))
5 apply(cut_tac prems)
6 apply(unfold trcl_def)
7 apply(drule spec[where x=P])
8 apply(assumption)
9 done
```

The proofs for the introduction rules are slightly more complicated. For the first one, we need to prove the following lemma:

```
1 lemma trcl_base:
2 shows "trcl R x x"
3 apply(unfold trcl_def)
4 apply(rule allI impI)+
5 apply(drule spec)
6 apply(assumption)
7 done
```

We again unfold first the definition and apply introduction rules for \forall and \longrightarrow as often as possible (Lines 3 and 4). We then end up in the goal state:

goal (1 subgoal):

1. $\bigwedge P. \llbracket \forall x. P\ x\ x; \forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z \rrbracket \Longrightarrow P\ x\ x$

The two assumptions come from the definition of *trcl* and correspond to the introduction rules. Thus, all we have to do is to eliminate the universal quantifier in front of the first assumption (Line 5), and then solve the goal by *assumption* (Line 6).

Next we have to show that the second introduction rule also follows from the definition. Since this rule has premises, the proof is a bit more involved. After unfolding the definitions and applying the introduction rules for \forall and \longrightarrow

lemma *trcl_step*:

shows " $R\ x\ y \Longrightarrow trcl\ R\ y\ z \Longrightarrow trcl\ R\ x\ z$ "

apply (*unfold trcl_def*)

apply (*rule allI impI*)⁺

we obtain the goal state

goal (1 subgoal):

1. $\bigwedge P. \llbracket R\ x\ y; \forall P. (\forall x. P\ x\ x) \longrightarrow (\forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z) \longrightarrow P\ y\ z; \forall x. P\ x\ x; \forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z \rrbracket \Longrightarrow P\ x\ z$

To see better where we are, let us explicitly name the assumptions by starting a subproof.

proof -

case (*goal1 P*)

have *p1*: " $R\ x\ y$ " **by fact**

have *p2*: " $\forall P. (\forall x. P\ x\ x)$

$\longrightarrow (\forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z) \longrightarrow P\ y\ z$ " **by fact**

have *r1*: " $\forall x. P\ x\ x$ " **by fact**

have *r2*: " $\forall x\ y\ z. R\ x\ y \longrightarrow P\ y\ z \longrightarrow P\ x\ z$ " **by fact**

show " $P\ x\ z$ "

The assumptions *p1* and *p2* correspond to the premises of the second introduction rule (unfolded); the assumptions *r1* and *r2* come from the definition of *trcl*. We apply *r2* to the goal $P\ x\ z$. In order for this assumption to be applicable as a rule, we have to eliminate the universal quantifier and turn the object-level implications into meta-level ones. This can be accomplished using the *rule_format* attribute. So we continue the proof with:

apply (*rule r2[rule_format]*)

This gives us two new subgoals

goal (2 subgoals):

1. $R\ x\ ?y$
2. $P\ ?y\ z$

which can be solved using assumptions $p1$ and $p2$. The latter involves a quantifier and implications that have to be eliminated before it can be applied. To avoid potential problems with higher-order unification, we explicitly instantiate the quantifier to P and also match explicitly the implications with $r1$ and $r2$. This gives the proof:

```

apply(rule p1)
apply(rule p2[THEN spec[where x=P], THEN mp, THEN mp, OF r1, OF r2])
done
qed

```

Now we are done. It might be surprising that we are not using the automatic tactics available in Isabelle/HOL for proving this lemmas. After all *blast* would easily dispense of it.

```

lemma trcl_step_blast:
shows "R x y  $\implies$  trcl R y z  $\implies$  trcl R x z"
apply(unfold trcl_def)
apply(blast)
done

```

Experience has shown that it is generally a bad idea to rely heavily on *blast*, *auto* and the like in automated proofs. The reason is that you do not have precise control over them (the user can, for example, declare new intro- or simplification rules that can throw automatic tactics off course) and also it is very hard to debug proofs involving automatic tactics whenever something goes wrong. Therefore if possible, automatic tactics in packages should be avoided or be constrained sufficiently.

The method of defining inductive predicates by impredicative quantification also generalises to mutually inductive predicates. The next example defines the predicates *even* and *odd* given by

$$\frac{}{\text{even } 0} \quad \frac{\text{odd } n}{\text{even } (\text{Suc } n)} \quad \frac{\text{even } n}{\text{odd } (\text{Suc } n)}$$

Since the predicates *even* and *odd* are mutually inductive, each corresponding definition must quantify over both predicates (we name them below P and Q).

```

definition "even  $\equiv$ 
 $\lambda n. \forall P Q. P\ 0 \longrightarrow (\forall m. Q\ m \longrightarrow P\ (\text{Suc } m))$ 
 $\longrightarrow (\forall m. P\ m \longrightarrow Q\ (\text{Suc } m)) \longrightarrow P\ n"$ 

```

```

definition "odd  $\equiv$ 
 $\lambda n. \forall P Q. P\ 0 \longrightarrow (\forall m. Q\ m \longrightarrow P\ (\text{Suc } m))$ 
 $\longrightarrow (\forall m. P\ m \longrightarrow Q\ (\text{Suc } m)) \longrightarrow Q\ n"$ 

```

For proving the induction principles, we use exactly the same technique as in the transitive closure example, namely:

```

lemma even_induct:
assumes "even n"
shows "P 0  $\implies$  ( $\bigwedge m. Q\ m \implies P\ (\text{Suc } m)$ )  $\implies$  ( $\bigwedge m. P\ m \implies Q\ (\text{Suc } m)$ )  $\implies$  P n"
apply(atomize (full))
apply(cut_tac prems)
apply(unfold even_def)

```

```

apply(drule spec[where  $x=P$ ])
apply(drule spec[where  $x=Q$ ])
apply(assumption)
done

```

The only difference with the proof *trcl_induct* is that we have to instantiate here two universal quantifiers. We omit the other induction principle that has *even n* as premise and *Q n* as conclusion. The proofs of the introduction rules are also very similar to the ones in the *trcl*-example. We only show the proof of the second introduction rule.

```

1 lemma evenS:
2 shows " $odd\ m \implies even\ (Suc\ m)$ "
3 apply (unfold odd_def even_def)
4 apply (rule allI impI)+
5 proof -
6   case (goal1 P Q)
7   have p1: " $\forall P\ Q.\ P\ 0 \longrightarrow (\forall m.\ Q\ m \longrightarrow P\ (Suc\ m))$ "
8           " $\longrightarrow (\forall m.\ P\ m \longrightarrow Q\ (Suc\ m)) \longrightarrow Q\ m$ " by fact
9   have r1: " $P\ 0$ " by fact
10  have r2: " $\forall m.\ Q\ m \longrightarrow P\ (Suc\ m)$ " by fact
11  have r3: " $\forall m.\ P\ m \longrightarrow Q\ (Suc\ m)$ " by fact
12  show " $P\ (Suc\ m)$ "
13    apply(rule r2[rule_format])
14    apply(rule p1[THEN spec[where  $x=P$ ], THEN spec[where  $x=Q$ ],
15            THEN mp, THEN mp, THEN mp, OF r1, OF r2, OF r3])
16  done
17 qed

```

The interesting lines are 7 to 15. Again, the assumptions fall into two categories: *p1* corresponds to the premise of the introduction rule; *r1* to *r3* come from the definition of *even*. In Line 13, we apply the assumption *r2* (since we prove the second introduction rule). In Lines 14 and 15 we apply assumption *p1* (if the second introduction rule had more premises we have to do that for all of them). In order for this assumption to be applicable, the quantifiers need to be instantiated and then also the implications need to be resolved with the other rules.

Next we define the accessible part of a relation *R* given by the single rule:

$$\frac{\bigwedge y. R\ y\ x \implies accpart\ R\ y}{accpart\ R\ x}$$

The interesting point of this definition is that it contains a quantification that ranges only over the premise and the premise has also a precondition. The definition of *accpart* is:

definition "*accpart* $\equiv \lambda R\ x.\ \forall P.\ (\forall x.\ (\forall y. R\ y\ x \longrightarrow P\ y) \longrightarrow P\ x) \longrightarrow P\ x$ "

The proof of the induction principle is again straightforward and omitted. Proving the introduction rule is a little more complicated, because the quantifier and the implication in the premise. The proof is as follows.

```

1 lemma accpartI:
2 shows " $(\bigwedge y. R\ y\ x \implies accpart\ R\ y) \implies accpart\ R\ x$ "

```

```

3 apply (unfold accpart_def)
4 apply (rule allI impI)+
5 proof -
6   case (goal1 P)
7   have p1: " $\bigwedge y. R y x \implies$ 
8              $(\forall P. (\forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x) \longrightarrow P y)$ " by fact
9   have r1: " $\forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x$ " by fact
10  show "P x"
11    apply(rule r1[rule_format])
12    proof -
13      case (goal1 y)
14      have r1_prem: "R y x" by fact
15      show "P y"
16        apply(rule p1[OF r1_prem, THEN spec[where x=P], THEN mp, OF r1])
17      done
18    qed
19  qed

```

As you can see, there are now two subproofs. The assumptions fall as usual into two categories (Lines 7 to 9). In Line 11, applying the assumption $r1$ generates a goal state with the new local assumption $R y x$, named $r1_prem$ in the second subproof (Line 14). This local assumption is used to solve the goal $P y$ with the help of assumption $p1$.

Exercise 6.1.1. Give the definition for the freshness predicate for lambda-terms. The rules for this predicate are:

$$\begin{array}{c}
 \frac{a \neq b}{\text{fresh } a \text{ (Var } b)} \qquad \frac{\text{fresh } a \ t \quad \text{fresh } a \ s}{\text{fresh } a \ (\text{App } t \ s)} \\
 \frac{}{\text{fresh } a \ (\text{Lam } a \ t)} \qquad \frac{a \neq b \quad \text{fresh } a \ t}{\text{fresh } a \ (\text{Lam } b \ t)}
 \end{array}$$

From the definition derive the induction principle and the introduction rules.

The point of all these examples is to get a feeling what the automatic proofs should do in order to solve all inductive definitions we throw at them. This is usually the first step in writing a package. We next explain the parsing and typing part of the package.

6.2 Parsing and Typing the Specification

To be able to write down the specifications of inductive predicates, we have to introduce a new command (see Section 4.8). As the keyword for the new command we chose **simple inductive**. Examples of specifications from the previous section are shown in Figure 6.1. The syntax used in these examples more or less translates directly into the parser:


```

simple inductive
  trcl :: "('a ⇒ 'a ⇒ bool) ⇒ 'a ⇒ 'a ⇒ bool"
where
  base: "trcl R x x"
  | step: "trcl R x y ⇒ R y z ⇒ trcl R x z"

simple inductive
  even and odd
where
  even0: "even 0"
  | evenS: "odd n ⇒ even (Suc n)"
  | oddS: "even n ⇒ odd (Suc n)"

simple inductive
  accpart :: "('a ⇒ 'a ⇒ bool) ⇒ 'a ⇒ bool"
where
  accpartI: "(λy. R y x ⇒ accpart R y) ⇒ accpart R x"

simple inductive
  fresh :: "string ⇒ trm ⇒ bool"
where
  fresh_var: "a≠b ⇒ fresh a (Var b)"
  | fresh_app: "[fresh a t; fresh a s] ⇒ fresh a (App t s)"
  | fresh_lam1: "fresh a (Lam a t)"
  | fresh_lam2: "[a≠b; fresh a t] ⇒ fresh a (Lam b t)"

```

Figure 6.1: Specification given by the user for the inductive predicates *trcl*, *even* and *odd*, *accpart* and *fresh*.

```

val spec_parser =
  OuterParse.fixes --
  Scan.optional
    (OuterParse.$$$ "where" |--
     OuterParse.!!!
     (OuterParse.enum1 "|"
      (SpecParse.opt_thm_name ":" -- OuterParse.prop))) []

```

which we explained in Section 4.7. There is no code included for parsing the keyword and what is called a *target*. The latter can be given optionally after the keyword. The target is an “advanced” feature which we will inherit for “free” from the infrastructure on which we shall build the package. The target stands for a locale and allows us to specify

```

locale rel =
  fixes R :: "'a ⇒ 'a ⇒ bool"

```

and then define the transitive closure and the accessible part of this locale as follows:

```

simple inductive (in rel)
  trcl'
where

```

```

base: "trcl' x x"
| step: "trcl' x y  $\implies$  R y z  $\implies$  trcl' x z"

```

simple_inductive (in *rel*)

accpart'

where

accpartI: " $(\wedge y. R y x \implies \text{accpart}' y) \implies \text{accpart}' x$ "

Note that in these definitions the parameter R , standing for the relation, is left implicit. For the moment we will ignore this kind of implicit parameters and rely on the fact that the infrastructure will deal with them. Later, however, we will come back to them.

If we feed into the parser the string that corresponds to our definition of *even* and *odd*

```

let
  val input = filtered_input
    ("even and odd " ^
     "where " ^
     "  even0[intro]: \"even 0\" " ^
     "| evenS[intro]: \"odd n  $\implies$  even (Suc n)\" " ^
     "| oddS[intro]: \"even n  $\implies$  odd (Suc n)\" ")
in
  parse spec_parser input
end
> ((([even, NONE, NoSyn), (odd, NONE, NoSyn)],
>    [(even0,...), "\^E\^Ftoken\^Even 0\^E\^F\^E"),
>    (evenS,...), "\^E\^Ftoken\^Eodd n  $\implies$  even (Suc n)\^E\^F\^E"),
>    ((oddS,...), "\^E\^Ftoken\^Even n  $\implies$  odd (Suc n)\^E\^F\^E"))], [])

```

then we get back the specifications of the predicates (with type and syntax annotations), and specifications of the introduction rules. This is all the information we need for calling the package and setting up the keyword. The latter is done in Lines 5 to 7 in the code below.

```

1 val specification =
2   spec_parser >>
3     (fn (pred_specs, rule_specs) => add_inductive_cmd pred_specs rule_specs)
4
5 val _ = OuterSyntax.local_theory "simple_inductive2"
6       "definition of simple inductive predicates"
7       OuterKeyword.thy_decl specification

```

We call *local_theory* with the kind-indicator *thy_decl* since the package does not need to open up any proof (see Section 4.8). The auxiliary function *specification* in Lines 1 to 3 gathers the information from the parser to be processed further by the function *add_inductive_cmd*, which we describe below.

Note that the predicates when they come out of the parser are just some “naked” strings: they have no type yet (even if we annotate them with types) and they are

also not defined constants yet (which the predicates eventually will be). Also the introduction rules are just strings. What we have to do first is to transform the parser's output into some internal datastructures that can be processed further. For this we can use the function `read_spec`. This function takes some strings (with possible typing annotations) and some rule specifications, and attempts to find a typing according to the given type constraints given by the user and the type constraints by the "ambient" theory. It returns the type for the predicates and also returns typed terms for the introduction rules. So at the heart of the function `add_inductive_cmd` is a call to `read_spec`.

```
fun add_inductive_cmd pred_specs rule_specs lthy =
let
  val ((pred_specs', rule_specs'), _) =
    Specification.read_spec pred_specs rule_specs lthy
in
  add_inductive pred_specs' rule_specs' lthy
end
```

Once we have the input data as some internal datastructure, we call the function `add_inductive`. This function does the heavy duty lifting in the package: it generates definitions for the predicates and derives from them corresponding induction principles and introduction rules. The description of this function will span over the next two sections.

6.3 The Code in a Nutshell

The inductive package will generate the reasoning infrastructure for mutually recursive predicates, say $pred_1 \dots pred_n$. In what follows we will have the convention that various, possibly empty collections of "things" (lists, terms, nested implications and so on) are indicated either by adding an "s" or by adding a superscript "*". The shorthand for the predicates will therefore be $preds$ or $pred^*$. In the case of the predicates there must be, of course, at least a single one in order to obtain a meaningful definition.

The input for the inductive package will be some $preds$ with possible typing and syntax annotations, and also some introduction rules. We call below the introduction rules short as $rules$. Borrowing some idealised Isabelle notation, one such $rule$ is assumed to be of the form

$$rule ::= \bigwedge xs. \underbrace{As}_{\text{non-recursive premises}} \implies \underbrace{(\bigwedge ys. Bs \implies pred\ ss)^*}_{\text{recursive premises}} \implies pred\ ts$$

For the purposes here, we will assume the $rules$ have this format and omit any code that actually tests this. Therefore "things" can go horribly wrong, if the $rules$ are not of this form. The As and Bs in a $rule$ stand for formulae not involving the inductive predicates $preds$; the instances $pred\ ss$ and $pred\ ts$ can stand for different predicates, like $pred_1\ ss$ and $pred_2\ ts$, in case mutual recursive predicates are defined; the terms ss and ts are the arguments of these predicates. Every formula

left of " $\implies pred\ ts$ " is a premise of the rule. The outermost quantified variables xs are usually omitted in the user's input. The quantification for the variables ys is local with respect to one recursive premise and must be given. Some examples of *rules* are

$$a \neq b \implies fresh\ a\ (Var\ b)$$

which has only a single non-recursive premise, whereas

$$odd\ n \implies even\ (Suc\ n)$$

has a single recursive premise; the rule

$$(\bigwedge y. R\ y\ x \implies accpart\ R\ y) \implies accpart\ R\ x$$

has a single recursive premise that has a precondition. As is custom all rules are stated without the leading meta-quantification $\bigwedge xs$.

The output of the inductive package will be definitions for the predicates, induction principles and introduction rules. For the definitions we need to have the *rules* in a form where the meta-quantifiers and meta-implications are replaced by their object logic equivalents. Therefore an *orule* is of the form

$$orule ::= \forall xs. As \longrightarrow (\forall ys. Bs \longrightarrow pred\ ss)^* \longrightarrow pred\ ts$$

A definition for the predicate *pred* has then the form

$$def ::= pred \equiv \lambda zs. \forall preds. orules \longrightarrow pred\ zs$$

The induction principles for every predicate *pred* are of the form

$$ind ::= pred\ ?zs \implies rules[preds := ?Ps] \implies ?P\ ?zs$$

where in the *rules*-part every *pred* is replaced by a fresh meta-variable *?P*.

In order to derive an induction principle for the predicate *pred*, we first transform *ind* into the object logic and fix the meta-variables. Hence we have to prove a formula of the form

$$pred\ zs \longrightarrow orules[preds := Ps] \longrightarrow P\ zs$$

If we assume *pred zs* and unfold its definition, then we have an assumption

$$\forall preds. orules \longrightarrow pred\ zs$$

and must prove the goal

$$orules[preds := Ps] \longrightarrow P\ zs$$

This can be done by instantiating the \forall *preds*-quantification with the *Ps*. Then we are done since we are left with a simple identity.

Although the user declares the introduction rules *rules*, they must also be derived from the *defs*. These derivations are a bit involved. Assuming we want to prove the introduction rule

$$\bigwedge_{xs}. As \implies (\bigwedge_{ys}. Bs \implies \text{pred } ss)^* \implies \text{pred } ts$$

then we have assumptions of the form

- (i) *As*
- (ii) $(\bigwedge_{ys}. Bs \implies \text{pred } ss)^*$

and must show the goal

pred ts

If we now unfold the definitions for the *preds*, we have assumptions

- (i) *As*
- (ii) $(\bigwedge_{ys}. Bs \implies \forall \text{preds}. \text{orules} \longrightarrow \text{pred } ss)^*$
- (iii) *orules*

and need to show

pred ts

In the last step we removed some quantifiers and moved the precondition *orules* into the assumption. The *orules* stand for all introduction rules that are given by the user. We apply the *orule* that corresponds to introduction rule we are proving. After transforming the object connectives into meta-connectives, this introduction rule must necessarily be of the form

$$As \implies (\bigwedge_{ys}. Bs \implies \text{pred } ss)^* \implies \text{pred } ts$$

When we apply this rule we end up in the goal state where we have to prove goals of the form

- (a) *As*
- (b) $(\bigwedge_{ys}. Bs \implies \text{pred } ss)^*$

We can immediately discharge the goals *As* using the assumptions in (i). The goals in (b) can be discharged as follows: we assume the *Bs* and prove *pred ss*. For this we resolve the *Bs* with the assumptions in (ii). This gives us the assumptions

$$(\forall \text{preds}. \text{orules} \longrightarrow \text{pred } ss)^*$$

```

(* even-odd example *)
val eo_defs = [@{thm even_def}, @{thm odd_def}]

val eo_rules =
  [@{prop "even 0"},
   @{prop " $\wedge n. \text{odd } n \implies \text{even } (\text{Suc } n)$ "},
   @{prop " $\wedge n. \text{even } n \implies \text{odd } (\text{Suc } n)$ "}]

val eo_orules =
  [@{prop "even 0"},
   @{prop " $\forall n. \text{odd } n \longrightarrow \text{even } (\text{Suc } n)$ "},
   @{prop " $\forall n. \text{even } n \longrightarrow \text{odd } (\text{Suc } n)$ "}]

val eo_preds = [@{term "even::nat  $\Rightarrow$  bool"}, @{term "odd::nat  $\Rightarrow$  bool"}]
val eo_prednames = [@{binding "even"}, @{binding "odd"}]
val eo_mxs = [NoSyn, NoSyn]
val eo_arg_tyss = [[@{typ "nat"}], [@{typ "nat"}]]
val e_pred = @{term "even::nat  $\Rightarrow$  bool"}
val e_arg_tys = [@{typ "nat"}]

(* freshness example *)
val fresh_rules =
  [@{prop " $\wedge a b. a \neq b \implies \text{fresh } a \text{ (Var } b)$ "},
   @{prop " $\wedge a s t. \text{fresh } a t \implies \text{fresh } a s \implies \text{fresh } a \text{ (App } t s)$ "},
   @{prop " $\wedge a t. \text{fresh } a \text{ (Lam } a t)$ "},
   @{prop " $\wedge a b t. a \neq b \implies \text{fresh } a t \implies \text{fresh } a \text{ (Lam } b t)$ "}]

val fresh_orules =
  [@{prop " $\forall a b. a \neq b \longrightarrow \text{fresh } a \text{ (Var } b)$ "},
   @{prop " $\forall a s t. \text{fresh } a t \longrightarrow \text{fresh } a s \longrightarrow \text{fresh } a \text{ (App } t s)$ "},
   @{prop " $\forall a t. \text{fresh } a \text{ (Lam } a t)$ "},
   @{prop " $\forall a b t. a \neq b \longrightarrow \text{fresh } a t \longrightarrow \text{fresh } a \text{ (Lam } b t)$ "}]

val fresh_pred = @{term "fresh::string  $\Rightarrow$  trm  $\Rightarrow$  bool"}
val fresh_arg_tys = [@{typ "string"}, @{typ "trm"}]

(* accessible-part example *)
val acc_rules =
  [@{prop " $\wedge R x. (\wedge y. R y x \implies \text{accpart } R y) \implies \text{accpart } R x$ "}]
val acc_pred = @{term "accpart::('a  $\Rightarrow$  'a  $\Rightarrow$  bool)  $\Rightarrow$  'a  $\Rightarrow$  bool"}

```

Figure 6.2: Shorthands for the inductive predicates *even/odd*, *fresh* and *accpart*. The names of these shorthands follow the convention *rules*, *orules*, *preds* and so on. The purpose of these shorthands is to simplify the construction of testcases in Section 6.4.

Instantiating the universal quantifiers and then resolving with the assumptions in (iii) gives us $pred\ ss$, which is the goal we are after. This completes the proof for introduction rules.

What remains is to implement in Isabelle the reasoning outlined in this section. We will describe the code in the next section. For building testcases, we use the shorthands for *even/odd*, *fresh* and *accpart* defined in Figure 6.2.

6.4 The Gory Details

As mentioned before the code falls roughly into three parts: the code that deals with the definitions, with the induction principles and with the introduction rules. In addition there are some administrative functions that string everything together.

Definitions

We first have to produce for each predicate the user specifies an appropriate definition, whose general form is

$$pred \equiv \lambda zs. \forall preds. orules \longrightarrow pred\ zs$$

and then “register” the definition inside a local theory. To do the latter, we use the following wrapper for the function *define*. The wrapper takes a predicate name, a syntax annotation and a term representing the right-hand side of the definition.

```

1 fun make_defn ((predname, mx), trm) lthy =
2   let
3     val arg = ((predname, mx), (Attrib.empty_binding, trm))
4     val ((_, (_ , thm)), lthy') = LocalTheory.define Thm.internalK arg lthy
5   in
6     (thm, lthy')
7   end

```

It returns the definition (as a theorem) and the local theory in which the definition has been made. In Line 4, *internalK* is a flag attached to the theorem (other possible flags are *definitionK* and *axiomK*). These flags just classify theorems and have no significant meaning, except for tools that, for example, find theorems in the theorem database.¹ We also use *empty_binding* in Line 3, since the definitions for our inductive predicates are not meant to be seen by the user and therefore do not need to have any theorem attributes. A testcase for this function is

```

local_setup {* fn lthy =>
let
  val arg = (@{binding "My_True"}, NoSyn), @{term True}

```

¹FIXME: put in the section about theorems.

```

    val (def, lthy') = make_defn arg lthy
  in
    tracing (string_of_thm_no_vars lthy' def); lthy'
  end *}

```

which introduces the definition $My_True \equiv True$ and then prints it out. Since we are testing the function inside **local_setup**, i.e., make actual changes to the ambient theory, we can query the definition with the usual command **thm**:

```

thm "My_True_def"
> My_True  $\equiv$  True

```

The next two functions construct the right-hand sides of the definitions, which are terms whose general form is:

$$\lambda zs. \forall preds. orules \longrightarrow pred\ zs$$

When constructing these terms, the variables zs need to be chosen so that they do not occur in the $orules$ and also be distinct from the $preds$.

The first function, named $defn_aux$, constructs the term for one particular predicate (the argument $pred$ in the code below). The number of arguments of this predicate is determined by the number of argument types given in arg_tys . The other arguments of the function are the $orules$ and all the $preds$.

```

1 fun defn_aux lthy orules preds (pred, arg_tys) =
2   let
3     fun mk_all x P = HOLogic.all_const (fastype_of x) $ lambda x P
4
5     val fresh_args =
6       arg_tys
7       |> map (pair "z")
8       |> Variable.variant_frees lthy (preds @ orules)
9       |> map Free
10  in
11    list_comb (pred, fresh_args)
12    |> fold_rev (curry HOLogic.mk_imp) orules
13    |> fold_rev mk_all preds
14    |> fold_rev lambda fresh_args
15  end

```

The function mk_all in Line 3 is just a helper function for constructing universal quantifications. The code in Lines 5 to 9 produces the fresh zs . For this it pairs every argument type with the string "z" (Line 7); then generates variants for all these strings so that they are unique w.r.t. to the predicates and $orules$ (Line 8); in Line 9 it generates the corresponding variable terms for the unique strings.

The unique variables are applied to the predicate in Line 11 using the function $list_comb$; then the $orules$ are prefixed (Line 12); in Line 13 we quantify over all predicates; and in line 14 we just abstract over all the zs , i.e., the fresh arguments of the predicate. A testcase for this function is


```

local_setup {* fn lthy =>
let
  val def = defn_aux lthy eo_orules eo_preds (e_pred, e_arg_tys)
in
  tracing (Syntax.string_of_term lthy def); lthy
end *}

```

where we use the shorthands defined in Figure 6.2. The testcase calls *defn_aux* for the predicate *even* and prints out the generated definition. So we obtain as printout

$$\lambda z. \forall \text{even odd. } (\text{even } 0) \longrightarrow (\forall n. \text{odd } n \longrightarrow \text{even } (\text{Suc } n)) \\ \longrightarrow (\forall n. \text{even } n \longrightarrow \text{odd } (\text{Suc } n)) \longrightarrow \text{even } z$$

If we try out the function with the rules for freshness

```

local_setup {* fn lthy =>
let
  val def = defn_aux lthy fresh_orules
                                [fresh_pred] (fresh_pred, fresh_arg_tys)
in
  tracing (Syntax.string_of_term lthy def); lthy
end *}

```

we obtain

$$\lambda z \text{ za.} \\ \forall \text{fresh.} \\ (\forall a \text{ b. } a \neq b \longrightarrow \text{fresh } a \text{ (Var } b)) \longrightarrow \\ (\forall a \text{ s t. } \text{fresh } a \text{ t} \longrightarrow \text{fresh } a \text{ s} \longrightarrow \text{fresh } a \text{ (App t s)}) \longrightarrow \\ (\forall a \text{ t. } \text{fresh } a \text{ (Lam a t)}) \longrightarrow \\ (\forall a \text{ b t. } a \neq b \longrightarrow \text{fresh } a \text{ t} \longrightarrow \text{fresh } a \text{ (Lam b t)}) \longrightarrow \text{fresh } z \text{ za}$$

The second function, named *defns*, has to iterate the function *defn_aux* over all predicates. The argument *preds* is again the list of predicates as *terms*; the argument *prednames* is the list of binding names of the predicates; *mxs* are the list of syntax, or mixfix, annotations for the predicates; *arg_tysss* is the list of argument-type-lists.

```

1 fun defns rules preds prednames mxs arg_tysss lthy =
2 let
3   val thy = ProofContext.theory_of lthy
4   val orules = map (ObjectLogic.atomize_term thy) rules
5   val defs = map (defn_aux lthy orules preds) (preds ~~ arg_tysss)
6 in
7   fold_map make_defn (prednames ~~ mxs ~~ defs) lthy
8 end

```

The user will state the introduction rules using meta-implications and meta-quantifications. In Line 4, we transform these introduction rules into the object logic (since

definitions cannot be stated with meta-connectives). To do this transformation we have to obtain the theory behind the local theory using the function *theory_of* (Line 3); with this theory we can use the function *atomize_term* to make the transformation (Line 4). The call to *defn_aux* in Line 5 produces all right-hand sides of the definitions. The actual definitions are then made in Line 7. The result of the function is a list of theorems and a local theory (the theorems are registered with the local theory). A testcase for this function is

```
local_setup {* fn lthy =>
let
  val (defs, lthy') =
    defns eo_rules eo_preds eo_prednames eo_mxs eo_arg_tyss lthy
in
  tracing (string_of_thms_no_vars lthy' defs); lthy
end *}
```

where we feed into the function all parameters corresponding to the *even/odd* example. The definitions we obtain are:

$$\begin{aligned} \text{even} &\equiv \lambda z. \forall \text{even odd. } (\text{even } 0) \longrightarrow (\forall n. \text{odd } n \longrightarrow \text{even } (\text{Suc } n)) \\ &\quad \longrightarrow (\forall n. \text{even } n \longrightarrow \text{odd } (\text{Suc } n)) \longrightarrow \text{even } z, \\ \text{odd} &\equiv \lambda z. \forall \text{even odd. } (\text{even } 0) \longrightarrow (\forall n. \text{odd } n \longrightarrow \text{even } (\text{Suc } n)) \\ &\quad \longrightarrow (\forall n. \text{even } n \longrightarrow \text{odd } (\text{Suc } n)) \longrightarrow \text{odd } z \end{aligned}$$

Note that in the testcase we return the local theory *lthy* (not the modified *lthy'*). As a result the test case has no effect on the ambient theory. The reason is that if we introduce the definition again, we pollute the name space with two versions of *even* and *odd*. We want to avoid this here.

This completes the code for introducing the definitions. Next we deal with the induction principles.

Induction Principles

Recall that the manual proof for the induction principle of *even* was:

```
lemma manual_ind_prin_even:
assumes prem: "even z"
shows "P 0  $\implies$  ( $\bigwedge m. Q m \implies P (\text{Suc } m)$ )  $\implies$  ( $\bigwedge m. P m \implies Q (\text{Suc } m)$ )  $\implies P z$ "
apply(atomize (full))
apply(cut_tac prem)
apply(unfold even_def)
apply(drule spec[where x=P])
apply(drule spec[where x=Q])
apply(assumption)
done
```

The code for automating such induction principles has to accomplish two tasks: constructing the induction principles from the given introduction rules and then automatically generating proofs for them using a tactic.

The tactic will use the following helper function for instantiating universal quantifiers.

```
fun inst_spec cterm =
  Drule.instantiate' [SOME (ctyp_of_term cterm)] [NONE, SOME cterm]
    @{thm spec}
```

This helper function uses the function *instantiate'* and instantiates the $?x$ in the theorem $\forall x. ?P x \implies ?P ?x$ with a given *cterm*. We call this helper function in the following tactic..

```
fun inst_spec_tac ctrms =
  EVERY' (map (dtac o inst_spec) ctrms)
```

This tactic expects a list of *cterm*s. It allows us in the proof below to instantiate the three quantifiers in the assumption.

lemma

fixes $P::\text{"nat} \Rightarrow \text{nat} \Rightarrow \text{nat} \Rightarrow \text{bool}"$

shows $\forall x y z. P x y z \implies \text{True}$

apply (tactic {*

$\text{inst_spec_tac } [@\{\text{cterm } \text{"a::nat"}\}, @\{\text{cterm } \text{"b::nat"}\}, @\{\text{cterm } \text{"c::nat"}\}] 1 \text{ *)}$

We obtain the goal state

goal (1 subgoal):

1. $P a b c \implies \text{True}$

The complete tactic for proving the induction principles can now be implemented as follows:

```
1 fun ind_tac defs prem insts =
2   EVERY1 [ObjectLogic.full_atomize_tac,
3           cut_facts_tac prem,
4           K (rewrite_goals_tac defs),
5           inst_spec_tac insts,
6           assume_tac]
```

We have to give it as arguments the definitions, the premise (a list of formulae) and the instantiations. The premise is *even n* in lemma *manual_ind_prin_even* shown above; in our code it will always be a list consisting of a single formula. Compare this tactic with the manual proof for the lemma *manual_ind_prin_even*: as you can see there is almost a one-to-one correspondence between the **apply**-script and the *ind_tac*. We first rewrite the goal to use only object connectives (Line 2), "cut in" the premise (Line 3), unfold the definitions (Line 4), instantiate the assumptions of the goal (Line 5) and then conclude with *assume_tac*.

Two testcases for this tactic are:

lemma *automatic_ind_prin_even*:

assumes *prem*: $\text{"even } z"$

shows $P 0 \implies (\bigwedge m. Q m \implies P (\text{Suc } m)) \implies (\bigwedge m. P m \implies Q (\text{Suc } m)) \implies P z$

```

by (tactic {* ind_tac eo_defs @ {thms prem}
      [ @ {cterm "P::nat⇒bool"}, @ {cterm "Q::nat⇒bool"} ] *})

lemma automatic_ind_prin_fresh:
assumes prem: "fresh z za"
shows "( $\bigwedge a b. a \neq b \implies P a (\text{Var } b)$ )  $\implies$ 
      ( $\bigwedge a t s. \llbracket P a t; P a s \rrbracket \implies P a (\text{App } t s)$ )  $\implies$ 
      ( $\bigwedge a t. P a (\text{Lam } a t)$ )  $\implies$ 
      ( $\bigwedge a b t. \llbracket a \neq b; P a t \rrbracket \implies P a (\text{Lam } b t)$ )  $\implies P z za$ "
by (tactic {* ind_tac @ {thms fresh_def} @ {thms prem}
      [ @ {cterm "P::string⇒trm⇒bool"} ] *})

```

While the tactic for proving the induction principles is relatively simple, it will be a bit more work to construct the goals from the introduction rules the user provides. Therefore let us have a closer look at the first proved theorem:

```

thm automatic_ind_prin_even
>  $\llbracket \text{even } ?z; ?P 0; \bigwedge m. ?Q m \implies ?P (\text{Suc } m); \bigwedge m. ?P m \implies ?Q (\text{Suc } m) \rrbracket \implies ?P ?z$ 

```

The variables z , P and Q are schematic variables (since they are not quantified in the lemma). These variables must be schematic, otherwise they cannot be instantiated by the user. To generate these schematic variables we use a common trick in Isabelle programming: we first declare them as *free, but fixed*, and then use the infrastructure to turn them into schematic variables.

In general we have to construct for each predicate $pred$ a goal of the form

```

pred ?zs  $\implies$  rules[preds := ?Ps]  $\implies$  ?P ?zs

```

where the predicates $preds$ are replaced in $rules$ by new distinct variables $?Ps$. We also need to generate fresh arguments $?zs$ for the predicate $pred$ and the $?P$ in the conclusion.

We generate these goals in two steps. The first function, named `prove_ind`, expects that the introduction rules are already appropriately substituted. The argument `srules` stands for these substituted rules; `cnewpreds` are the certified terms corresponding to the variables $?Ps$; `pred` is the predicate for which we prove the induction principle; `newpred` is its replacement and `arg_tys` are the argument types of this predicate.

```

1 fun prove_ind lthy defs srules cnewpreds ((pred, newpred), arg_tys) =
2   let
3     val zs = replicate (length arg_tys) "z"
4     val (newargnames, lthy') = Variable.variant_fixes zs lthy;
5     val newargs = map Free (newargnames ~~ arg_tys)
6
7     val prem = HOLogic.mk_Trueprop (list_comb (pred, newargs))
8     val goal = Logic.list_implies
9       (srules, HOLogic.mk_Trueprop (list_comb (newpred, newargs)))
10  in
11    Goal.prove lthy' [] [prem] goal

```

```

12   (fn {prems, ...} => ind_tac defs prems cnewpreds)
13   |> singleton (ProofContext.export lthy' lthy)
14 end

```

In Line 3 we produce names *zs* for each type in the argument type list. Line 4 makes these names unique and declares them as free, but fixed, variables in the local theory *lthy'*. That means they are not schematic variables (yet). In Line 5 we construct the terms corresponding to these variables. The variables are applied to the predicate in Line 7 (this corresponds to the first premise *pred zs* of the induction principle). In Line 8 and 9, we first construct the term *P zs* and then add the (substituted) introduction rules as preconditions. In case that no introduction rules are given, the conclusion of this implication needs to be wrapped inside a *Trueprop*, otherwise the Isabelle's goal mechanism will fail.²

In Line 11 we set up the goal to be proved using the function *prove*; in the next line we call the tactic for proving the induction principle. As mentioned before, this tactic expects the definitions, the premise and the (certified) predicates with which the introduction rules have been substituted. The code in these two lines will return a theorem. However, it is a theorem proved inside the local theory *lthy'*, where the variables *zs* are free, but fixed (see Line 4). By exporting this theorem from *lthy'* (which contains the *zs* as free variables) to *lthy* (which does not), we obtain the desired schematic variables *?zs*. A testcase for this function is

```

local setup {* fn lthy =>
let
  val newpreds = [@{term "P::nat => bool"}, @{term "Q::nat => bool"}]
  val cnewpreds = [@{cterm "P::nat => bool"}, @{cterm "Q::nat => bool"}]
  val newpred = @{term "P::nat => bool"}
  val srules = map (subst_free (eo_preds ~~ newpreds)) eo_rules
  val intro =
    prove_ind lthy eo_defs srules cnewpreds ((e_pred, newpred), e_arg_tys)
in
  tracing (string_of_thm lthy intro); lthy
end *}

```

This prints out the theorem:

$$\llbracket \text{even } ?z; P\ 0; \bigwedge n. Q\ n \implies P\ (\text{Suc } n); \bigwedge n. P\ n \implies Q\ (\text{Suc } n) \rrbracket \implies P\ ?z$$

The export from *lthy'* to *lthy* in Line 13 above has correctly turned the free, but fixed, *z* into a schematic variable *?z*; the variables *P* and *Q* are not yet schematic.

We still have to produce the new predicates with which the introduction rules are substituted and iterate *prove_ind* over all predicates. This is what the second function, named *inds* does.

²FIXME: check with Stefan...is this so?

```

1 fun inds rules defs preds arg_tyss lthy =
2 let
3   val Ps = replicate (length preds) "P"
4   val (newprednames, lthy') = Variable.variant_fixes Ps lthy
5
6   val thy = ProofContext.theory_of lthy'
7
8   val tyss' = map (fn tys => tys ---> HOLogic.boolT) arg_tyss
9   val newpreds = map Free (newprednames ~~ tyss')
10  val cnewpreds = map (cterm_of thy) newpreds
11  val srules = map (subst_free (preds ~~ newpreds)) rules
12
13 in
14   map (prove_ind lthy' defs srules cnewpreds)
15       (preds ~~ newpreds ~~ arg_tyss)
16       |> ProofContext.export lthy' lthy
17 end

```

In Line 3, we generate a string "P" for each predicate. In Line 4, we use the same trick as in the previous function, that is making the *Ps* fresh and declaring them as free, but fixed, in the new local theory *lthy'*. From the local theory we extract the ambient theory in Line 6. We need this theory in order to certify the new predicates. In Line 8, we construct the types of these new predicates using the given argument types. Next we turn them into terms and subsequently certify them (Line 9 and 10). We can now produce the substituted introduction rules (Line 11) using the function *subst_free*. Line 14 and 15 just iterate the proofs for all predicates. From this we obtain a list of theorems. Finally we need to export the fixed variables *Ps* to obtain the schematic variables *?Ps* (Line 16).

A testcase for this function is

```

local_setup {* fn lthy =>
let
  val ind_thms = inds eo_rules eo_defs eo_preds eo_arg_tyss lthy
in
  tracing (string_of_thms lthy ind_thms); lthy
end *}

```

which prints out

```

even ?z ==> ?P1 0 ==>
  (∧m. ?Pa1 m ==> ?P1 (Suc m)) ==> (∧m. ?P1 m ==> ?Pa1 (Suc m)) ==> ?P1 ?z,
odd ?z ==> ?P1 0 ==>
  (∧m. ?Pa1 m ==> ?P1 (Suc m)) ==> (∧m. ?P1 m ==> ?Pa1 (Suc m)) ==> ?Pa1 ?z

```

Note that now both, the *?Ps* and the *?zs*, are schematic variables. The numbers attached to these variables have been introduced by the pretty-printer and are *not* important for the user.

This completes the code for the induction principles. The final peice of reasoning infrastructure we need are the introduction rules.

Introduction Rules

Constructing the goals for the introduction rules is easy: they are just the rules given by the user. However, their proofs are quite a bit more involved than the ones for the induction principles. To explain the general method, our running example will be the introduction rule

$$\wedge a b t. \llbracket a \neq b; \text{fresh } a \ t \rrbracket \implies \text{fresh } a \ (\text{Lam } b \ t)$$

about freshness for lambdas. In order to ease somewhat our work here, we use the following two helper functions.

```
val all_elims = fold (fn ct => fn th => th RS inst_spec ct)
val imp_elims = fold (fn th => fn th' => [th', th] MRS @ {thm mp})
```

To see what these functions do, let us suppose we have the following three theorems.

```
lemma all_elims_test:
fixes P::"nat => nat => nat => bool"
shows "\x y z. P x y z" sorry
```

```
lemma imp_elims_test:
shows "A -> B -> C" sorry
```

```
lemma imp_elims_test':
shows "A" "B" sorry
```

The function `all_elims` takes a list of (certified) terms and instantiates theorems of the form `all_elims_test`. For example we can instantiate the quantifiers in this theorem with `a`, `b` and `c` as follows:

```
let
  val ctrms = [@{cterm "a::nat"}, @{cterm "b::nat"}, @{cterm "c::nat"}]
  val new_thm = all_elims ctrms @ {thm all_elims_test}
in
  tracing (string_of_thm_no_vars @ {context} new_thm)
end
> P a b c
```

Note the difference with `inst_spec_tac` from Page 125: `inst_spec_tac` is a tactic which operates on a goal state; in contrast `all_elims` operates on theorems.

Similarly, the function `imp_elims` eliminates preconditions from implications. For example we can eliminate the preconditions `A` and `B` from `imp_elims_test`:

```
let
  val res = imp_elims @ {thms imp_elims_test'} @ {thm imp_elims_test}
in
  tracing (string_of_thm_no_vars @ {context} res)
end
> C
```

Now we set up the proof for the introduction rule as follows:

lemma *fresh_Lam*:

shows " $\bigwedge a\ b\ t. \llbracket a \neq b; \text{fresh } a\ t \rrbracket \implies \text{fresh } a\ (\text{Lam } b\ t)$ "

The first step in the proof will be to expand the definitions of freshness and then introduce quantifiers and implications. For this we will use the tactic

```
1 fun expand_tac defs =
2   ObjectLogic.rulify_tac 1
3   THEN rewrite_goals_tac defs
4   THEN (REPEAT (resolve_tac [ @{thm allI}, @{thm impI} ] 1))
```

The function in Line 2 “rulifies” the lemma.³ This will turn out to be important later on. Applying this tactic in our proof of *fresh_Lem*

```
apply(tactic { * expand_tac @{thms fresh_def} * })
```

gives us the goal state

```
goal (1 subgoal):
```

```
1.  $\bigwedge a\ b\ t\ \text{fresh}.$ 
```

```
    $\llbracket a \neq b;$ 
```

```
      $\forall \text{fresh}.$ 
```

```
        $(\forall a\ b. a \neq b \longrightarrow \text{fresh } a\ (\text{Var } b)) \longrightarrow$ 
```

```
        $(\forall a\ t\ s. \text{fresh } a\ t \longrightarrow \text{fresh } a\ s \longrightarrow \text{fresh } a\ (\text{App } t\ s)) \longrightarrow$ 
```

```
        $(\forall a\ t. \text{fresh } a\ (\text{Lam } a\ t)) \longrightarrow$ 
```

```
        $(\forall a\ b\ t. a \neq b \longrightarrow \text{fresh } a\ t \longrightarrow \text{fresh } a\ (\text{Lam } b\ t)) \longrightarrow \text{fresh } a\ t;$ 
```

```
    $\forall a\ b. a \neq b \longrightarrow \text{fresh } a\ (\text{Var } b);$ 
```

```
    $\forall a\ t\ s. \text{fresh } a\ t \longrightarrow \text{fresh } a\ s \longrightarrow \text{fresh } a\ (\text{App } t\ s);$ 
```

```
    $\forall a\ t. \text{fresh } a\ (\text{Lam } a\ t);$ 
```

```
    $\forall a\ b\ t. a \neq b \longrightarrow \text{fresh } a\ t \longrightarrow \text{fresh } a\ (\text{Lam } b\ t) \rrbracket$ 
```

```
 $\implies \text{fresh } a\ (\text{Lam } b\ t)$ 
```

As you can see, there are parameters (namely *a*, *b* and *t*) which come from the introduction rule and parameters (in the case above only *fresh*) which come from the universal quantification in the definition *fresh a (App t s)*. Similarly, there are assumptions that come from the premises of the rule (namely the first two) and assumptions from the definition of the predicate (assumption three to six). We need to treat these parameters and assumptions differently. In the code below we will therefore separate them into *params1* and *params2*, respectively *prems1* and *prems2*. To do this separation, it is best to open a subproof with the tactic *SUBPROOF*, since this tactic provides us with the parameters (as list of *cterm*s) and the assumptions (as list of *thm*s). The problem with *SUBPROOF*, however, is that it always expects us to completely discharge the goal (see Section 5.2). This is a bit inconvenient for our gradual explanation of the proof here. Therefore we use first the function *FOCUS*, which does same as *SUBPROOF* but does not require us to completely discharge the goal.

³FIXME: explain this better


```

fun chop_print params1 params2 prems1 prems2 ctxt =
let
  val s = ["Params1 from the rule:", string_of_cterms ctxt params1]
          @ ["Params2 from the predicate:", string_of_cterms ctxt params2]
          @ ["Prems1 from the rule:"] @ (map (string_of_thm ctxt) prems1)
          @ ["Prems2 from the predicate:"] @ (map (string_of_thm ctxt) prems2)
in
  s |> cat_lines
    |> tracing
end

```

Figure 6.3: A helper function that prints out the parameters and premises that need to be treated differently.

First we calculate the values for *params1/2* and *prems1/2* from *params* and *prems*, respectively. To better see what is going in our example, we will print out these values using the printing function in Figure 6.3. Since *FOCUS* will supply us the *params* and *prems* as lists, we can separate them using the function *chop*.

```

1 fun chop_test_tac preds rules =
2   Subgoal.FOCUS (fn {params, prems, context, ...} =>
3     let
4       val cparams = map snd params
5       val (params1, params2) = chop (length cparams - length preds) cparams
6       val (prems1, prems2) = chop (length prems - length rules) prems
7     in
8       chop_print params1 params2 prems1 prems2 context; all_tac
9     end)

```

For the separation we can rely on the fact that Isabelle deterministically produces parameters and premises in a goal state. The last parameters that were introduced come from the quantifications in the definitions (see the tactic *expand_tac*). Therefore we only have to subtract in Line 5 the number of predicates (in this case only 1) from the lengths of all parameters. Similarly with the *prems* in line 6: the last premises in the goal state come from unfolding the definition of the predicate in the conclusion. So we can just subtract the number of rules from the number of all premises. To check our calculations we print them out in Line 8 using the function *chop_print* from Figure 6.3 and then just do nothing, that is *all_tac*. Applying this tactic in our example

```
apply(tactic {* chop_test_tac [fresh_pred] fresh_rules @{context} 1 *})
```

gives

```

Params1 from the rule:
a, b, t
Params2 from the predicate:
fresh

```

Prem1 from the rule:

$a \neq b$

$\forall \text{fresh.}$

$(\forall a b. a \neq b \longrightarrow \text{fresh } a \text{ (Var } b)) \longrightarrow$

$(\forall a t s. \text{fresh } a t \longrightarrow \text{fresh } a s \longrightarrow \text{fresh } a \text{ (App } t s)) \longrightarrow$

$(\forall a t. \text{fresh } a \text{ (Lam } a t)) \longrightarrow$

$(\forall a b t. a \neq b \longrightarrow \text{fresh } a t \longrightarrow \text{fresh } a \text{ (Lam } b t)) \longrightarrow \text{fresh } a t$

Prem2 from the predicate:

$\forall a b. a \neq b \longrightarrow \text{fresh } a \text{ (Var } b)$

$\forall a t s. \text{fresh } a t \longrightarrow \text{fresh } a s \longrightarrow \text{fresh } a \text{ (App } t s)$

$\forall a t. \text{fresh } a \text{ (Lam } a t)$

$\forall a b t. a \neq b \longrightarrow \text{fresh } a t \longrightarrow \text{fresh } a \text{ (Lam } b t)$

We now have to select from *prems2* the premise that corresponds to the introduction rule we prove, namely:

$\forall a b t. a \neq b \longrightarrow \text{fresh } a t \longrightarrow \text{fresh } a \text{ (Lam } a t)$

To use this premise with *rtac*, we need to instantiate its quantifiers (with *params1*) and transform it into rule format (using *rulify*). So we can modify the code as follows:

```

1 fun apply_prem_tac i prems rules =
2   Subgoal.FOCUS (fn {params, prems, context, ...} =>
3     let
4       val cparams = map snd params
5       val (params1, params2) = chop (length cparams - length prems) cparams
6       val (prems1, prems2) = chop (length prems - length rules) prems
7     in
8       rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
9     end)

```

The argument *i* corresponds to the number of the introduction we want to prove. We will later on let it range from 0 to the number of *rules* - 1. Below we apply this function with 3, since we are proving the fourth introduction rule.

apply(tactic {* apply_prem_tac 3 [fresh_pred] fresh_rules @{context} 1 *})

The goal state we obtain is:

1. ... $\implies a \neq b$
2. ... $\implies \text{fresh } a t$

As expected there are two subgoals, where the first comes from the non-recursive premise of the introduction rule and the second comes from the recursive one. The first goal can be solved immediately by *prems1*. The second needs more work. It can be solved with the other premise in *prems1*, namely

$\forall \text{fresh.}$

$(\forall a b. a \neq b \longrightarrow \text{fresh } a \text{ (Var } b)) \longrightarrow$

$(\forall a t s. \text{fresh } a t \longrightarrow \text{fresh } a s \longrightarrow \text{fresh } a \text{ (App } t s)) \longrightarrow$

$(\forall a t. \text{fresh } a \text{ (Lam } a t)) \longrightarrow$

$(\forall a b t. a \neq b \longrightarrow \text{fresh } a t \longrightarrow \text{fresh } a \text{ (Lam } b t)) \longrightarrow \text{fresh } a t$

but we have to instantiate it appropriately. These instantiations come from *params1* and *prems2*. We can determine whether we are in the simple or complicated case by checking whether the topmost connective is an \forall . The premises in the simple case cannot have such a quantification, since the first step of *expand_tac* was to “rulify” the lemma. The premise of the complicated case must have at least one \forall coming from the quantification over the *preds*. So we can implement the following function

```
fun prepare_prem params2 prems2 prem =
  rtac (case prop_of prem of
    _ $ (Const (@{const_name All}, _) $ _) =>
      prem /> all_elims params2
      /> imp_elims prems2
    | _ => prem)
```

which either applies the premise outright (the default case) or if it has an outermost universal quantification, instantiates it first with *params1* and then *prems1*. The following tactic will therefore prove the lemma completely.

```
fun prove_intro_tac i preds rules =
  SUBPROOF (fn {params, prems, ...} =>
  let
    val cparams = map snd params
    val (params1, params2) = chop (length cparams - length preds) cparams
    val (prems1, prems2) = chop (length prems - length rules) prems
  in
    rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
    THEN EVERY1 (map (prepare_prem params2 prems2) prems1)
  end)
```

Note that the tactic is now *SUBPROOF*, not *FOCUS* anymore. The full proof of the introduction rule is as follows:

lemma *fresh_Lam*:

shows " $\bigwedge a b t. \llbracket a \neq b; \text{fresh } a \ t \rrbracket \implies \text{fresh } a \ (\text{Lam } b \ t)$ "

apply(tactic {* expand_tac @{thms fresh_def} *})

apply(tactic {* prove_intro_tac 3 [fresh_pred] fresh_rules @{context} 1 *})

done

Phew!...

Unfortunately, not everything is done yet. If you look closely at the general principle outlined for the introduction rules in Section 6.3, we have not yet dealt with the case where recursive premises have preconditions. The introduction rule of the accessible part is such a rule.

lemma *accpartI*:

shows " $\bigwedge R x. (\bigwedge y. R \ y \ x \implies \text{accpart } R \ y) \implies \text{accpart } R \ x$ "

apply(tactic {* expand_tac @{thms accpart_def} *})

apply(tactic {* chop_test_tac [acc_pred] acc_rules @{context} 1 *})

apply(tactic {* apply_prem_tac 0 [acc_pred] acc_rules @{context} 1 *})

Here *chop_test_tac* prints out the following values for *params1/2* and *prems1/2*

Params1 from the rule:

x

Params2 from the predicate:

P

Prem1 from the rule:

$R \text{ ?}y \ x \implies \forall P. (\forall x. (\forall y. R \ y \ x \longrightarrow P \ y) \longrightarrow P \ x) \longrightarrow P \ \text{?}y$

Prem2 from the predicate:

$\forall x. (\forall y. R \ y \ x \longrightarrow P \ y) \longrightarrow P \ x$

and after application of the introduction rule using `apply_prem_tac`, we are in the goal state

1. $\bigwedge y. R \ y \ x \implies P \ y$

In order to make progress, we have to use the precondition $R \ y \ x$ (in general there can be many of them). The best way to get a handle on these preconditions is to open up another subproof, since the preconditions will then be bound to `prems`. Therefore we modify the function `prepare_prem` as follows

```

1 fun prepare_prem params2 prems2 ctxt prem =
2   SUBPROOF (fn {prems, ...} =>
3     let
4       val prem' = prems MRS prem
5     in
6       rtac (case prop_of prem' of
7         _ $ (Const (@{const_name All}, _) $ _) =>
8           prem' |> all_elims params2
9           |> imp_elims prems2
10        | _ => prem') 1
11     end) ctxt

```

In Line 4 we use the `prems` from the `SUBPROOF` and resolve them with `prem`. In the simple cases, that is where the `prem` comes from a non-recursive premise of the rule, `prems` will be just the empty list and the function `MRS` does nothing. Similarly, in the cases where the recursive premises of the rule do not have preconditions. In case there are preconditions, then Line 4 discharges them. After that we can proceed as before, i.e., check whether the outermost connective is \forall .

The function `prove_intro_tac` only needs to be changed so that it gives the context to `prepare_prem` (Line 8). The modified version is below.

```

1 fun prove_intro_tac i preds rules =
2   SUBPROOF (fn {params, prems, context, ...} =>
3     let
4       val cparams = map snd params
5       val (params1, params2) = chop (length cparams - length preds) cparams
6       val (prems1, prems2) = chop (length prems - length rules) prems
7     in
8       rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
9       THEN EVERY1 (map (prepare_prem params2 prems2 context) prems1)
10     end)

```

With these two functions we can now also prove the introduction rule for the accessible part.

```

lemma accpartI:
shows " $\bigwedge R x. (\bigwedge y. R y x \implies \text{accpart } R y) \implies \text{accpart } R x$ "
apply(tactic {* expand_tac @{thms accpart_def} *})
apply(tactic {* prove_intro_tac 0 [acc_pred] acc_rules @{context} 1 *})
done

```

Finally we need two functions that string everything together. The first function is the tactic that performs the proofs.

```

1 fun intro_tac defs rules preds i ctxt =
2   EVERY1 [ObjectLogic.rulify_tac,
3           K (rewrite_goals_tac defs),
4           REPEAT o (resolve_tac [@[thm allI], @[thm impI]]),
5           prove_intro_tac i preds rules ctxt]

```

Lines 2 to 4 in this tactic correspond to the function `expand_tac`. Some testcases for this tactic are:

```

lemma even0_intro:
shows "even 0"
by (tactic {* intro_tac eo_defs eo_rules eo_preds 0 @{context} *})

```

```

lemma evenS_intro:
shows " $\bigwedge m. \text{odd } m \implies \text{even } (\text{Suc } m)$ "
by (tactic {* intro_tac eo_defs eo_rules eo_preds 1 @{context} *})

```

```

lemma fresh_App:
shows " $\bigwedge a t s. [\text{fresh } a t; \text{fresh } a s] \implies \text{fresh } a (\text{App } t s)$ "
by (tactic {*
  intro_tac @{thms fresh_def} fresh_rules [fresh_pred] 1 @{context} *})

```

The second function sets up in Line 4 the goals to be proved (this is easy for the introduction rules since they are exactly the rules given by the user) and iterates `intro_tac` over all introduction rules.

```

1 fun intros rules preds defs lthy =
2   let
3     fun intros_aux (i, goal) =
4       Goal.prove lthy [] [] goal
5         (fn {context, ...} => intro_tac defs rules preds i context)
6   in
7     map_index intros_aux rules
8   end

```

The iteration is done with the function `map_index` since we need the introduction rule together with its number (counted from 0). This completes the code for the functions deriving the reasoning infrastructure. It remains to implement some administrative code that strings everything together.

Administrative Functions

We have produced various theorems (definitions, induction principles and introduction rules), but apart from the definitions, we have not yet registered them with the theorem database. This is what the functions *note* does.

For convenience, we use the following three wrappers this function:

```

fun note_many qname ((name, attrs), thms) =
  LocalTheory.note Thm.theoremK
    ((Binding.qualify false qname name, attrs), thms)

fun note_single1 qname ((name, attrs), thm) =
  note_many qname ((name, attrs), [thm])

fun note_single2 name attrs (qname, thm) =
  note_many (Binding.name_of qname) ((name, attrs), [thm])

```

The function that “holds everything together” is *add_inductive*. Its arguments are the specification of the predicates *pred_specs* and the introduction rules *rule_spec*.

```

1 fun add_inductive pred_specs rule_specs lthy =
2   let
3     val mxs = map snd pred_specs
4     val pred_specs' = map fst pred_specs
5     val prednames = map fst pred_specs'
6     val preds = map (fn (p, ty) => Free (Binding.name_of p, ty)) pred_specs'
7     val tyss = map (binder_types o fastype_of) preds
8
9     val (namesattrs, rules) = split_list rule_specs
10
11    val (defs, lthy') = defns rules preds prednames mxs tyss lthy
12    val ind_prins = inds rules defs preds tyss lthy'
13    val intro_rules = intros rules preds defs lthy'
14
15    val mut_name = space_implode "_" (map Binding.name_of prednames)
16    val case_names = map (Binding.name_of o fst) namesattrs
17  in
18    lthy' |> note_many mut_name (@{binding "intros"}, []), intro_rules)
19    ||>> note_many mut_name (@{binding "inducts"}, []), ind_prins)
20    ||>> fold_map (note_single1 mut_name) (namesattrs ~~ intro_rules)
21    ||>> fold_map (note_single2 @{binding "induct"}
22      [Attrib.internal (K (RuleCases.case_names case_names)),
23       Attrib.internal (K (RuleCases.consumes 1)),
24       Attrib.internal (K (Induct.induct_pred ""))])
25      (prednames ~~ ind_prins)
26    |> snd
27  end

```

In Line 3 the function extracts the syntax annotations from the predicates. Lines 4 to 6 extract the names of the predicates and generate the variables terms (with

types) corresponding to the predicates. Line 7 produces the argument types for each predicate.

Line 9 extracts the introduction rules from the specifications and stores also in *namesattrs* the names and attributes the user may have attached to these rules.

Line 11 produces the definitions and also registers the definitions in the local theory *lthy'*. The next two lines produce the induction principles and the introduction rules (all of them as theorems). Both need the local theory *lthy'* in which the definitions have been registered.

Lines 15 produces the name that is used to register the introduction rules. It is custom to collect all introduction rules under *string.intros*, whereby *string* stands for the "-"-separated list of predicate names (for example *even_odd*). Also by custom, the case names in intuction proofs correspond to the names of the introduction rules. These are generated in Line 16.

Lines 18 and 19 now add to *lthy'* all the introduction rules und induction principles under the name *mut_name.intros* and *mut_name.inducts*, respectively (see previous paragraph).

Line 20 add further every introduction rule under its own name (given by the user).⁴ Line 21 registers the induction principles. For this we have to use some specific attributes. The first *case_names* corresponds to the case names that are used by Isar to reference the proof obligations in the induction. The second *consumes 1* indicates that the first premise of the induction principle (namely the predicate over which the induction proceeds) is eliminated.

This completes all the code and fits in with the "front end" described in Section 6.2.⁵

6.5 Extensions of the Package (TBD)

Things to include at the end:

- include the code for the parameters
- say something about add-inductive to return the rules
- say something about the two interfaces for calling packages

Exercise 6.5.1. In Section 6.3 we required that introduction rules must be of the form

$$\text{rule} ::= \bigwedge xs. As \implies (\bigwedge ys. Bs \implies \text{pred } ss)^* \implies \text{pred } ts$$

where the *As* and *Bs* can be any collection of formulae not containing the *preds*. This requirement is important, because if violated, the theory behind the inductive package does not work and also the proofs break. Write code that tests whether the introduction rules given by the user fit into the scheme described above. Hint: It is not important in which order the premises are given; the *As* and $(\bigwedge ys. Bs \implies \text{pred } ss)$ premises can occur in any order.

⁴FIXME: what happens if the user did not give any name.

⁵FIXME: Describe *Induct.induct_pred*. Why the mut-name? What does *Binding.qualify* do?

Exercise 6.5.2. *If you define `even` and `odd` with the standard inductive package*

inductive

`even_2 and odd_2`

where

```
even0_2: "even_2 0"
| evenS_2: "odd_2 m  $\implies$  even_2 (Suc m)"
| oddS_2: "even_2 m  $\implies$  odd_2 (Suc m)"
```

you will see that the generated induction principle for `even`' (namely `even_2_odd_2.inducts` has the additional assumptions `odd_2 m` and `even_2 m` in the recursive cases. These additional assumptions can sometimes make "life easier" in proofs. Since more assumptions can be made when proving properties, these induction principles are called strong inductions principles. However, it is the case that the "weak" induction principles imply the "strong" ones. Hint: Prove a property taking a pair (or tuple in case of more than one predicate) as argument: the property that you originally want to prove and the predicate(s) over which the induction proceeds. Write code that automates the derivation of the strong induction principles from the weak ones.

Read More

The standard inductive package is based on least fix-points. It allows more general introduction rules that can include any monotone operators and also provides a richer reasoning infrastructure. The code of this package can be found in [HOL/Tools/inductive.ML](#).

6.6 Definitional Packages

Type declarations

```
Typedef.add_typedef false NONE (@{binding test}, [], NoSyn)
  @{term "{1}::nat set"} NONE (simp_tac @{simpset} 1) @{theory}
```


Appendix A

Recipes

Possible topics:

- translations/print translations; *ProofContext.print_syntax*
- user space type systems (in the form that already exists)
- unification and typing algorithms (*Pure/pattern.ML* implements HOPU)
- useful datastructures: discrimination nets, association lists

A.1 Useful Document Antiquotations

Problem: How to keep your ML-code inside a document synchronised with the actual code?

Solution: This can be achieved with document antiquotations.

Document antiquotations can be used for ensuring consistent type-setting of various entities in a document. They can also be used for sophisticated \TeX -hacking. If you type on the Isabelle level

print antiquotations

you obtain a list of all currently available document antiquotations and their options. Below we will give the code for two additional document antiquotations both of which are intended to typeset ML-code. The crucial point of these document antiquotations is that they not just print the ML-code, but also check whether it compiles. This will provide a sanity check for the code and also allows you to keep documents in sync with other code, for example Isabelle.

We first describe the antiquotation *ML_checked* with the syntax:

```
@{ML_checked "a_piece_of_code"}
```

The code is checked by sending the ML-expression *"val _ = a_piece_of_code"* to the ML-compiler (i.e. the function *ML_Context.eval_in* in Line 4 below). The complete code of the document antiquotation is as follows:

```

1 fun ml_val code_txt = "val _ = " ^ code_txt
2
3 fun output_ml {context = ctxt, ...} code_txt =
4   (ML_Context.eval_in (SOME ctxt) false Position.none (ml_val code_txt);
5    ThyOutput.output (map Pretty.str (space_explode "\n" code_txt)))
6
7 val _ = ThyOutput.antiquotation "ML_checked" (Scan.lift Args.name) output_ml

```

The parser (`Scan.lift Args.name`) in Line 7 parses a string, in this case the code, and then calls the function `output_ml`. As mentioned before, the parsed code is sent to the ML-compiler in Line 4 using the function `ml_val`, which constructs the appropriate ML-expression, and using `eval_in`, which calls the compiler. If the code is “approved” by the compiler, then the output function `output` in the next line pretty prints the code. This function expects that the code is a list of (pretty)strings where each string correspond to a line in the output. Therefore the use of (`space_explode "\n" txt`) which produces such a list according to linebreaks. There are a number of options for antiquotations that are observed by the function `output` when printing the code (including `[display]` and `[quotes]`). The function `antiquotation` in Line 7 sets up the new document antiquotation.

Read More

For more information about options of document antiquotations see [Isar Ref.Man., Sec. 4.2]).

Since we used the argument `Position.none`, the compiler cannot give specific information about the line number, in case an error is detected. We can improve the code above slightly by writing

```

1 fun output_ml {context = ctxt, ...} (code_txt, pos) =
2   (ML_Context.eval_in (SOME ctxt) false pos (ml_val code_txt);
3    ThyOutput.output (map Pretty.str (space_explode "\n" code_txt)))
4
5 val _ = ThyOutput.antiquotation "ML_checked"
6   (Scan.lift (OuterParse.position Args.name)) output_ml

```

where in Lines 1 and 2 the positional information is properly treated. The parser `OuterParse.position` encodes the positional information in the result.

We can now write `@{ML_checked "2 + 3"}` in a document in order to obtain `2 + 3` and be sure that this code compiles until somebody changes the definition of addition.

The second document antiquotation we describe extends the first by a pattern that specifies what the result of the ML-code should be and checks the consistency of the actual result with the given pattern. For this we are going to implement the document antiquotation:

```
@{ML_resp "a_piece_of_code" "a_pattern"}
```

To add some convenience and also to deal with large outputs, the user can give a partial specification by using ellipses. For example `(... , ...)` for specifying a pair.

In order to check consistency between the pattern and the output of the code, we have to change the ML-expression that is sent to the compiler: in *ML_checked* we sent the expression `"val _ = a_piece_of_code"` to the compiler; now the wildcard `_` must be replaced by the given pattern. However, we have to remove all ellipses from it and replace them by `"_"`. The following function will do this:

```
fun ml_pat (code_txt, pat) =
  let val pat' =
        implode (map (fn "... " => "_" | s => s) (Symbol.explode pat))
      in
        "val " ^ pat' ^ " = " ^ code_txt
      end
```

Next we add a response indicator to the result using:

```
fun add_resp pat = map (fn s => "> " ^ s) pat
```

The rest of the code of *ML_resp* is:

```
1 fun output_ml_resp {context = ctxt, ...} ((code_txt, pat), pos) =
2   (ML_Context.eval_in (SOME ctxt) false pos (ml_pat (code_txt, pat)));
3   let
4     val code_output = space_explode "\n" code_txt
5     val resp_output = add_resp (space_explode "\n" pat)
6   in
7     ThyOutput.output (map Pretty.str (code_output @ resp_output))
8   end
9
10 val _ = ThyOutput.antiquotation "ML_resp"
11   (Scan.lift (OuterParse.position (Args.name -- Args.name)))
12   output_ml_resp
```

In comparison with *ML_checked*, we only changed the line about the compiler (Line 2), the lines about the output (Lines 4 to 7) and the parser in the setup (Line 11). Now you can write

```
@{ML_resp [display] "true andalso false" "false"}
```

to obtain

```
true andalso false
> false
```

or

```
@{ML_resp [display] "let val i = 3 in (i * i, "foo") end" "(9, ...)"}
```

to obtain

```
let val i = 3 in (i * i, "foo") end
> (9, ...)
```

In both cases, the check by the compiler ensures that code and result match. A limitation of this document antiquotation, however, is that the pattern can only be given for values that can be constructed. This excludes values that are abstract datatypes, like *thms* and *cterm*s.

A.2 Restricting the Runtime of a Function

Problem: Your tool should run only a specified amount of time.

Solution: In PolyML 5.2.1 and later, this can be achieved using the function *timeLimit*.

Assume you defined the Ackermann function on the ML-level.

```
fun ackermann (0, n) = n + 1
  | ackermann (m, 0) = ackermann (m - 1, 1)
  | ackermann (m, n) = ackermann (m - 1, ackermann (m, n - 1))
```

Now the call

```
ackermann (4, 12)
> ...
```

takes a bit of time before it finishes. To avoid this, the call can be encapsulated in a time limit of five seconds. For this you have to write

```
TimeLimit.timeLimit (Time.fromSeconds 5) ackermann (4, 12)
  handle TimeLimit.TimeOut => ~1
> ~1
```

where *TimeOut* is the exception raised when the time limit is reached.

Note that *timeLimit* is only meaningful when you use PolyML 5.2.1 or later, because this version of PolyML has the infrastructure for multithreaded programming on which *timeLimit* relies.

Read More

The function *timeLimit* is defined in the structure *TimeLimit* which can be found in the file [Pure/ML-Systems/multithreading_polyml.ML](#).

A.3 Measuring Time

Problem: You want to measure the running time of a tactic or function.

Solution: Time can be measured using the function *start_timing* and *end_timing*.

Suppose you defined the Ackermann function on the Isabelle level.

```

fun
  ackermann:: "(nat × nat) ⇒ nat"
where
  "ackermann (0, n) = n + 1"
  | "ackermann (m, 0) = ackermann (m - 1, 1)"
  | "ackermann (m, n) = ackermann (m - 1, ackermann (m, n - 1))"

```

You can measure how long the simplifier takes to verify a datapoint of this function. The actual timing is done inside the wrapper function:

```

1 fun timing_wrapper tac st =
2   let
3     val t_start = start_timing ();
4     val res = tac st;
5     val t_end = end_timing t_start;
6   in
7     (writeln (#message t_end); res)
8   end

```

Note that this function, in addition to a tactic, also takes a state *st* as argument and applies this state to the tactic (Line 4). The reason is that tactics are lazy functions and you need to force them to run, otherwise the timing will be meaningless. The simplifier tactic, amongst others, can be forced to run by just applying the state to it. But “fully” lazy tactics, such as *resolve_tac*, need even more “standing-on-ones-head” to force them to run.

The time between start and finish of the simplifier will be calculated as the end time minus the start time. An example of the wrapper is the proof

```

lemma "ackermann (3, 4) = 125"
apply(tactic {*
  timing_wrapper (simp_tac (@{simpset} addsimps @{thms "nat_number"}) 1) *)}
done

```

where it returns something on the scale of 3 seconds. We chose to return this information as a string, but the timing information is also accessible in number format.

Read More

Basic functions regarding timing are defined in [Pure/ML-Systems/polymml_common.ML](#) (for the PolyML compiler). Some more advanced functions are defined in [Pure/General/output.ML](#).

A.4 Configuration Options

Problem: You would like to enhance your tool with options that can be changed by the user without having to resort to the ML-level.

Solution: This can be achieved using configuration values.

Assume you want to control three values, say *bval* containing a boolean, *ival* containing an integer and *sval* containing a string. These values can be declared on the ML-level by

```
val (bval, setup_bval) = Attrib.config_bool "bval" false
val (ival, setup_ival) = Attrib.config_int "ival" 0
val (sval, setup_sval) = Attrib.config_string "sval" "some string"
```

where each value needs to be given a default. To enable these values, they need to be set up with

```
setup {* setup_bval *}
setup {* setup_ival *}
```

or on the ML-level with

```
setup_sval @{theory}
```

The user can now manipulate the values from within Isabelle with the command

```
declare [[bval = true, ival = 3]]
```

On the ML-level these values can be retrieved using the function *Config.get*:

```
Config.get @{context} bval
> true
```

```
Config.get @{context} ival
> 3
```

The function *Config.put* manipulates the values. For example

```
Config.put sval "foo" @{context}; Config.get @{context} sval
> foo
```

The same can be achieved using the command **setup**.

```
setup {* Config.put_thy sval "bar" *}
```

Now the retrieval of this value yields:

```
Config.get @{context} sval
> "bar"
```

We can apply a function to a value using *Config.map*. For example incrementing *ival* can be done by:

```

let
  val ctxt' = Config.map ival (fn i => i + 1) @{context}
in
  Config.get ctxt' ival
end
> 4

```

Read More

For more information see [Pure/Isar/attrib.ML](#) and [Pure/config.ML](#).

There are many good reasons to control parameters in this way. One is that no global reference is needed, which would cause many headaches with the multithreaded execution of Isabelle.

A.5 Storing Data (TBD)

Problem: Your tool needs to manage data.

Solution: This can be achieved using a generic data slot.

Every generic data slot may keep data of any kind which is stored in the context.

```

local

structure Data = GenericDataFun
  ( type T = int Symtab.table
    val empty = Symtab.empty
    val extend = I
    fun merge _ = Symtab.merge (K true)
  )

in
  val lookup = Symtab.lookup o Data.get
  fun update k v = Data.map (Symtab.update (k, v))
end

```

```

setup {* Context.theory_map (update "foo" 1) *}

```

```

lookup (Context.Proof @{context}) "foo"
> SOME 1

```

alternatives: TheoryDataFun, ProofDataFun Code: [Pure/context.ML](#)

A.6 Executing an External Application (TBD)

Problem: You want to use an external application.

Solution: The function `system_out` might be the right thing for you.

This function executes an external command as if printed in a shell. It returns the output of the program and its return value.

For example, consider running an ordinary shell commands:

```
system_out "echo Hello world!"
> ("Hello world!\n", 0)
```

Note that it works also fine with timeouts (see Recipe A.2 on Page 142), i.e. external applications are killed properly. For example, the following expression takes only approximately one second:

```
TimeLimit.timeLimit (Time.fromSeconds 1) system_out "sleep 30"
  handle TimeLimit.TimeOut => ("timeout", ~1)
> ("timeout", ~1)
```

The function `system_out` can also be used for more reasonable applications, e.g. coupling external solvers with Isabelle. In that case, one has to make sure that Isabelle can find the particular executable. One way to ensure this is by adding a Bash-like variable binding into one of Isabelle's settings file (prefer the user settings file usually to be found at `$HOME/.isabelle/etc/settings`).

For example, assume you want to use the application `foo` which is here supposed to be located at `/usr/local/bin/`. The following line has to be added to one of Isabelle's settings file:

```
FOO=/usr/local/bin/foo
```

In Isabelle, this application may now be executed by

```
system_out "$FOO"
> ...
```

A.7 Writing an Oracle (TBD)

Problem: You want to use a fast, new decision procedure not based on Isabelle's tactics, and you do not care whether it is sound.

Solution: Isabelle provides the oracle mechanisms to bypass the inference kernel. Note that theorems proven by an oracle carry a special mark to inform the user of their potential incorrectness.

Read More

A short introduction to oracles can be found in [isar-ref: no suitable label for section 3.11].
 A simple example, which we will slightly extend here, is given in [FOL/ex/Iff_Oracle.thy](#).
 The raw interface for adding oracles is `add_oracle` in [Pure/thm.ML](#).

For our explanation here, we restrict ourselves to decide propositional formulae which consist only of equivalences between propositional variables, i.e. we want to decide whether $(P = (Q = P)) = Q$ is a tautology.

Assume, that we have a decision procedure for such formulae, implemented in ML. Since we do not care how it works, we will use it here as an “external solver”:

```
use "external_solver.ML"
```

We do, however, know that the solver provides a function `IffSolver.decide`. It takes a string representation of a formula and returns either `true` if the formula is a tautology or `false` otherwise. The input syntax is specified as follows:

```
formula ::= atom | ( formula <=> formula )
```

and all tokens are separated by at least one space.

(FIXME: is there a better way for describing the syntax?)

We will proceed in the following way. We start by translating a HOL formula into the string representation expected by the solver. The solver’s result is then used to build an oracle, which we will subsequently use as a core for an Isar method to be able to apply the oracle in proving theorems.

Let us start with the translation function from Isabelle propositions into the solver’s string representation. To increase efficiency while building the string, we use functions from the `Buffer` module.

```
fun translate t =
  let
    fun trans t =
      (case t of
        @term "op = :: bool => bool => bool" $ t $ u =>
          Buffer.add " (" #>
            trans t #>
            Buffer.add "<=>" #>
            trans u #>
            Buffer.add ") "
        | Free (n, @typ bool) =>
          Buffer.add " " #>
          Buffer.add n #>
          Buffer.add " "
        | _ => error "inacceptable term")
      in Buffer.content (trans t Buffer.empty) end
```

Here is the string representation of the term $p = (q = p)$:

```
translate @term "p = (q = p)"
> " ( p <=> ( q <=> p ) ) "
```

Let us check, what the solver returns when given a tautology:

```
IffSolver.decide (translate @{term "p = (q = p) = q"})
> true
```

And here is what it returns for a formula which is not valid:

```
IffSolver.decide (translate @{term "p = (q = p)"})
> false
```

Now, we combine these functions into an oracle. In general, an oracle may be given any input, but it has to return a certified proposition (a special term which is type-checked), out of which Isabelle’s inference kernel “magically” makes a theorem.

Here, we take the proposition to be show as input. Note that we have to first extract the term which is then passed to the translation and decision procedure. If the solver finds this term to be valid, we return the given proposition unchanged to be turned then into a theorem:

```
oracle iff_oracle = {* fn ct =>
  if IffSolver.decide (translate (HOLogic.dest_Trueprop (Thm.term_of ct)))
  then ct
  else error "Proof failed."*}
```

Here is what we get when applying the oracle:

```
iff_oracle @{cprop "p = (p::bool)"}
> p = p
```

(FIXME: is there a better way to present the theorem?)

To be able to use our oracle for Isar proofs, we wrap it into a tactic:

```
val iff_oracle_tac =
  CSUBGOAL (fn (goal, i) =>
    (case try iff_oracle goal of
      NONE => no_tac
      | SOME thm => rtac thm i))
```

and create a new method solely based on this tactic:

```
method_setup iff_oracle = {*
  Scan.succeed (K (Method.SIMPLE_METHOD' iff_oracle_tac))
*} "Oracle-based decision procedure for chains of equivalences"
```

Finally, we can test our oracle to prove some theorems:

```
lemma "p = (p::bool)"
  by iff_oracle
```

```
lemma "p = (q = p) = q"
  by iff_oracle
```

(FIXME: say something about what the proof of the oracle is ... what do you mean?)

A.8 SAT Solvers

Problem: You like to use a SAT solver to find out whether an Isabelle formula is satisfiable or not.

Solution: Isabelle contains a general interface for a number of external SAT solvers (including ZChaff and Minisat) and also contains a simple internal SAT solver that is based on the DPLL algorithm.

The SAT solvers expect a propositional formula as input and produce a result indicating that the formula is either satisfiable, unsatisfiable or unknown. The type of the propositional formula is `PropLogic.prop_formula` with the usual constructors such as `And`, `Or` and so on.

The function `PropLogic.prop_formula_of_term` translates an Isabelle term into a propositional formula. Let us illustrate this function by translating $A \wedge \neg A \vee B$. The function will return a propositional formula and a table. Suppose

```
val (pform, table) =
  PropLogic.prop_formula_of_term @{term "A ∧ ¬A ∨ B"} Termtab.empty
```

then the resulting propositional formula `pform` is

```
Or (And (BoolVar 1, Not (BoolVar 1)), BoolVar 2)
```

where indices are assigned for the variables A and B , respectively. This assignment is recorded in the table that is given to the translation function and also returned (appropriately updated) in the result. In the case above the input table is empty (i.e. `Termtab.empty`) and the output table is

```
Termtab.dest table
> [(Free ("A", "bool"), 1), (Free ("B", "bool"), 2)]
```

An index is also produced whenever the translation function cannot find an appropriate propositional formula for a term. Attempting to translate $\forall x. P x$

```
val (pform', table') =
  PropLogic.prop_formula_of_term @{term "∀x::nat. P x"} Termtab.empty
```

returns `BoolVar 1` for `pform'` and the table `table'` is:

```
map (apfst (Syntax.string_of_term @{context})) (Termtab.dest table')
> (∀x. P x, 1)
```

In the print out of the tabel, we used some pretty printing scaffolding to see better which assignment the table contains.

Having produced a propositional formula, you can now call the SAT solvers with the function `SatSolver.invoke_solver`. For example

```
SatSolver.invoke_solver "dpll" pform
> SatSolver.SATISFIABLE assg
```

determines that the formula *pform* is satisfiable. If we inspect the returned function *assg*

```
let
  val SatSolver.SATISFIABLE assg = SatSolver.invoke_solver "dpll" pform
in
  (assg 1, assg 2, assg 3)
end
> (SOME true, SOME true, NONE)
```

we obtain a possible assignment for the variables *A* and *B* that makes the formula satisfiable.

Note that we invoked the SAT solver with the string "dpll". This string specifies which SAT solver is invoked (in this case the internal one). If instead you invoke the SAT solver with the string "auto"

```
SatSolver.invoke_solver "auto" pform
```

several external SAT solvers will be tried (assuming they are installed). If no external SAT solver is installed, then the default is "dpll".

There are also two tactics that make use of SAT solvers. One is the tactic *sat_tac*. For example

```
lemma "True"
apply(tactic {* sat.sat_tac @context 1 *})
done
```

However, for proving anything more exciting using *sat_tac* you have to use a SAT solver that can produce a proof. The internal one is not usable for this.

Read More

The interface for the external SAT solvers is implemented in [HOL/Tools/sat_solver.ML](#). This file contains also a simple SAT solver based on the DPLL algorithm. The tactics for SAT solvers are implemented in [HOL/Tools/sat_funcs.ML](#). Functions concerning propositional formulas are implemented in [HOL/Tools/prop_logic.ML](#). The tables used in the translation function are implemented in [Pure/General/table.ML](#).

A.9 User Space Type-Systems (TBD)

Appendix B

Solutions to Most Exercises

Solution for Exercise 3.2.1.

```
fun rev_sum ((p as Const (@{const_name plus}, _)) $ t $ u) =
  p $ u $ rev_sum t
| rev_sum t = t
```

An alternative solution using the function `mk_binop` is:

```
fun rev_sum t =
let
  fun dest_sum (Const (@{const_name plus}, _) $ u $ u') = u' :: dest_sum u
  | dest_sum u = [u]
in
  foldl1 (HOLogic.mk_binop @{const_name plus}) (dest_sum t)
end
```

Solution for Exercise 3.2.2.

```
fun make_sum t1 t2 =
  HOLogic.mk_nat (HOLogic.dest_nat t1 + HOLogic.dest_nat t2)
```

Solution for Exercise 3.2.3.

```
fun P n = @{term "P::nat ⇒ bool"} $ (HOLogic.mk_number @{typ "nat"} n)

fun rhs 1 = P 1
  | rhs n = HOLogic.mk_conj (P n, rhs (n - 1))

fun lhs 1 n = HOLogic.mk_imp (HOLogic.mk_eq (P 1, P n), rhs n)
  | lhs m n = HOLogic.mk_conj (HOLogic.mk_imp
    (HOLogic.mk_eq (P (m - 1), P m), rhs n), lhs (m - 1) n)

fun de_bruijn n =
  HOLogic.mk_Trueprop (HOLogic.mk_imp (lhs n n, rhs n))
```

Solution for Exercise 4.1.1.

```

val any = Scan.one (Symbol.not_eof)

val scan_cmt =
let
  val begin_cmt = Scan.this_string "("
  val end_cmt = Scan.this_string ")"
in
  begin_cmt |-- Scan.repeat (Scan.unless end_cmt any) --| end_cmt
  >> (enclose "**" "**)" o implode)
end

val parser = Scan.repeat (scan_cmt || any)

val scan_all =
  Scan.finite Symbol.stopper parser >> implode #> fst

```

By using `#> fst` in the last line, the function `scan_all` retruns a string, instead of the pair a parser would normally return. For example:

```

let
  val input1 = (explode "foo bar")
  val input2 = (explode "foo (*test*) bar (*test*)")
in
  (scan_all input1, scan_all input2)
end
> ("foo bar", "foo (**test**) bar (**test**)")

```

Solution for Exercise 5.3.1.

The axiom rule can be implemented with the function `atac`. The other rules correspond to the theorems:

\wedge_R	<i>conjI</i>	<i>False</i>	<i>FalseE</i>
\vee_{R1}	<i>disjI1</i>	\wedge_L	<i>conjE</i>
\vee_{R2}	<i>disjI2</i>	\vee_L	<i>disjE</i>
\longrightarrow_R	<i>impI</i>	$=_L$	<i>iffE</i>
$=_R$	<i>iffI</i>		

For the other rules we need to prove the following lemmas.

lemma *impE1*:
shows " $\llbracket A \longrightarrow B; A; B \Longrightarrow R \rrbracket \Longrightarrow R$ "
by *iprover*

lemma *impE2*:
shows " $\llbracket (C \wedge D) \longrightarrow B; C \longrightarrow (D \longrightarrow B) \Longrightarrow R \rrbracket \Longrightarrow R$ "
by *iprover*

```

lemma impE3:
  shows " $\llbracket (C \vee D) \longrightarrow B; \llbracket C \longrightarrow B; D \longrightarrow B \rrbracket \Longrightarrow R \rrbracket \Longrightarrow R$ "
  by iprover

lemma impE4:
  shows " $\llbracket (C \longrightarrow D) \longrightarrow B; D \longrightarrow B \Longrightarrow C \longrightarrow D; B \Longrightarrow R \rrbracket \Longrightarrow R$ "
  by iprover

lemma impE5:
  shows " $\llbracket (C = D) \longrightarrow B; (C \longrightarrow D) \longrightarrow ((D \longrightarrow C) \longrightarrow B) \Longrightarrow R \rrbracket \Longrightarrow R$ "
  by iprover

```

Now the tactic which applies a single rule can be implemented as follows.

```

1  val apply_tac =
2  let
3    val intros = [@{thm conjI}], @{thm disjI1}], @{thm disjI2}],
4                @{thm impI}], @{thm iffI}]
5    val elims = [@{thm FalseE}], @{thm conjE}], @{thm disjE}], @{thm iffE}],
6                @{thm impE2}], @{thm impE3}], @{thm impE4}], @{thm impE5}]
7  in
8    atac
9    ORELSE' resolve_tac intros
10   ORELSE' eresolve_tac elims
11   ORELSE' (etac @{thm impE1} THEN' atac)
12 end

```

In Line 11 we apply the rule *impE1* in conjunction with *atac* in order to reduce the number of possibilities that need to be explored. You can use the tactic as follows.

```

lemma
  shows " $((((P \longrightarrow Q) \longrightarrow P) \longrightarrow P) \longrightarrow Q) \longrightarrow Q$ "
apply(tactic {* (DEPTH_SOLVE o apply_tac) 1 *})
done

```

We can use the tactic to prove or disprove automatically the de Bruijn formulae from Exercise 3.2.3.

```

fun de_bruijn_prove ctxt n =
let
  val goal = HOLogic.mk_Trueprop (HOLogic.mk_imp (lhs n n, rhs n))
in
  Goal.prove ctxt ["P"] [] goal
  (fn _ => (DEPTH_SOLVE o apply_tac) 1)
end

```

You can use this function to prove de Bruijn formulae.

```

de_bruijn_prove @{context} 3

```

Solution for Exercise 5.5.1.

```

fun dest_sum term =
  case term of
    (@{term "(op +):: nat ⇒ nat ⇒ nat"} $ t1 $ t2) =>
      (snd (HOLogic.dest_number t1), snd (HOLogic.dest_number t2))
  | _ => raise TERM ("dest_sum", [term])

fun get_sum_thm ctxt t (n1, n2) =
  let
    val sum = HOLogic.mk_number @{typ "nat"} (n1 + n2)
    val goal = Logic.mk_equals (t, sum)
  in
    Goal.prove ctxt [] [] goal (K (Arith_Data.arith_tac ctxt 1))
  end

fun add_sp_aux ss t =
  let
    val ctxt = Simplifier.the_context ss
    val t' = term_of t
  in
    SOME (get_sum_thm ctxt t' (dest_sum t'))
  handle TERM _ => NONE
end

```

The setup for the simproc is

```

simproc_setup add_sp ("t1 + t2") = {* K add_sp_aux *}

```

and a test case is the lemma

```

lemma "P (Suc (99 + 1)) ((0 + 0)::nat) (Suc (3 + 3 + 3)) (4 + 1)"
  apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [#{@simproc add_sp}]) 1 *})

```

where the simproc produces the goal state

```

goal (1 subgoal):
  1. P (Suc 100) 0 (Suc 9) ((4::'a) + (1::'a))

```

Solution for Exercise 5.6.1.

The following code assumes the function *dest_sum* from the previous exercise.

```

fun add_simple_conv ctxt ctrm =
  let
    val trm = Thm.term_of ctrm
  in
    get_sum_thm ctxt trm (dest_sum trm)
  end

fun add_conv ctxt ctrm =
  (case Thm.term_of ctrm of
    @{term "(op +)::nat ⇒ nat ⇒ nat"} $ _ $ _ =>
      (Conv.binop_conv (add_conv ctxt)

```



```

      then_conv (Conv.try_conv (add_simple_conv ctxt)) ctrm
| _ $ _ => Conv.combination_conv
      (add_conv ctxt) (add_conv ctxt) ctrm
| Abs _ => Conv.abs_conv (fn (_, ctxt) => add_conv ctxt) ctxt ctrm
| _ => Conv.all_conv ctrm

fun add_tac ctxt = CSUBGOAL (fn (goal, i) =>
  CONVERSION
  (Conv.params_conv ~1 (fn ctxt =>
    (Conv.prems_conv ~1 (add_conv ctxt) then_conv
     Conv.concl_conv ~1 (add_conv ctxt))) ctxt) i)

```

A test case for this conversion is as follows

```

lemma "P (Suc (99 + 1)) ((0 + 0)::nat) (Suc (3 + 3 + 3)) (4 + 1)"
  apply(tactic {* add_tac @context 1 *})?

```

where it produces the goal state

```

goal (1 subgoal):
  1. P (Suc 100) 0 (Suc 9) ((4::'a) + (1::'a))

```

Solution for Exercise 5.6.1.

We use the timing function *timing_wrapper* from Recipe A.3. To measure any difference between the *simproc* and *conversion*, we will create mechanically terms involving additions and then set up a goal to be simplified. We have to be careful to set up the goal so that other parts of the simplifier do not interfere. For this we construct an unprovable goal which, after simplification, we are going to “prove” with the help of “**sorry**”, that is the method *SkipProof.cheat_tac*

For constructing test cases, we first define a function that returns a complete binary tree whose leaves are numbers and the nodes are additions.

```

fun term_tree n =
let
  val count = ref 0;

  fun term_tree_aux n =
    case n of
      0 => (count := !count + 1; HOLogic.mk_number @{typ nat} (!count))
    | _ => Const (@{const_name "plus"}, @{typ "nat=>nat=>nat"})
      $ (term_tree_aux (n - 1)) $ (term_tree_aux (n - 1))
in
  term_tree_aux n
end

```

This function generates for example:

```

writeln (Syntax.string_of_term @context (term_tree 2))
> (1 + 2) + (3 + 4)

```

The next function constructs a goal of the form $P \dots$ with a term produced by `term_tree` filled in.

```
fun goal n = HOLLogic.mk_Trueprop (@{term "P::nat⇒ bool"} $ (term_tree n))
```

Note that the goal needs to be wrapped in a `Trueprop`. Next we define two tactics, `c_tac` and `s_tac`, for the conversion and `simproc`, respectively. The idea is to first apply the conversion (respectively `simproc`) and then prove the remaining goal using `cheat_tac`.

```
local
  fun mk_tac tac =
    timing_wrapper (EVERY1 [tac, K (SkipProof.cheat_tac @{theory})])
in
  val c_tac = mk_tac (add_tac @{context})
  val s_tac = mk_tac (simp_tac (HOL_basic_ss addsimprocs [@{simproc add_sp}]))
end
```

This is all we need to let the conversion run against the `simproc`:

```
val _ = Goal.prove @{context} [] [] (goal 8) (K c_tac)
val _ = Goal.prove @{context} [] [] (goal 8) (K s_tac)
```

If you do the exercise, you can see that both ways of simplifying additions perform relatively similar with perhaps some advantages for the `simproc`. That means the simplifier, even if much more complicated than conversions, is quite efficient for tasks it is designed for. It usually does not make sense to implement general-purpose rewriting using conversions. Conversions only have clear advantages in special situations: for example if you need to have control over innermost or outermost rewriting, or when rewriting rules are lead to non-termination.

Appendix C

Comments for Authors

- This tutorial can be compiled on the command-line with:

```
$ isabelle make
```

You very likely need a recent snapshot of Isabelle in order to compile the tutorial. Some parts of the tutorial also rely on compilation with PolyML.

- You can include references to other Isabelle manuals using the reference names from those manuals. To do this the following four \LaTeX commands are defined:

	Chapters	Sections
Implementation Manual	<code>\ichcite{...}</code>	<code>\isccite{...}</code>
Isar Reference Manual	<code>\rchcite{...}</code>	<code>\rsccite{...}</code>

So `\ichcite{ch:logic}` yields a reference for the chapter about logic in the implementation manual, namely [Impl. Man., Ch. 2].

- There are various document antiquotations defined for the tutorial. They allow to check the written text against the current Isabelle code and also allow to show responses of the ML-compiler. Therefore authors are strongly encouraged to use antiquotations wherever appropriate.

The following antiquotations are defined:

- `@{ML "expr" for vars in structs}` should be used for displaying any ML-expression, because the antiquotation checks whether the expression is valid ML-code. The *for*- and *in*-arguments are optional. The former is used for evaluating open expressions by giving a list of free variables. The latter is used to indicate in which structure or structures the ML-expression should be evaluated. Examples are:

<code>@{ML "1 + 3"}</code>		<code>1 + 3</code>
<code>@{ML "a + b" for a b}</code>	produce	<code>a + b</code>
<code>@{ML Ident in OuterLex}</code>		<code>Ident</code>

- `@{ML_response "expr" "pat"}` should be used to display ML-expressions and their response. The first expression is checked like in the antiquotation `@{ML "expr"}`; the second is a pattern that specifies the result the first expression produces. This pattern can contain "... " for parts that you like to omit. The response of the first expression will be checked against this pattern. Examples are:

```
@{ML_response "1+2" "3"}
@{ML_response "(1+2,3)" "(3,...)"}

```

which produce respectively

```
1+2          (1+2,3)
> 3          > (3,...)

```

Note that this antiquotation can only be used when the result can be constructed: it does not work when the code produces an exception or returns an abstract datatype (like *thm* or *cterm*).

- `@{ML_response_fake "expr" "pat"}` works just like the antiquotation `@{ML_response "expr" "pat"}` above, except that the result-specification is not checked. Use this antiquotation when the result cannot be constructed or the code generates an exception. Examples are:

```
@{ML_response_fake "cterm_of @{theory} @{term \"a + b = c\"}"
  "a + b = c"}
@{ML_response_fake "($$ \"x\") (explode \"world\")"
  "Exception FAIL raised"}

```

which produce respectively

```
cterm_of @{theory} @{term "a + b = c"}
> a + b = c
($$ "x") (explode "world")
> Exception FAIL raised

```

This output mimics to some extent what the user sees when running the code.

- `@{ML_response_fake_both "expr" "pat"}` can be used to show erroneous code. Neither the code nor the response will be checked. An example is:

```
@{ML_response_fake_both "@{cterm \"1 + True\"}"
  "Type unification failed ..."}

```

- `@{ML_file "name"}` should be used when referring to a file. It checks whether the file exists. An example is

```
@{ML_file "Pure/General/basics.ML"}

```

The listed antiquotations honour options including `[display]` and `[quotes]`. For example

`@{ML [quotes] "\"foo\" ^ \"bar\""} produces "foobar"`

whereas

`@{ML "\"foo\" ^ \"bar\""} produces only foobar`

- Functions and value bindings cannot be defined inside antiquotations; they need to be included inside **ML** `{* ... *}` environments. In this way they are also checked by the compiler. Some \LaTeX -hack in the tutorial, however, ensures that the environment markers are not printed.
- Line numbers can be printed using **ML** `%linenos {* ... *}` for ML-code or **lemma** `%linenos ...` for proofs. The tag is `%linenosgray` when the numbered text should be gray.

Bibliography

- [1] R. Bornat. In Defence of Programming. Available online via http://www.cs.mdx.ac.uk/staffpages/r_bornat/lectures/revisedinauguraltext.pdf, April 2005. Corrected and revised version of inaugural lecture, delivered on 22nd January 2004 at the School of Computing Science, Middlesex University.
- [2] R. Dyckhoff. Contraction-Free Sequent Calculi for Intuitionistic Logic. *The Journal of Symbolic Logic*, 57(3):795–807, 1992.
- [3] M. Gordon, R. Milner, and C. P. Wadsworth. *Edinburgh LCF*, volume 78 of *Lecture Notes in Computer Science*. Springer, 1979.
- [4] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*. Springer, 2002. LNCS Tutorial 2283.
- [5] D. C. Oppen. Pretty printing. *ACM Transactions on Programming Languages and Systems*, pages 465–483, 1980.
- [6] L. C. Paulson. *ML for the Working Programmer*. Cambridge University Press, 2nd edition, 1996.

Index

--, 54, 61
--->, 31
--/, 55
-->, 31
#->, 17
##>, 17
##>>, 17
#>, 15, 17
#>>, 17
!!, 56
!!! (in *OuterParse*), 62
/--, 55
/->, 16, 17
//, 55
//>, 15, 17
//>>, 16, 17
/>, 13, 17
/>>, 15, 17
\$\$, 53
\$\$\$ (in *OuterParse*), 61
>>, 58
' , 15

abs_conv (in *Conv*), 103
add_consts_i (in *Sign*), 37
add_del (in *Attrib*), 41
add_tfrees (in *Term*), 32
add_thm (in *Thm*), 40, 41
add_thms_dynamic (in *PureThy*), 44
add_tvars (in *Term*), 32
addcongs, 91
addsimprocs, 91, 97
addsimps, 91
ahead (in *Scan*), 56
all_conv (in *Conv*), 101
all_tac, 75
ALLGOALS, 85
arg_conv (in *Conv*), 103

asm_full_simp_tac, 89
asm_lr_simp_tac, 89
asm_simp_tac, 89
atac, 74, 78
atomize_term (in *ObjectLogic*), 124
axiomK (in *Thm*), 121

base_name (in *Long_Name*), 31
beta_conversion (in *Thm*), 101
binder_types, 14
blk (in *Pretty*), 45, 46
Bound, 28
breaks (in *Pretty*), 45

case_names (in *RuleCases*), 137
cat_lines, 12
CHANGED, 86
cheat_tac (in *SkipProof*), 78
check_terms (in *Syntax*), 17
chop, 131
comb_conv (in *Conv*), 104
combination_conv (in *Conv*), 103
Command (in *OuterLex*), 60
commas, 10, 11
commas (in *Pretty*), 46
concl_conv (in *Conv*), 105
COND, 88
context (in *Args*), 63
CONVERSION, 105
CSUBGOAL, 82
cterm_of, 33
curry, 14
cut_facts_tac, 79

debug (in *Toplevel*), 10
declaration_attribute (in *Thm*), 40, 41
define (in *LocalTheory*), 20, 121
definitionK (in *Thm*), 121

- del_thm* (in *Thm*), 40
- delcongs*, 91
- delsimprocs*, 91
- delsimps*, 91
- DEPTH_SOLVE*, 89
- dest_number* (in *HOLogic*), 99
- dest_ss* (in *Simplifier*), 92
- dest_ss* (in *MetaSimplifier*), 19
- DETERM*, 88
- dresolve_tac*, 79
- dtac*, 78
- dummyT*, 35

- else_conv*, 102
- empty_binding* (in *Attrib*), 121
- empty_ss*, 91
- enum* (in *Pretty*), 46
- enum1* (in *OuterParse*), 62
- eq_thm_prop* (in *Thm*), 40
- eresolve_tac*, 79
- error*, 10
- error* (in *Scan*), 57
- etac*, 74, 78
- EVERY'*, 84
- EVERY1*, 85
- explode*, 54
- explode* (in *Symbol*), 54
- extern_const* (in *Sign*), 31

- fastype_of*, 14, 34
- fconv_rule* (in *Conv*), 105
- finite* (in *Scan*), 57
- FIRST'*, 85
- FIRST1*, 85
- fixes* (in *OuterParse*), 65
- FOCUS* (in *Subgoal*), 81, 130
- ftac*, 78
- full_simp_tac*, 89
- fun_conv* (in *Conv*), 103

- HOL_basic_ss*, 91, 92
- HOL_ss*, 92

- I*, 12
- Ident* (in *OuterLex*), 60
- import* (in *Variable*), 11
- indent* (in *Pretty*), 46
- inline* (in *ML_Antiquote*), 20

- instantiate'* (in *Drule*), 125
- internalK* (in *Thm*), 121
- is_proper* (in *OuterLex*), 60

- K*, 12, 13
- Keyword* (in *OuterLex*), 62
- keyword* (in *OuterKeyword*), 60, 69

- lambda*, 27
- lift* (in *Scan*), 59
- list_comb*, 14, 27
- local_theory* (in *OuterSyntax*), 57, 67, 116

- makestring* (in *PolyML*), 9
- map_index*, 135
- map_types*, 32
- Mixfix*, 66
- mk_binop* (in *HOLogic*), 151
- mk_eq* (in *HOLogic*), 29
- ML_source* (in *OuterParse*), 63
- MRL*, 80
- MRS*, 80, 134

- no_conv* (in *Conv*), 101
- no_tac*, 75
- NoSyn*, 66
- not_eof* (in *Symbol*), 58
- note* (in *LocalTheory*), 71, 136

- one* (in *Scan*), 54, 58
- opt_thm_name* (in *SpecParse*), 66
- option* (in *Scan*), 56
- ORELSE'*, 85

- params_conv* (in *Conv*), 105
- parse* (in *YXML*), 64
- prems_conv* (in *Conv*), 105
- Pretty.T* (type), 44
- pretty_term* (in *Syntax*), 48
- pretty_ttyp* (in *Syntax*), 48
- print_depth*, 61
- print_tac*, 78
- profiling* (in *Toplevel*), 10
- prop* (in *OuterParse*), 64, 66
- prop_of*, 11
- prove* (in *Goal*), 36, 74, 127

- quote* (in *Pretty*), 48

read_prop (in *Syntax*), 71
read_spec (in *Specification*), 117
read_term_pattern (in *ProofContext*),
 20
REPEAT, 86
repeat (in *Scan*), 57, 58
REPEAT1, 87
repeat1 (in *Scan*), 57
REPEAT_ALL_NEW, 87
reserved (in *OuterParse*), 61
resolve_tac, 79
rewr_conv (in *Conv*), 102
rewrite_goals_tac, 93
RS, 38, 80
RSN, 80
rtac, 74, 78, 79
rule_attribute (in *Thm*), 38, 40
rulify (in *ObjectLogic*), 36, 132

scan (in *OuterSyntax*), 60
setup (in *Attrib*), 39
simp_tac, 89
SIMPLE_METHOD, 72
stopper (in *Symbol*), 57
string_of (in *Pretty*), 45
string_of_term (in *Syntax*), 10
SUBGOAL, 82
SUBPROOF, 81, 130
subst_bounds, 28
subst_free, 28, 128
succeed (in *Scan*), 69

tap, 15
term (in *OuterParse*), 64
term_of, 11
THEN, 74
THEN', 75
then_conv, 102
theorem_i (in *Proof*), 71
theory_name (in *Context*), 18
theory_of (in *ProofContext*), 124
this_string (in *Scan*), 55
thm_name (in *SpecParse*), 66
thy_decl (in *OuterKeyword*), 70, 116
thy_goal (in *OuterKeyword*), 70
tracing, 9, 10
TRY, 86, 88
try_conv (in *Conv*), 103

TRYALL, 88
type_of, 34
unless (in *Scan*), 58

variant_frees (in *Variable*), 14

where_alt_specs (in *SpecParse*), 66
writeln, 9, 10